

# BASIC COMPUTER

David H. Ahl

# Spill

**Bind 3**

BORGEN









**BASIC Computerspil  
bind 3**

*BASIC Computerspil, 3*

Oversat fra amerikansk af Christian Borup.

Originalens titel:

BASIC Computer Games

Copyright © 1978 by Creative Computing Press

Danske udgave © Borgens Forlag 1986

Omslag: Jesper Bach Jensen

Tegninger: George Beker

Published by Borgens Forlag, Valbygaardsvej 33,

DK-2500 Copenhagen Valby

Trykt hos Narayana Press, Gylling

ISBN 87-418-7686-5

1. udgave. 1. oplag. 1986

**David Ahl**

# **BASIC** **computerspil**

**Bind 3**

**Redigeret af David H. Ahl**

**Programændringer: Steve North**

**Dansk oversættelse, bearbejdning og programmering:  
Christian Borup**

**Borgen**

# Forord til den danske udgave

I den danske udgave af Ahl's »BASIC Computer Games« er der anvendt to forskellige programformater. Det ene er beregnet for en skærm med en linielængde på 40 tegn, mens det andet format er beregnet til en skærm med en linielængde på 80. Visse af bogens programmer er beregnet til kørsel på linieskriver, og derfor kan kørslerne af disse programmer virke forvirrende ved første øjekast.

Der er overalt anvendt så »skrabet« en BASIC som muligt, og visse af programmerne er grundet forskelle mellem engelsk og dansk grammatik revideret kraftigt. Det samme gælder for visse af programmernes udskriftsrutiner.

I den danske oversættelse er der, i bogens appendiks, tilføjet programmet BASIC-TEST, der giver mulighed for at se en fuld prøvekørsel af et program, der indeholder alle de BASIC-instruktioner som resten af bogens programmer indeholder. Det anbefales derfor, at man starter med at indtaste og prøvekøre BASIC-TEST, idet man derved hurtigt får et overblik over, hvor eventuelle afvigelser fra bogens BASIC kan optræde.

Christian Borup  
Oktober 1985



# Indhold · bind 1

Indledning · 6	
Om denne bog · 6	
BASIC · 7	
Oversættelse til andre former for BASIC · 9	
Til slut · 10	
Forord til den danske udgave · 11	
Spillene:	
AKTIER · 12	
AMERIKANSK FODBOLD · 15	
AMERIKANSK FODBOLD, VERSION 2 · 19	
AWARI · 22	
BAGELS · 24	
BAGLÆNS · 26	
BANNER · 28	
BASKETBALL · 30	
BILLE · 33	
BOGSTAV · 36	
BOKSNING · 37	
BOMBARDEMENT · 39	
BONDESPIL · 41	
BORGERKRIG · 44	
BOWLING · 49	
BUNKE · 52	
BYTTEPENGE · 53	
CIFFER · 54	
CRAPS · 56	
DAM · 57	
DAM FOR EN · 59	
DRONNINGESKAK · 61	
DYRERIGET · 63	
ENOGTYVE · 65	
FLIP-FLOP · 68	
FRASEGENERATOR · 69	
GODDAV · 70	
GOLF · 72	
Appendix · 75	
Fortegnelse over spillene opdelt i kategorier · 77	

# Indhold · bind 2

Om denne bog · 6
GOMOKO · 7
GÆT · 8
HAMMURABI · 9
HAPS · 11
HASARD · 13
HESTEVÆDDELØB · 14
HOP · 16
HURKLE · 17
HÆNGT MAND · 18
HØJT-LAVT · 21
HØVDING · 22
ISHOCKEY · 23
KALENDER · 27
KAMP · 29
KANIN · 31
KAST · 32
KAST BOMBERNE · 33
KEMIKER · 35
KONGE · 36

KRIG · 39
KRYDS OG BOLLE · 40
KRYDS OG BOLLE, VERSION 2 · 41
KRYDS OG BOLLE I 3-D · 42
KUBUS · 45
LABYRINT · 47
LIGE VINDER · 49
LIGE VINDER, VERSION 2 · 50
LITTERATUR QUIZ · 52
LIV · 53
LIV FOR TO · 56
LOVE · 58
MASTER MIND · 59
MATEMATISK TERNING · 62
MÅL · 63
MÅNELANDING · 65
MÅNELANDING. L.E.M. · 67
MÅNELANDING, RAKET · 70
NAVN · 72
Appendix · 73
Fortegnelse over spillene opdelt i kategorier · 75

# Indhold · bind 3

Om denne bog · 6
NICOMACHUS · 7
NIM · 8
OMRING · 10
ORBIT · 11
ORD · 13
PELSHANDLER · 14
PILESPIL · 16
PIZZA · 18
PLOT I 3-D · 20
POESI · 21
POKER · 22
ROULETTE · 26
RUDERE · 29
RUSSISK ROULETTE · 30
SAKS-STEN-PAPIR · 31
SINUSKURVE · 32
SKYTTE · 33
SLAGSKIB · 34
SLALOM · 37
SMUGORM · 39
SOLITAIRE · 41
SPLAT · 44
STJERNER · 46
SUPER STAR TREK INSTRUKTIONER · 47
SUPER STAR TREK X · 50
SYNONYM · 57
SÆNKE SLAGSKIBE · 58
TAL · 61
TERNING · 62
TOG · 63
TYREFÆGTNING · 64
TYVEKNÆGT · 66
TÆNDSTIKKER · 68
TÅRNET I HANOI · 69
UBÅDSJAGT · 71
UGEDAG · 72
Appendix · 74
Fortegnelse over spillene opdelt i kategorier · 76

# Om denne bog

BASIC computerspil: Nærværende mikrodatamat-udgave er en kraftig bearbejdet version af min første bog: »101 BASIC Computer Games«, der i 1973 blev udgivet af Digital Equipment Corporation. Programmerne i den første bog var skrevet i seks forskellige versioner af BASIC. Programmerne i nærværende bog er alle skrevet i Microsoft BASIC. I den første bog fandtes flere programmer, der enten ikke kunne omskrives til Microsoft BASIC, eller som ikke var besværet værd. Disse programmer er fjernet og erstattet af nye. Jeg siger tak til alle, som med tålmodighed og forstørrelsesglas tydede de næsten ulæselige programmer i den første bog og fik dem til at virke.

David H. Ahl  
September 1978

## Om forfatteren:

David H. Ahl har været interesseret i datamater i mange år. I 1956 lærte han at programmere i ALGOL og CAL (Cornell Assembly Language) på en Burroughs B-200. Siden da har han prøvet kræfter med så forskellige uhyrer som: Bendix G-15 og G-20, IBM 650, 704, 7090, 1401, 1130, 360/40 og 360/50, CDC 3200 og 3600, GE 115, 235 og 635, HP 2000, NCR Century 50 og 100, DEC PDP-8, 10, 11, 12 og 15 samt så at sige alle eksisterende mikrodatamater.

Han skrev sit første computerspil allerede ugen efter, at han havde lært at programmere. Siden da har han arbejdet med både alvorlige spil (som f.eks. Carnegie-Mellon Management Game og energisimulering i USA) og underholdningsspil.

Hans liv har altid været knyttet til datamater inden for områder som marketing, produktprognose, pædagogisk forskning, salg af datamater til skoler og universiteter, og han har fungeret som lærer og forelæser. Som udgiver af det amerikanske tidsskrift »Creative Computing« har han fortsat med at udvikle nye områder, hvor datamater kan anvendes.

# Nicomachus

En af de ældste former for matematiske hovedbrud kaldes nogle gange for »boomerang-spil«. Alle mennesker er vel på et eller andet tidspunkt i deres liv blevet udsat for at skulle »tænke på et tal«, at gøre nogle beregninger på det, hvorefter den, der spurgte, kan fortælle, hvad det var for et tal, man tænkte på. Der findes hundredvis af varianter af dette spil.

Det ældste dokumenterede eksempel synes at være det som findes i »Arithmetica«, skrevet af Nicomachus, som døde ca. år 120. Han beder dig tænke på et tal, dele det med 3, 5 og 7, og derefter fortælle hvad der blev til rest ved hver division. Når han har fået resterne at vide, så kan han med det samme sige, hvilket tal du tænkte på.

Kan du komme på en simpel metode, hvor du kan klare denne opgave i hovedet? Hvis ikke, må du kigge i programlinje 390, hvor du kan se, hvorledes de gamle matematikere løste dette problem.

Nicomachus er skrevet af David Ahl.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
NICOMACHUS  
-----

BOOMERANG-SPIL FRA 'ARITHMETICA' AF  
NICOMACHUS - 90 ÅR F.KR.

TÆNK PÅ ET TAL MELLEM 1 OG 100.

NÅR DIT TAL DELES MED 3,  
HVAD BLIVER DER SÅ TIL REST ?1

NÅR DIT TAL DELES MED 5,  
HVAD BLIVER DER SÅ TIL REST ?0

NÅR DIT TAL DELES MED 7,  
HVAD BLIVER DER SÅ TIL REST ?5

LAD MIG LIGE TÆNKE ET ØJEBLIK.....

DIT TAL VAR 40

IKKE SANDT ?J

HVAD SIGER DU TIL DET??

LAD OS PRØVE ET NYT TAL.

TÆNK PÅ ET TAL MELLEM 1 OG 100.

NÅR DIT TAL DELES MED 3,  
HVAD BLIVER DER SÅ TIL REST ?1

NÅR DIT TAL DELES MED 5,  
HVAD BLIVER DER SÅ TIL REST ?0

NÅR DIT TAL DELES MED 7,  
HVAD BLIVER DER SÅ TIL REST ?2

LAD MIG LIGE TÆNKE ET ØJEBLIK.....

DIT TAL VAR 100

IKKE SANDT ?J

HVAD SIGER DU TIL DET??

LAD OS PRØVE ET NYT TAL.

TÆNK PÅ ET TAL MELLEM 1 OG 100.

NÅR DIT TAL DELES MED 3,  
HVAD BLIVER DER SÅ TIL REST ?0

NÅR DIT TAL DELES MED 5,  
HVAD BLIVER DER SÅ TIL REST ?4

NÅR DIT TAL DELES MED 7,  
HVAD BLIVER DER SÅ TIL REST ?4

LAD MIG LIGE TÆNKE ET ØJEBLIK.....

DIT TAL VAR 39

IKKE SANDT ?N

DU ER VIST IKKE SÅ GOD TIL AT REGNE?

LAD OS PRØVE ET NYT TAL.

TÆNK PÅ ET TAL MELLEM 1 OG 100.

NÅR DIT TAL DELES MED 3,  
HVAD BLIVER DER SÅ TIL REST ?1

NÅR DIT TAL DELES MED 5,  
HVAD BLIVER DER SÅ TIL REST ?1

NÅR DIT TAL DELES MED 7,  
HVAD BLIVER DER SÅ TIL REST ?1

LAD MIG LIGE TÆNKE ET ØJEBLIK.....

DIT TAL VAR 1

IKKE SANDT ?J

HVAD SIGER DU TIL DET??

LAD OS PRØVE ET NYT TAL.

TÆNK PÅ ET TAL MELLEM 1 OG 100.

NÅR DIT TAL DELES MED 3,  
HVAD BLIVER DER SÅ TIL REST ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          NICOMACHUS"
130 PRINT "-----"
140 PRINT
150 PRINT "BOOMERANG-SPIL FRA 'ARITHMETICA' AF"
160 PRINT "NICOMACHUS - 90 ÅR F.KR."
170 PRINT
180 PRINT "TÆNK PÅ ET TAL MELLEM 1 OG 100."
190 PRINT
200 PRINT "NÅR DIT TAL DELES MED 3,"
210 PRINT "HVAD BLIVER DER SÅ TIL REST ";
220 INPUT A
230 PRINT
240 PRINT "NÅR DIT TAL DELES MED 5,"
250 PRINT "HVAD BLIVER DER SÅ TIL REST ";
260 INPUT B
270 PRINT
280 PRINT "NÅR DIT TAL DELES MED 7,"
290 PRINT "HVAD BLIVER DER SÅ TIL REST ";
300 INPUT C
310 PRINT
320 PRINT "LAD MIG LIGE TÆNKE ET ØJEBLIK....."
330 PRINT
340 FOR I = 1 TO 1500
350 NEXT I
360 D = 70 * A + 21 * B + 15 * C
370 IF D < = 105 THEN 400
380 D = D - 105
390 GOTO 370
400 PRINT "DIT TAL VAR";D
410 PRINT
420 PRINT "IKKE SANDT ";
430 INPUT AS
440 PRINT
450 IF LEFT$(AS,1) = "J" THEN 520
460 IF LEFT$(AS,1) = "N" THEN 540
470 PRINT "HVAD I £%*(%)£ SIGER DU?"
480 PRINT AS;" FORSTÅR JEG IKKE"
490 PRINT "SKRIV 'JA' ELLER 'NEJ'."
500 PRINT
510 GOTO 430
520 PRINT "HVAD SIGER DU TIL DET??"
530 GOTO 550
540 PRINT "DU ER VIST IKKE SÅ GOD TIL AT REGNE?"
550 PRINT
560 PRINT "LAD OS PRØVE ET NYT TAL."
570 GOTO 170
580 END
```

# Nim

Nim er et af de ældste spil, mennesket kender til, og det antages, at spillet er opstået i det gamle Kina. Navnet – som spillet fik af en af de første matematikere, der analyserede det – kommer fra et gammelt germansk udsagnsord, der betyder »at stjæle« eller »at tage bort«.

Nogle objekter (brikker, spillemærker, etc.) lægges op mellem de to spillere. F.eks. følgende:

```
X X X X X X X 1. række, 7 objekter
X X X X X      2. række, 5 objekter
X X X          3. række, 3 objekter
X              4. række, 1 objekt
```

Spillerne skiftes til at fjerne objekter, til der ikke er flere tilbage. Den, som fjerner det sidste objekt vinder. Trækkene foretages ifølge nedenstående regler:

1. Du kan frit vælge den række og det antal objekter, du vil tage. Men man kan selvfølgelig ikke tage flere, end der findes i en række.
2. Du må ikke springe et objekt over, og du må ikke tage nul objekter.

Vinderstrategien kan defineres matematisk, men i stedet for at jeg fortæller dig, hvad den går ud på, får du lov til at finde ud af den selv. Et godt tip er at spille nogle omgange med datamaten og på et stykke papir skrive ned, hvor mange objekter der var tilbage i bunken efter hvert træk. Skriv antallet med binære tal. Er der et mønster, der gentager sig?

Denne version af Nim kommer fra Dartmouth College og er et generaliseret spil, som giver dig mulighed for at angive størrelserne på de fire rækker, og du kan også bestemme, hvilken gevinstvariant der skal gælde, dvs. det gælder om at tage det sidste objekt eller om at undgå at tage det sidste objekt. Hvis du vil spille klassisk Nim, skal du indtaste 7, 5, 3 og 1 for antal objekter pr. række, og gevinstalternativ 1.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

NIM

DETTE ER SPILLET 'NIM'.

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

SPILLET SPILLES MED ET ANTAL BUNKER MED OBJEKTER. EFTER TUR SKIFTES DU OG DATAMATEN TIL AT TAGE ET ANTAL OBJEKTER FRA BUNKEN.

NÅR DET ER DIN TUR KAN DU TAGE ALLE DE OBJEKTER DER ER TILBAGE I EN BUNKE, MEN DU SKAL TAGE MINDST 1 OBJEKT, OG DET ER KUN TILLADT AT TAGE FRA EN BUNKE AD GANGEN.

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

DU SKAL VED SPILLET BEGYNDELSE SELV OPLYSE OM GEVINST AFGØRES VED AT TAGE ELLER IKKE TAGE DET SIDSTE OBJEKT.

DU SKAL OGSÅ BESTEMME ANTAL BUNKER, SAMT HVOR MANGE OBJEKTER DER SKAL VÆRE I HVER BUNKE NÅR SPILLET STARTER.

BUNKERNE KAN INDEHOLDE FORSKELLIGE ANTAL OBJEKTER.

DATAMATEN VISER SINE TRÆK VED AT UD-SKRIVE HVER BUNKE OG DET ANTAL OBJEKTER DER ER TILBAGE I BUNKERNE EFTER HVERT TRÆK SOM DATAMATEN FORETAGER.

INDTAST GEVINSTALTERNATIV

1=TAG SIDSTE 2= UNDGÅ SIDSTE ?2

INDTAST ANTAL BUNKER ?4

INDTAST ANTAL OBJEKTER

ANTAL OBJEKTER I BUNKE 1 ?10

ANTAL OBJEKTER I BUNKE 2 ?8

ANTAL OBJEKTER I BUNKE 3 ?7

ANTAL OBJEKTER I BUNKE 4 ?3

VIL DU BEGYNDE ?N

BUNKE	STØRRELSE
1	10
2	8
3	1
4	3

DIT TRÆK

INDTAST BUNKE,ANTAL DER TAGES ?1,9

BUNKE	STØRRELSE
1	1
2	3
3	1
4	3

DIT TRÆK

INDTAST BUNKE,ANTAL DER TAGES ?2,2

BUNKE	STØRRELSE
1	1
2	1
3	1
4	0

DIT TRÆK

INDTAST BUNKE,ANTAL DER TAGES ?3,1

DATAMATEN VINDER.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                NIM"
130 PRINT "-----"
140 PRINT
150 DIM A(100),B(100,10),D(2)
160 PRINT "DETTE ER SPILLET 'NIM'."
170 PRINT
180 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
190 INPUT Z$
200 PRINT
210 IF LEFT$(Z$,1) = "N" THEN 540
220 IF LEFT$(Z$,1) = "J" THEN 260
230 PRINT "TAST 'J' FOR JA ELLER 'N' FOR NEJ. ";
240 PRINT
250 GOTO 190
260 PRINT "SPILLET SPILLES MED ET ANTAL BUNKER MED"
270 PRINT "OBJEKTER. EFTER TUR SKIFTES DU OG DATA-"
280 PRINT "MATEN TIL AT TAGE ET ANTAL OBJEKTER FRA"
290 PRINT "BUNKEN."
300 PRINT
310 PRINT "NÅR DET ER DIN TUR KAN DU TAGE ALLE DE"
320 PRINT "OBJEKTER DER ER TILBAGE I EN BUNKE, MEN"
330 PRINT "DU SKAL TAGE MINDST 1 OBJEKT, OG DET ER"
340 PRINT "KUN TILLADT AT TAGE FRA EN BUNKE AD"
350 PRINT "GANGEN."
360 PRINT
```

```

370 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
380 INPUT A$
390 PRINT
400 PRINT "DU SKAL VED SPILLET BEGYNDELSE SELV"
410 PRINT "OPLYSE OM GEVINST AFGØRES VED AT TAGE"
420 PRINT "ELLER IKKE TAGE DET SIDSTE OBJEKT."
430 PRINT
440 PRINT "DU SKAL OGSÅ BESTEMME ANTAL BUNKER,"
450 PRINT "SAMT HVOR MANGE OBJEKTER DER SKAL VÆRE"
460 PRINT "I HVER BUNKE NÅR SPILLET STARTER."
470 PRINT
480 PRINT "BUNKERNE KAN INDEHOLDE FORSKELLIGE"
490 PRINT "ANTAL OBJEKTER."
500 PRINT "DATAMATEN VISER SINE TRÆK VED AT UD-"
510 PRINT "SKRIVE HVER BUNKE OG DET ANTAL OBJEKTER"
520 PRINT "DER ER TILBAGE I BUNKERNE EFTER HVERT"
530 PRINT "TRÆK SOM DATAMATEN FORETAGER."
540 PRINT
550 PRINT "INDTAST GEVINSTALTERNATIV"
560 PRINT "1=TAG SIDSTE 2= UNDGÅ SIDSTE ";
570 INPUT W
580 PRINT
590 IF W = 1 THEN 610
600 IF W < > 2 THEN 550
610 PRINT "INDTAST ANTAL BUNKER ";
620 INPUT N
630 PRINT
640 IF N > 100 THEN 610
650 IF N < 1 THEN 610
660 IF N < > INT (N) THEN 610
670 PRINT "INDTAST ANTAL OBJEKTER"
680 PRINT "-----"
690 FOR I = 1 TO N
700 PRINT "ANTAL OBJEKTER I BUNKE";I;
710 INPUT A(I)
720 PRINT
730 IF A(I) > 2000 THEN 700
740 IF A(I) < 1 THEN 700
750 IF A(I) < > INT (A(I)) THEN 700
760 NEXT I
770 PRINT "VIL DU BEGYNDE ";
780 INPUT Q9$
790 PRINT
800 IF LEFT$(Q9$,1) = "J" THEN 1630
810 IF LEFT$(Q9$,1) = "N" THEN 850
820 PRINT "VENLIGST 'J' ELLER 'N'"
830 PRINT
840 GOTO 780
850 IF W = 1 THEN 1110
860 C = 0
870 FOR I = 1 TO N
880 IF A(I) = 0 THEN 920
890 C = C + 1
900 IF C = 3 THEN 1010
910 D(C) = I
920 NEXT I
930 IF C = 2 THEN 1090
940 IF A(D(1)) > 1 THEN 980
950 PRINT "DATAMATEN HAR TABT."
960 PRINT
970 GOTO 1860
980 PRINT "DATAMATEN VINDER."
990 PRINT
1000 GOTO 1860
1010 C = 0
1020 FOR I = 1 TO N
1030 IF A(I) > 1 THEN 1110
1040 IF A(I) = 0 THEN 1060
1050 C = C + 1
1060 NEXT I
1070 IF C / 2 < > INT (C / 2) THEN 950
1080 GOTO 1110
1090 IF A(D(1)) = 1 THEN 980
1100 IF A(D(2)) = 1 THEN 980
1110 FOR I = 1 TO N
1120 E = A(I)
1130 FOR J = 0 TO 10
1140 F = E / 2
1150 B(I,J) = 2 * (F - INT (F))
1160 E = INT (F)
1170 NEXT J
1180 NEXT I
1190 FOR J = 10 TO 0 STEP - 1
1200 C = 0
1210 H = 0
1220 FOR I = 1 TO N
1230 IF B(I,J) = 0 THEN 1280
1240 C = C + 1

```

```

1250 IF A(I) < = H THEN 1280
1260 H = A(I)
1270 G = I
1280 NEXT I
1290 IF C / 2 < > INT (C / 2) THEN 1360
1300 NEXT J
1310 E = INT (N * RND (1) + 1)
1320 IF A(E) = 0 THEN 1310
1330 F = INT (A(E) * RND (1) + 1)
1340 A(E) = A(E) - F
1350 GOTO 1550
1360 A(G) = 0
1370 FOR J = 0 TO 10
1380 B(G,J) = 0
1390 C = 0
1400 FOR I = 1 TO N
1410 IF B(I,J) = 0 THEN 1430
1420 C = C + 1
1430 NEXT I
1440 A(G) = A(G) + 2 * (C / 2 - INT (C / 2)) * 2 ^ J
1450 NEXT J
1460 IF W = 1 THEN 1550
1470 C = 0
1480 FOR I = 1 TO N
1490 IF A(I) > 1 THEN 1550
1500 IF A(I) = 0 THEN 1520
1510 C = C + 1
1520 NEXT I
1530 IF C / 2 < > INT (C / 2) THEN 1550
1540 A(G) = 1 - A(G)
1550 PRINT "BUNKE          STØRRELSE"
1560 PRINT "-----          -----"
1570 FOR I = 1 TO N
1580 PRINT " ";I;"          ";A(I)
1590 NEXT I
1600 IF W = 2 THEN 1630
1610 GOSUB 1790
1620 IF Z = 1 THEN 980
1630 PRINT
1640 PRINT "DIT TRÆK"
1650 PRINT
1660 PRINT "INDTAST BUNKE,ANTAL DER TAGES ";
1670 INPUT X,Y
1680 PRINT
1690 IF X > N THEN 1630
1700 IF X < 1 THEN 1630
1710 IF X < > INT (X) THEN 1630
1720 IF Y > A(X) THEN 1630
1730 IF Y < 1 THEN 1630
1740 IF Y < > INT (Y) THEN 1630
1750 A(X) = A(X) - Y
1760 GOSUB 1790
1770 IF Z = 1 THEN 950
1780 GOTO 850
1790 Z = 0
1800 FOR I = 1 TO N
1810 IF A(I) = 0 THEN 1830
1820 RETURN
1830 NEXT I
1840 Z = 1
1850 RETURN
1860 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1870 INPUT Q9$
1880 PRINT
1890 IF LEFT$(Q9$,1) = "J" THEN 1940
1900 IF LEFT$(Q9$,1) = "N" THEN 1950
1910 PRINT "VENLIGST: SVAR 'JA' ELLER 'NEJ'."
1920 PRINT
1930 GOTO 1870
1940 GOTO 540
1950 END

```

# Omring

Dette er endnu et program af typen »gæt det hemmelige tal«. I Omring vælger datamaten et tilfældigt tal mellem 1 og 200 (eller en anden maksimumgrænse (N), bestemt af programlinie 160). Du skal derefter forsøge at gætte tallet. Ved hvert gæt skal du angive to tal, idet du skal forsøge at »omringe« det hemmelige tal med dine to gæt. Derefter fortæller datamaten dig, om dens tal er højere, lavere end din omringning, eller om det er lykkedes dig at omringe det hemmelige tal.

For at vinde spillet skal du angive det hemmelige tal ved at indtaste den samme værdi for begge gæt. Du har 6 forsøg, men dette kan du selv lave om i programlinie 150 (G), hvilket især er en god idé hvis du ændrer på maksimumsværdien i linie 160.

Når du har spillet Gæt, Stjerner og Omring, så prøv at sammenligne de strategier, der blev anvendt i de forskellige spil. Kan du finde forskelle mellem dem? Ligheder? Kan du selv skrive et sådant gætte-spil med en helt anden strategi?

Omring opstod efter et forslag fra en 10-årig, da han spillede Gæt. Det blev oprindeligt programmeret af Steve Ullmann og blev modificeret til sin nuværende form af Bob Albrecht fra People's Computer Company.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
                        OMRING  
-----
```

```
ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J
```

```
JEG TÆNKER PÅ ET TAL MELLEM 1 OG 100 .
```

```
PRØV AT GÆTTE DET TAL JEG TÆNKER PÅ!
```

```
DU SKAL INDTASTE TO TAL, OG  
DERVED FØRSØGE AT 'OMRINGE' MIT TAL  
MELLEM DINE TO TAL. JEG FORTÆLLER  
DIG OM DU HAR 'OMRINGET' MIT TAL,  
ELLER OM MIT TAL ER MINDRE ELLER  
STØRRE END DINE TO TAL.
```

```
HVIS DU ØNSKER AT GÆTTE ET ENKELT TAL,  
SÅ SKAL DU BLOT INDTASTE TO ENS TAL.
```

```
DU FÅR 6 GÆT TIL OPGAVEN.
```

```
GÆT NUMMER 1 -- ?30,70
```

```
MIT TAL ER STØRRE END DINE 2 TAL.
```

```
GÆT NUMMER 2 -- ?80,90
```

```
MIT TAL ER MINDRE END DINE 2 TAL.
```

```
GÆT NUMMER 3 -- ?71,79
```

```
DU HAR FANGET MIT TAL.
```

```
GÆT NUMMER 4 -- ?74,76
```

```
DU HAR FANGET MIT TAL.
```

```
GÆT NUMMER 5 -- ?75,75
```

```
MIT TAL ER MINDRE END DINE 2 TAL.
```

```
GÆT NUMMER 6 -- ?74,74
```

```
DU GÆTTEDE DET!!!
```

```
PRØV IGEN.
```

```
GÆT NUMMER 1 -- ?30,70
```

```
MIT TAL ER MINDRE END DINE 2 TAL.
```

```
GÆT NUMMER 2 -- ?10,20
```

```
MIT TAL ER STØRRE END DINE 2 TAL.
```

```
GÆT NUMMER 3 -- ?21,29
```

```
DU HAR FANGET MIT TAL.
```

```
GÆT NUMMER 4 -- ?23,27
```

```
MIT TAL ER STØRRE END DINE 2 TAL.
```

```
GÆT NUMMER 5 -- ?28,28
```

```
DU GÆTTEDE DET!!!
```

```
PRØV IGEN.
```


```
GÆT NUMMER 1 -- ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                        OMRING"  
130 PRINT "-----"  
140 PRINT  
150 G = 6  
160 N = 100  
170 REM Omring  
180 REM Af Steve Ullman, 1/8 1972  
190 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";  
200 INPUT Z$  
210 PRINT  
220 IF LEFT$(Z$,1) < > "J" THEN 380  
230 PRINT "JEG TÆNKER PÅ ET TAL MELLEM 1 OG";N;"."  
240 PRINT  
250 PRINT "PRØV AT GÆTTE DET TAL JEG TÆNKER PÅ!"  
260 PRINT  
270 PRINT "DU SKAL INDTASTE TO TAL, OG"  
280 PRINT "DERVED FØRSØGE AT 'OMRINGE' MIT TAL"  
290 PRINT "MELLEM DINE TO TAL. JEG FORTÆLLER"  
300 PRINT "DIG OM DU HAR 'OMRINGET' MIT TAL,"  
310 PRINT "ELLER OM MIT TAL ER MINDRE ELLER"  
320 PRINT "STØRRE END DINE TO TAL."  
330 PRINT  
340 PRINT "HVIS DU ØNSKER AT GÆTTE ET ENKELT TAL,"  
350 PRINT "SÅ SKAL DU BLOT INDTASTE TO ENS TAL."  
360 PRINT  
370 PRINT "DU FÅR";G;"GÆT TIL OPGAVEN."  
380 X = INT (N * RND (1)) + 1  
390 FOR Q = 1 TO G  
400 PRINT  
410 PRINT "GÆT NUMMER";Q;" -- ";  
420 INPUT Z,B  
430 PRINT  
440 IF Z = B AND X = Z THEN 640  
450 IF Z < = B THEN 470  
460 GOSUB 600  
470 IF Z < = X AND X < = B THEN 530  
480 IF X < Z THEN 510  
490 PRINT "MIT TAL ER STØRRE END DINE 2 TAL."  
500 GOTO 540  
510 PRINT "MIT TAL ER MINDRE END DINE 2 TAL."  
520 GOTO 540  
530 PRINT "DU HAR FANGET MIT TAL."  
540 NEXT Q  
550 PRINT  
560 PRINT "DESVÆRRE, DINE";G;"GÆT ER BRUGT."  
570 PRINT  
580 PRINT "MIT TAL VAR:";X  
590 GOTO 660  
600 R = Z  
610 Z = B  
620 B = R  
630 RETURN  
640 PRINT  
650 PRINT "DU GÆTTEDE DET!!!"  
660 PRINT  
670 PRINT "PRØV IGEN."  
680 PRINT  
690 GOTO 380  
700 END
```

# Orbit

Orbit udfordrer dig til at forsøge at se tredimensionalt ved hjælp af polære koordinater. Det gælder om at detonere en photon-bombe i nærheden af et bakteriebefængt romulansk rumskib. Dette skib kredser rundt om din planet i konstant højde og periferihastighed (grader/time). Rumskibets position er gemt af et apparat, som gør det usynligt, men efter hver bombe du affyrer, får du at vide, hvor tæt på skibet den eksploderede. Udfordringen består i at træffe et usynligt objekt, der bevæger sig, med et begrænset antal bomber.


Planeten kan erstattes af et punkt i planetens centrum (kaldet origo). Skibets position kan derefter angives som en afstand fra origo og en vinkel, der viser afstanden fra planetens østlige horisont til rumskibet.



Afstanden mellem bomben og skibet beregnes ved hjælp af cosinus-formlen, der lyder:

$$D = \text{SQR}(R^2 + D1^2 + R * D1 * \text{COS}(A - A1))$$

hvor D er afstanden mellem skibet og bomben, mens R er skibets højde, og D1 er bombens højde, og (A-A1) er vinklen mellem skibet og bomben.



Orbit hed oprindeligt Spacewar og blev skrevet af Jeff Lederer fra projekt SOLO i Pittsburgh, Pennsylvania.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
 ORBIT  
 -----

ET STED OVEN OVER DIN PLANET BEFINDER DER SIG ET ROMULANSK RUMSKIB.

RUMSKIBET CIKLER KONSTANT OM DIN PLANETS NORDPOL. AFSTANDEN FRA DIN PLANETS CENTRUM VARIERER MELLEM 10000 OG 30000 KILOMETER, OG MED DETS NUVÆRENDE HASTIGHED KAN DET CIRKLE RUNDT OM DIN PLANET PÅ MELLEM 12 OG 36 TIMER.

DESVÆRRE BRUGER ROMULANERNE ET ELEKTRONSKJOLD, SÅ DU KAN IKKE SE RUMSKIBET, MEN VED HJÆLP AF ET SÆRLIGT INSTRUMENT KAN DU FÅ AT VIDE HVOR LANGT VÆK FRA RUMSKIBET DINE FOTON-BOMBER EKSPLODERER.

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?


DU SKAL NEDSKYDE RUMSKIBET PÅ 7 TIMER, FOR DET TAGER NETOP RUMSKIBET 7 TIMER AT OPSAMLE NOK ENRGI TIL AT SLIPPE UD AF DIN PLANETS TYNGDEFELT.

DIN PLANET HAR ENERGI NOK TIL AT AFFYRE EN FOTON-BOMBE I TIMEN.

VED HVER TIMES BEGYNDELSE VIL DU BLIVE BEDT OM AT ANGIVE EN VINKEL (MELLEM 0 OG 360 GRADER) OG EN AFSTAND I 100 KILOMETERS-ENHEDER (MELLEM 100 OG 300).  
 DEREFTER UDSKRIVES, HVOR LANGT VÆK FRA SKIBET DIN BOMBE EKSPLODEREDE.

EN EKSPLOSION DER LIGGER INDEN FOR EN 5000 KILOMETERS RADIUS AF DET ROMULANSKE RUMSKIB VIL ØDELÆGGE DET!

NEDENFOR ER ET DIAGRAM, DER VIL HJÆLPE DIG TIL AT FORESTILLE DIG SITUATIONEN.



X = DIN PLANET  
 O = DET ROMULANSKE RUMSKIBS BANE

PÅ OVENSTRENDTE DIAGRAM KREDSER DET ROMULANSKE RUMSKIB RUNDT OM DIN PLANET I MODSAT RETNING AF SOLEN. GLEM IKKE, AT NÅR DER ER UTILSTRÆKKELIG ENERGI VIL RUMSKIBETS HØJDE OG HASTIGHED VÆRE KONSTANT.

HELD OG LYKKE.  
 STJERNERÅDET REGNER MED DIN HJÆLP.

TIME 1 :  
 HVILKEN AFSKYDNINGSVINKEL SKAL DIN FOTON-BOMBE HAVE ?90  
 HVOR LANGT UDE SKAL DEN DETONERE ?150

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS  
 402.564675 \*10<sup>2</sup> KILOMETER  
 FRA RUMSKIBET.

TIME 2 :  
 HVILKEN AFSKYDNINGSVINKEL SKAL DIN FOTON-BOMBE HAVE ?270  
 HVOR LANGT UDE SKAL DEN DETONERE ?170

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS  
 123.074149 \*10<sup>2</sup> KILOMETER  
 FRA RUMSKIBET.

TIME 3 :  
 HVILKEN AFSKYDNINGSVINKEL SKAL DIN FOTON-BOMBE HAVE ?330  
 HVOR LANGT UDE SKAL DEN DETONERE ?150

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS  
 124.142719 \*10<sup>2</sup> KILOMETER  
 FRA RUMSKIBET.

TIME 4 :  
 HVILKEN AFSKYDNINGSVINKEL SKAL DIN FOTON-BOMBE HAVE ?300  
 HVOR LANGT UDE SKAL DEN DETONERE ?150

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS  
 133.997518 \*10<sup>2</sup> KILOMETER  
 FRA RUMSKIBET.

TIME 5 :  
 HVILKEN AFSKYDNINGSVINKEL SKAL DIN FOTON-BOMBE HAVE ?150  
 HVOR LANGT UDE SKAL DEN DETONERE ?150

```

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS
 402.564675 *10^2 KILOMETER
FRA RUMSKIBET.

TIME 6 :
HVILKEN AFSKYDNINGSVINKEL SKAL DIN
FOTON-BOMBE HAVE ?350
HVOR LANGT UDE SKAL DEN DETONERE ?150

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS
 105.384835 *10^2 KILOMETER
FRA RUMSKIBET.

TIME 7 :
HVILKEN AFSKYDNINGSVINKEL SKAL DIN
FOTON-BOMBE HAVE ?30
HVOR LANGT UDE SKAL DEN DETONERE ?150

DIN FOTON-BOMBE EKSPLODEREDE PRÆCIS
 124.142719 *10^2 KILOMETER
FRA RUMSKIBET.

DU HAR TILLADT ROMULANERNE AT SLIPPE
BORT. ET NYT ROMULANSK RUMSKIB ER KOM-
MET INDEN FOR DIN PLANETS LUFTRUM.

ØNSKER DU AT ØDELÆGGE DET ?NEJ
FARVEL.....

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                ORBIT"
130 PRINT "-----"
140 PRINT
150 PRINT "ET STED OVEN OVER DIN PLANET BEFINDER"
160 PRINT "DER SIG ET ROMULANSK RUMSKIB."
170 PRINT
180 PRINT "RUMSKIBET CIKLER KONSTANT OM DIN PLA-"
190 PRINT "NETS NORDPOL. AFSTANDEN FRA DIN PLANETS"
200 PRINT "CENTRUM VARIERER MELLEM 10000 OG 30000"
210 PRINT "KILOMETER, OG MED DETS NUVÆRENDE"
220 PRINT "HASTIGHED KAN DET CIRKLE RUNDT OM DIN"
230 PRINT "PLANET PÅ MELLEM 12 OG 36 TIMER."
240 PRINT
250 PRINT "DESVÆRRE BRUGER ROMULANERNE ET ELEK-"
260 PRINT "TRONSKJOLD, SÅ DU KAN IKKE SE RUMSKI-"
270 PRINT "BET, MEN VED HJÆLP AF ET SÆRLIGT"
280 PRINT "INSTRUMENT KAN DU FÅ AT VIDE HVOR LANGT"
290 PRINT "VÆK FRA RUMSKIBET DINE FOTON-BOMBER"
300 PRINT "EKSPLODERER."
310 PRINT
320 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
330 INPUT C$
340 PRINT
350 PRINT "DU SKAL NEDSKYDE RUMSKIBET PÅ 7 TIMER,"
360 PRINT "FOR DET TAGER NETOP RUMSKIBET 7 TIMER"
370 PRINT "AT OPSAMLE NOK ENRGI TIL AT SLIPPE UD"
380 PRINT "AF DIN PLANETS TYNGDEFELT."
390 PRINT

```

```

400 PRINT "DIN PLANET HAR ENERGI NOK TIL AT AFFYRE"
410 PRINT "EN FOTON-BOMBE I TIMEN."
420 PRINT
430 PRINT "VED HVER TIMES BEGYNDELSE VIL DU BLIVE"
440 PRINT "BEDT OM AT ANGIVE EN VINKEL (MELLEM 0"
450 PRINT "OG 360 GRADER) OG EN AFSTAND I"
460 PRINT "100 KILOMETERS-ENHEDER (MELLEM 100 OG"
470 PRINT "300)."

```



# Ord

Ord er en blanding af Hængt mand og Bagels. I dette spil skal du gætte et ord med udgangspunkt i de bogstavpositioner, som datamaten giver dig. I stedet for at gætte et bogstav ad gangen, kan du i Ord gætte hele ord (eller andre grupper af bogstaver, som f.eks. ABCDE). Datamaten fortæller dig derefter, om de bogstaver, du har gættet, findes i det ukendte ord, og om nogen af dem stod på den rigtige plads. Ved hjælp af disse ledetråde skal du fortsætte med at gætte, indtil du har fundet det rigtige ord. Hvis du ikke kan gætte ordet, kan du indtaste et »?«, og så fortæller datamaten, hvad det var for et ord.

Du kan selv ændre ordene i programlinierne 1070, 1080 og 1090. Men husk, at der skal være tale om ord på fem bogstaver.

Forfatteren af dette program er Charles Reid fra Lexington High School, Lexington, Massachusetts.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
ORD  
-----
```

```
JEG TÆNKER PÅ ET ORD.  
DU SKAL GÆTTE HVILKET ORD DET ER!
```

```
DU FÅR LIDT HJÆLP AF MIG.  
HELD OG LYKKE!!
```

```
NU BEGYNDER ET NYT SPIL!
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?ABCDE
```

```
1 AF BOGSTAVERNE PASSEDE,  
DET VAR BOGSTAVERNE: E  
PÅ RIGTIG PLADS VAR: ----E
```

```
HVIS DU GIVER OP, SÅ TAST '?' SOM SVAR
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?FGHIJ
```

```
2 AF BOGSTAVERNE PASSEDE,  
DET VAR BOGSTAVERNE: JG  
PÅ RIGTIG PLADS VAR: ----E
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?JODLM
```

```
1 AF BOGSTAVERNE PASSEDE,  
DET VAR BOGSTAVERNE: J  
PÅ RIGTIG PLADS VAR: J----E
```

```
HVIS DU GIVER OP, SÅ TAST '?' SOM SVAR
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?JIGGU
```

```
4 AF BOGSTAVERNE PASSEDE,  
DET VAR BOGSTAVERNE: JUGG  
PÅ RIGTIG PLADS VAR: J--GE
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?JØNSE
```

```
3 AF BOGSTAVERNE PASSEDE,  
DET VAR BOGSTAVERNE: JNE  
PÅ RIGTIG PLADS VAR: J-NGE
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ??
```

```
DET HEMMELIGE ORD ER: JUNGE
```

```
NU BEGYNDER ET NYT SPIL!
```

```
GÆT ET ORD PÅ FEM BOGSTAVER ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "ORD"  
130 PRINT "-----"  
140 PRINT  
150 DIM S(7),A(7),L(7),D(7),P(7)  
160 PRINT "JEG TÆNKER PÅ ET ORD."  
170 PRINT "DU SKAL GÆTTE HVILKET ORD DET ER!"  
180 PRINT  
190 PRINT "DU FÅR LIDT HJÆLP AF MIG."  
200 PRINT "HELD OG LYKKE!!"  
210 PRINT
```

```
220 PRINT "NU BEGYNDER ET NYT SPIL!"  
230 PRINT  
240 RESTORE  
250 READ N  
260 C = INT ( RND (1) * N + 1)  
270 FOR I = 1 TO C  
280 READ S$  
290 NEXT I  
300 G = 0  
310 S(O) = LEN (S$)  
320 FOR I = 1 TO LEN (S$)  
330 S(I) = ASC ( MID$ (S$,I,1))  
340 NEXT I  
350 FOR I = 1 TO 5  
360 A(I) = 45  
370 NEXT I  
380 FOR J = 1 TO 5  
390 P(J) = 0  
400 NEXT J  
410 PRINT "GÆT ET ORD PÅ FEM BOGSTAVER ";  
420 INPUT L$  
430 PRINT  
440 G = G + 1  
450 IF S$ = G$ THEN 990  
460 FOR I = 1 TO 7  
470 P(I) = 0  
480 NEXT I  
490 L(O) = LEN (L$)  
500 FOR I = 1 TO LEN (L$)  
510 L(I) = ASC ( MID$ (L$,I,1))  
520 NEXT I  
530 IF L(1) = 63 THEN 860  
540 IF L(O) < > 5 THEN 940  
550 M = 0  
560 Q = 1  
570 FOR I = 1 TO 5  
580 FOR J = 1 TO 5  
590 IF S(I) < > L(J) THEN 650  
600 P(Q) = L(J)  
610 Q = Q + 1  
620 IF I < > J THEN 640  
630 A(J) = L(J)  
640 M = M + 1  
650 NEXT J  
660 NEXT I  
670 A(O) = 5  
680 P(O) = M  
690 A$ = ""  
700 FOR I = 1 TO A(O)  
710 A$ = A$ + CHR$ (A(I))  
720 NEXT I  
730 P$ = ""  
740 FOR I = 1 TO P(O)  
750 P$ = P$ + CHR$ (P(I))  
760 NEXT I  
770 PRINT M;"AF BOGSTAVERNE PASSEDE,"  
780 PRINT "DET VAR BOGSTAVERNE: ";P$  
790 PRINT "PÅ RIGTIG PLADS VAR: ";A$  
800 IF A$ = S$ THEN 990  
810 IF M > 1 THEN 840  
820 PRINT  
830 PRINT "HVIS DU GIVER OP, SÅ TAST '?' SOM SVAR"  
840 PRINT  
850 GOTO 410  
860 S$ = ""  
870 FOR I = 1 TO 7  
880 S$ = S$ + CHR$ (S(I))  
890 NEXT I  
900 PRINT  
910 PRINT "DET HEMMELIGE ORD ER: ";S$  
920 PRINT  
930 GOTO 210  
940 PRINT "DET ER ET ORD PÅ 5 BOGSTAVER DU SKAL"  
950 PRINT "GÆTTE!! PRØV IGEN."  
960 PRINT  
970 G = G - 1  
980 GOTO 410  
990 PRINT  
1000 PRINT "DU HAR GÆTTET ORDET!!!!"  
1010 PRINT  
1020 PRINT "DU VAR";G;"FORSØG OM DET!"  
1030 PRINT  
1040 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";  
1050 INPUT Q$  
1060 IF LEFT$ (Q$,1) = "J" THEN 210  
1070 DATA 12,"SPURV","TRANE","KARPE","BILLE"  
1080 DATA "PRÆST","HØSTE","BEIGE","GRYDE"  
1090 DATA "JUNGE","KEDEL","FÆRGE","FORAN"  
1100 END
```

# Pelshandler

Du er leder af en fransk pelshandlerekspedition, som i år 1776 forlader Ontario-området for at sælge skind og få proviant til næste års ekspeditioner. Du kan vælge mellem tre forskellige forter, hvor du kan handle. Priserne for de varer, du har brug for, og det beløb du kan få for dine skind, er afhængige af, hvilket fort du vælger at rejse til. Du skal også angive, hvilke skind du har at sælge.

Programmet fortsætter, indtil du vælger ikke at købe og sælge mere.

Forfatteren er Dan Bachor, University of Calgary, Alberta, Canada.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
PELSHANDLER  
-----

DU ER LEDER AF EN FRANSK PELSHANDLER-EKSPEDITION I 1776. EKSPEDITIONEN STARTER VED LAKE ONTARIO OG SKAL SÆLGE PELSE OG SKAFFE FORSYNINGER TIL NÆSTE ÅR. DU KAN VÆLGE MELLEM 3 FORTER, HVOR DU KAN HANDLE. HVAD DINE FORSYNINGER KOMMER TIL AT KOSTE, OG HVOR MEGET DU FÅR FOR DINE SKIND AFHÆNGER AF, HVILKET FORT DU VÆLGER.

ØNSKER DU AT HANDLE MED SKIND  
SVAR 'J' FOR JA OG 'N' FOR NEJ ?J

DU HAR OPSPARET 600 DOLLARS OG  
190 SKIND, DER DANNER GRUNDLAGET  
FOR DIN EKSPEDITION.

DINE 190 SKIND KAN VÆRE AF FØLGENDE  
SLAGS: MINK, BÆVER, HERMELIN OG RÆV.

HVOR MANGE MINKSKIND HAR DU ?40

HVOR MANGE BÆVERSKIND HAR DU ?50

HVOR MANGE HERMELINSKIND HAR DU ?60

HVOR MANGE RÆVESKIND HAR DU ?40

VIL DU HANDLE I FORT 1, 2 ELLER 3

FORT 1 ER FORT HOCHELAGA (MONTREAL)  
OG ER UNDER DEN FRANSKE HÆRS  
BESKYTTELSE.

FORT 2 FORT STADACONA (QUEBEC) ER OGSÅ  
UNDER DEN FRANSKE HÆRS  
BESKYTTELSE, MEN DU SKAL BÅRE  
BÅRDENE OVER LAND, FORBI LACHINE  
VANDFALDENE.

FORT 3 ER FORT NEW YORK, OG DET TILHØ-  
RER HOLLÆNDERNE, OG FOR AT NÅ  
DET SKAL DU GENNEM IROQUOIS-  
INDIANERNES LAND.

SVAR '1', '2' ELLER '3' ?3

DU HAR VALGT DEN VANSKELIGSTE RUTE.  
NÅR DU KOMMER TIL FORT NEW YORK VIL DU  
FÅ DEN HØJESTE PRIS FOR DINE SKIND.  
PRISEN DU SKAL BETALE FOR FORSYNINGER  
VIL VÆRE LAVERE END PÅ ALLE ANDRE FOR-  
TER.

ØNSKER DU AT HANDLE I ET ANDET FORT  
SVAR 'J' FOR JA OG 'N' FOR NEJ ?J

SVAR '1', '2' ELLER '3' ?2

DU HAR VALGT EN SVÆR RUTE. DEN ER LET-  
TERE END NEW YORK-RUTEN, MEN HÅRDERE  
END RUTEN TIL HOCHELAGA. DU VIL FÅ EN  
GENNEMSNITLIG VÆRDI FOR DINE SKIND, OG  
OGSÅ FORSYNINGSPRISERNE VIL VÆRE RIME-  
LIGE.

ØNSKER DU AT HANDLE I ET ANDET FORT  
SVAR 'J' FOR JA OG 'N' FOR NEJ ?N

DU ANKOM HELSKINDET TIL FORT STADACONA.  
PÅ FORT STADACONA KOSTER FORSYNINGER  
125.00 DOLLARS. DINE REJSEOMKOSTNINGER  
HAR VÆRET 15.00 DOLLARS.

BÆVERSKIND SOLGT FOR 52.5  
RÆVESKIND SOLGT FOR 0  
HERMELINSKIND SOLGT FOR 55.2  
MINKSKIND SOLGT FOR 43.6

NU HAR DU 611.3 DOLLARS.  
(DINE SPAREPENGE MEDREGNET)

VIL DU OGSÅ HANDLE MED SKIND NÆSTE ÅR  
SVAR 'J' FOR JA OG 'N' FOR NEJ ?N

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                PELSHANDLER"  
130 PRINT "-----"  
140 PRINT  
150 DIM F(4)  
160 GOSUB 540  
170 I = 600  
180 PRINT  
190 PRINT "ØNSKER DU AT HANDLE MED SKIND "  
200 GOSUB 1950  
210 IF LEFT$(B$,1) = "J" THEN 230  
220 GOTO 2140  
230 PRINT  
240 PRINT "DU HAR OPSPARET";I;"DOLLARS OG"  
250 PRINT "190 SKIND, DER DANNER GRUNDLAGET"  
260 PRINT "FOR DIN EKSPEDITION."  
270 E1 = INT ((.15 * RND (1) + .95) * 10 ^ 2 + .5)  
/ 10 ^ 2  
280 B1 = INT ((.25 * RND (1) + 1.0) * 10 ^ 2 + .5)  
/ 10 ^ 2  
290 PRINT  
300 PRINT "DINE 190 SKIND KAN VÆRE AF FØLGENDE"  
310 PRINT "SLAGS: MINK, BÆVER, HERMELIN OG RÆV."  
320 GOSUB 2090  
330 RESTORE  
340 FOR J = 1 TO 4  
350 READ B$  
360 PRINT  
370 PRINT "HVOR MANGE ";B$;"SKIND HAR DU ";  
380 INPUT F(J)  
390 F(0) = F(1) + F(2) + F(3) + F(4)  
400 IF F(0) = 190 THEN 430  
410 IF F(0) > 190 THEN 450  
420 NEXT J  
430 PRINT  
440 GOTO 640  
450 PRINT  
460 PRINT "DU KAN IKKE HAVE SÅ MANGE SKIND.....LAD"  
470 PRINT "LAD VÆRE MED AT SNYDE."  
480 PRINT "JEG KAN GODT REGNE! START FORFRA."  
490 PRINT  
500 GOTO 160  
510 PRINT  
520 PRINT "VIL DU OGSÅ HANDLE MED SKIND NÆSTE ÅR "  
530 GOTO 200  
540 PRINT "DU ER LEDER AF EN FRANSK PELSHANDLER-"  
550 PRINT "EKSPEDITION I 1776. EKSPEDITIONEN"  
560 PRINT "STARTER VED LAKE ONTARIO OG SKAL SÆLGE"  
570 PRINT "PELSE OG SKAFFE FORSYNINGER TIL NÆSTE"  
580 PRINT "ÅR. DU KAN VÆLGE MELLEM 3 FORTER, HVOR"  
590 PRINT "DU KAN HANDLE. HVAD DINE FORSYNINGER"  
600 PRINT "KOMMER TIL AT KOSTE, OG HVOR MEGET DU"  
610 PRINT "FÅR FOR DINE SKIND AFHÆNGER AF, HVILKET"  
620 PRINT "FORT DU VÆLGER."  
630 RETURN  
640 PRINT "VIL DU HANDLE I FORT 1, 2 ELLER 3 "  
650 PRINT  
660 PRINT "FORT 1 ER FORT HOCHELAGA (MONTREAL)"  
670 PRINT "                OG ER UNDER DEN FRANSKE HÆRS"  
680 PRINT "                BESKYTTELSE."  
690 PRINT  
700 PRINT "FORT 2 FORT STADACONA (QUEBEC) ER OGSÅ"  
710 PRINT "                UNDER DEN FRANSKE HÆRS"  
720 PRINT "                BESKYTTELSE, MEN DU SKAL BÅRE"  
730 PRINT "                BÅRDENE OVER LAND, FORBI LACHINE"  
740 PRINT "                VANDFALDENE."  
750 PRINT  
760 PRINT "FORT 3 ER FORT NEW YORK, OG DET TILHØ-"  
770 PRINT "                RER HOLLÆNDERNE, OG FOR AT NÅ"  
780 PRINT "                DET SKAL DU GENNEM IROQUOIS-"  
790 PRINT "                INDIANERNES LAND."  
800 PRINT  
810 PRINT "SVAR '1', '2' ELLER '3' ";  
820 INPUT B  
830 PRINT  
840 IF B = 1 THEN 880  
850 IF B = 2 THEN 960  
860 IF B = 3 THEN 1050  
870 GOTO 810  
880 PRINT "DU HAR VALGT DEN LETTESTE RUTE, MEN"  
890 PRINT "FORTET LIGGER LANGT FRA EN HAVN. DINE"  
900 PRINT "SKIND VIL KUN INDBRINGE RINGE VÆRDI, OG"  
910 PRINT "FORSYNINGSPRISERNE ER HØJERE END I FORT"  
920 PRINT "STADACONA OG FORT NEW YORK."
```

```

930 GOSUB 1930
940 IF LEFT$(B$,1) = "J" THEN 810
950 GOTO 1140
960 PRINT "DU HAR VALGT EN SVÆR RUTE. DEN ER LET-"
970 PRINT "TERE END NEW YORK-RUTEN, MEN HÅRDERE"
980 PRINT "END RUTEN TIL HOCHELAGA. DU VIL FRÆ EN"
990 PRINT "GENNEMSNITLIG VÆRDI FOR DINE SKIND, OG"
1000 PRINT "OGSÅ FORSYNINGSPRISERNE VIL VÆRE RIME-"
1010 PRINT "LIGE."
1020 GOSUB 1930
1030 IF LEFT$(B$,1) = "J" THEN 810
1040 GOTO 1250
1050 PRINT "DU HAR VALGT DEN VANSKELIGSTE RUTE."
1060 PRINT "NÅR DU KOMMER TIL FORT NEW YORK VIL DU"
1070 PRINT "FRÆ DEN HØJESTE PRIS FOR DINE SKIND."
1080 PRINT "PRISEN DU SKAL BETALE FOR FORSYNINGER"
1090 PRINT "VIL VÆRE LAVERE END PÅ ALLE ANDRE FOR-"
1100 PRINT "TER."
1110 GOSUB 1930
1120 IF LEFT$(B$,1) = "J" THEN 810
1130 GOTO 1580
1140 I = I - 160
1150 PRINT
1160 M1 = INT ((.2 * RND (1) + .70) * 10 ^ 2 + .5)
/ 10 ^ 2
1170 E1 = INT ((.2 * RND (1) + .65) * 10 ^ 2 + .5)
/ 10 ^ 2
1180 B1 = INT ((.2 * RND (1) + .75) * 10 ^ 2 + .5)
/ 10 ^ 2
1190 D1 = INT ((.2 * RND (1) + .80) * 10 ^ 2 + .5)
/ 10 ^ 2
1200 PRINT "I FORT HOCHELAGA KOSTER FORSYNINGER"
1210 PRINT "150.00 DOLLARS."
1220 PRINT "DINE REJSEOMKOSTNINGER TIL HOCHELAGA"
1230 PRINT "BELØBER SIG TIL 10.00 DOLLARS."
1240 GOTO 1990
1250 I = I - 140
1260 PRINT
1270 M1 = INT ((.30 * RND (1) + .85) * 10 ^ 2 + .5)
/ 10 ^ 2
1280 E1 = INT ((.15 * RND (1) + .80) * 10 ^ 2 + .5)
/ 10 ^ 2
1290 B1 = INT ((.20 * RND (1) + .90) * 10 ^ 2 + .5)
/ 10 ^ 2
1300 P = INT (10 * RND (1)) + 1
1310 IF P < = 2 THEN 1350
1320 IF P < = 6 THEN 1420
1330 IF P < = 8 THEN 1440
1340 IF P < = 10 THEN 1490
1350 F(2) = 0
1360 PRINT "DINE BÆVERSKIND VAR FOR TUNGE TIL AT"
1370 PRINT "BÆRE OVER VANDFALDEN SAMMEN MED BÅDEN."
1380 PRINT "DA DU VENDER TILBAGE EFTER DEM ER DE"
1390 PRINT "STJÅLET."
1400 GOSUB 1540
1410 GOTO 2010
1420 PRINT "DU ANKOM HELSKINDET TIL FORT STADACONA."
1430 GOTO 1520
1440 GOSUB 2090
1450 PRINT "DIN KANO KÆNTREDE I LACHINE VANDFALDEN"
1460 PRINT "DU HAR MISTET ALLE DINE SKIND."
1470 GOSUB 1540
1480 GOTO 2040
1490 F(4) = 0
1500 PRINT "DINE RÆVE-SKIND ER IKKE GARVET RIGTIGT."
1510 PRINT "INGEN VIL KØBE DEM."
1520 GOSUB 1540
1530 GOTO 1990
1540 PRINT "PÅ FORT STADACONA KOSTER FORSYNINGER"
1550 PRINT "125.00 DOLLARS. DINE REJSEOMKOSTNINGER"
1560 PRINT "HAR VÆRET 15.00 DOLLARS."
1570 RETURN
1580 I = I - 105
1590 PRINT
1600 M1 = INT ((.15 * RND (1) + 1.05) * 10 ^ 2 +
.5) / 10 ^ 2
1610 D1 = INT ((.25 * RND (1) + 1.10) * 10 ^ 2 +
.5) / 10 ^ 2
1620 P = INT (10 * RND (1)) + 1
1630 IF P < = 2 THEN 1670
1640 IF P < = 6 THEN 1710
1650 IF P < = 8 THEN 1740
1660 IF P < = 10 THEN 1810
1670 PRINT "DU BLEV ANGREBET AF EN GRUPPE IROQUIS-"
1680 PRINT "INDIANERE. ALLE DELTAGERNE I DIN PELS-"
1690 PRINT "EKSPEDITION BLEV DRÆBT. SPILLET ER UDE!"
1700 GOTO 2140
1710 PRINT "DU VAR HELDIG. DU ANKOM SIKKERT TIL"
1720 PRINT "FORT NEW YORK."
1730 GOTO 1870
1740 GOSUB 2090
1750 PRINT "DU UNDSLAP MED NØD OG NÆPPE EN FJENDT-"
1760 PRINT "SINDET GRUPPE IROQUOIS-INDIANERNE, MEN"
1770 PRINT "DU VAR NØD TIL AT LADE DINE SKIND LIG-"
1780 PRINT "GE FOR AT REDDE LIVET."
1790 GOSUB 1890
1800 GOTO 2040
1810 B1 = B1 / 2
1820 M1 = M1 / 2
1830 PRINT "DINE MINK- OG BÆVERSKIND BLEV ØDELAGT"
1840 PRINT "UNDER TRANSPORTEN. DU FIK KUN HALVDELEN"
1850 PRINT "AF DEN PRIS DU HAVDE REGNET MED FOR"
1860 PRINT "DISSE SKIND."
1870 GOSUB 1890
1880 GOTO 1990
1890 PRINT "I NEW YORK KOSTER FORSYNINGERNE 80.00"
1900 PRINT "DOLLARS. DINE REJSEUDGIFTER VAR PÅ"
1910 PRINT "25.00 DOLLARS."
1920 RETURN
1930 PRINT
1940 PRINT "ØNSKER DU AT HANDLE I ET ANDET FORT "
1950 PRINT "SVAR 'J' FOR JA OG 'N' FOR NEJ ";
1960 INPUT B$
1970 PRINT
1980 RETURN
1990 PRINT
2000 PRINT "BÆVERSKIND SOLGT FOR";B1 * F(2)
2010 PRINT "RÆVESKIND SOLGT FOR";D1 * F(4)
2020 PRINT "HERMELINSKIND SOLGT FOR";E1 * F(3)
2030 PRINT "MINKSKIND SOLGT FOR";M1 * F(1)
2040 I = M1 * F(1) + B1 * F(2) + E1 * F(3) + D1 *
F(4) + I
2050 PRINT
2060 PRINT "NU HAR DU";I;"DOLLARS."
2070 PRINT "(DINE SPAREPENGE MEDREGNET)"
2080 GOTO 510
2090 FOR J = 1 TO 4
2100 F(J) = 0
2110 NEXT J
2120 RETURN
2130 DATA "MINK","BÆVER","HERMELIN","RÆVE"
2140 END

```

# Pilespil

I dette spil kaster spillerne pile mod en skydeskive med ringe, der har 10-, 20-, 30- og 40 points. Op til 20 spillere kan deltage. Det gælder om at få 200 points først.

Du kan vælge mellem tre forskellige måder at kaste på:

- 1 = Hurtig overarm (rammer enten i centrum eller helt ved siden af)
- 2 = Kontrolleret overarm (giver 10, 20 eller 30 point)
- 3 = Underarm (kan give hvad som helst)

Efter at have spillet et stykke tid vil du opdage, at forskellige spillere sværger til bestemte metoder. Du kan imidlertid selv bedømme det forventede resultat pr. kast, når kun kast nummer 3 anvendes (se også programlinierne 590 og 600)

POINT (S)	SANDSYNLIGHED (P)	S*P
40	$1.00 - 0.95 = 0.05$	2
30	$0.95 - 0.75 = 0.20$	6
20	$0.75 - 0.45 = 0.30$	6
10	$0.45 - 0.05 = 0.40$	4
0	$0.05 - 0.00 = 0.05$	0
Forventet antal point pr. kast =		18

Prøv at beregne de forventede points for de andre former for kast, og du vil blive svært forbavset!

Programmet er skrevet af David Ahl fra Creative Computing.

## RUNDE 5

PETER'S KAST ?2  
30 POINT!!!!  
TOTALE ANTAL POINT = 90

KELDE'S KAST ?2  
30 POINT!!!!  
TOTALE ANTAL POINT = 130

## RUNDE 6

PETER'S KAST ?2  
20 POINT!!  
TOTALE ANTAL POINT = 110

KELDE'S KAST ?1  
30 POINT!!!!  
TOTALE ANTAL POINT = 160

## RUNDE 7

PETER'S KAST ?1  
LIGE I PLET! 40 POINT!  
TOTALE ANTAL POINT = 150

KELDE'S KAST ?1  
LIGE I PLET! 40 POINT!  
TOTALE ANTAL POINT = 200

VI HAR EN VINDER!!

KELDE FIK 200 POINTS.

TAK FOR SPILLET.

## MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

### PILESPIL

I DETTE SPIL KAN OP TIL 20 SPILLERE  
KASTE MED PILE MOD EN SKIVE, DER KAN  
GIVE ENTEN 10, 20, 30 ELLER 40 POINT.

DET GÆLDER OM AT FÅ 200 POINT.

DU KAN VÆLGE MELLEM TRE FORSKELLIGE  
FORMER FOR KAST:

- 1=HURTIG OVERARM (CENTRUM ELLER HELT  
VED SIDEN AF)
- 2=KONTROLLERET OVERARM (10, 20 ELLER  
30 POINT)
- 3=UNDERARM (HVAD SOM HELST)

HVOR MANGE VIL SPILLE ?2

HVAD HEDDER SPILLER NR. 1 ?PETER  
HVAD HEDDER SPILLER NR. 2 ?KELDE

### RUNDE 1

PETER'S KAST ?1  
LIGE I PLET! 40 POINT!  
TOTALE ANTAL POINT = 40

KELDE'S KAST ?2  
30 POINT!!!!  
TOTALE ANTAL POINT = 30

### RUNDE 2

PETER'S KAST ?0  
DU RAMTE VED SIDEN AF SKIVEN!!  
TOTALE ANTAL POINT = 40

KELDE'S KAST ?2  
30 POINT!!!!  
TOTALE ANTAL POINT = 60

### RUNDE 3

PETER'S KAST ?1  
DU RAMTE VED SIDEN AF SKIVEN!!  
TOTALE ANTAL POINT = 40

KELDE'S KAST ?1  
30 POINT!!!!  
TOTALE ANTAL POINT = 90

### RUNDE 4


PETER'S KAST ?2  
20 POINT!!  
TOTALE ANTAL POINT = 60

KELDE'S KAST ?2  
SØRENS OSSE, 10 POINT  
TOTALE ANTAL POINT = 100

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          PILESPIL"
130 PRINT "-----"
140 PRINT
150 PRINT "I DETTE SPIL KAN OP TIL 20 SPILLERE"
160 PRINT "KASTE MED PILE MOD EN SKIVE, DER KAN"
170 PRINT "GIVE ENTEN 10, 20, 30 ELLER 40 POINT."
180 PRINT
190 PRINT "DET GÆLDER OM AT FR 200 POINT."
200 PRINT
210 PRINT "DU KAN VÆLGE MELLEM TRE FORSKELLIGE"
220 PRINT "FORMER FOR KAST:"
230 PRINT
240 PRINT "1=HURTIG OVERARM (CENTRUM ELLER HELT"
250 PRINT "          VED SIDEN AF)"
260 PRINT "2=KONTROLLERET OVERARM (10, 20 ELLER"
270 PRINT "          30 POINT)"
280 PRINT "3=UNDERARM          (HVAD SOM HELST)"
290 PRINT
300 DIM A$(20),S(20),W(10)
310 M = 0:R = 0
320 FOR I = 1 TO 20
330 S(I) = 0
340 NEXT I
350 PRINT "HVOR MANGE VIL SPILLE ";
360 INPUT N
370 PRINT
380 FOR I = 1 TO N
390 PRINT "HVAD HEDDER SPILLER NR. ";I;
400 INPUT A$(I)
410 NEXT I
420 R = R + 1
430 PRINT
440 PRINT
450 PRINT "RUNDE";R
460 PRINT "-----"
470 FOR I = 1 TO N
480 PRINT
490 PRINT A$(I)"'S KAST ";
500 INPUT T
510 IF T < 0 OR T > 3 THEN PRINT "TAST 1, 2 ELLER
3!": GOTO 490
520 ON T GOTO 530,560,530
530 P1 = .65:P2 = .55
540 P3 = .50:P4 = .50
550 GOTO 610
560 P1 = .99:P2 = .77
570 P3 = .43:P4 = .01
580 GOTO 610
590 P1 = .95:P2 = .75
600 P3 = .45:P4 = .05
610 U = RND (1)
620 IF U > = P1 THEN PRINT "LIGE I PLET! 40
POINT!":B = 40: GOTO 680
630 IF U > = P2 THEN PRINT "30 POINT!!!!":B = 30:
GOTO 680
640 IF U > = P3 THEN PRINT "20 POINT!!!!":B = 20:
GOTO 680
650 IF U > = P4 THEN PRINT "SØRENS OSSE, 10
POINT":B = 10: GOTO 680
660 PRINT "DU RAMTE VED SIDEN AF SKIVEN!!"
670 B = 0
680 S(I) = S(I) + B
690 PRINT "TOTALE ANTAL POINT =" ;S(I)
700 NEXT I
710 FOR I = 1 TO N
720 IF S(I) > = 200 THEN M = M + 1:W(M) = I
730 NEXT I
740 IF M = 0 THEN 420
750 PRINT
760 PRINT "VI HAR EN VINDER!!"
770 PRINT
780 FOR I = 1 TO M
790 PRINT A$(W(I))" FIK";S(W(I));"POINTS."
800 NEXT I
810 PRINT
820 PRINT "TAK FOR SPILLET."
830 END

```



# Pizza

I dette spil gælder det om at levere pizzaer til kunder, der bor i Ædeby. Ved hjælp af et kort over byen skal du fortælle bydrengen, hvor han skal køre hen med pizzaen, dvs. give ham den rigtige adresse. Hvis pizzaen er leveret til den rigtige kunde, ringer denne dig op og takker for pizzaen. Hvis den ikke når frem, er du nødt til at give bydrengen en ny adresse, indtil pizzaen når frem til den rigtige kunde.

Der kan foretages nogle interessante ændringer i programmet, f.eks. at pizzaerne bliver kolde efter at være afleveret til to forkerte adresser, eller man kan tage imod to eller tre bestillinger samtidig og derefter regne den korteste leveringsvej ud.

Dette program synes oprindeligt at være blevet skrevet på University of Georgia i Athens, Georgia. Forfatteren er ukendt.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

PIZZA

PIZZABESTILLINGSSPIL

HVAD HEDDER DU TIL FORNAVN ?CHRISTIAN

DAVS, CHRISTIAN. I DETTE SPIL TAGER DU IMOD BESTILLINGER PÅ PIZZAER. NÅR ALLE BESTILLINGER ER INDLØBET SKAL DU FORTÆLLE BYDRENGEN HVOR PIZZAERNE SKAL LEVERES.

KORT OVER ÆDEBY:

```
-----1-----2-----3-----4-----
-
-
-
-
4   M   N   O   P   4
-
-
-
3   I   J   K   L   3
-
-
-
2   E   F   G   H   2
-
-
-
1   A   B   C   D   1
-
-
-
-----1-----2-----3-----4-----
```

OVENSTÅENDE ER ET KORT OVER DE HUSE HVOR DU SKAL LEVERE PIZZAER.

DET ER DIN OPGAVE AT FORTÆLLE BYDRENGEN HVILKE STEDER, DVS. HVILKE KOORDINATER DE HUSE HAR, HVOR DER ER BESTILT PIZZAER.

HAR DU BRUG FOR MERE INSTRUKTION ?JA

EN ELLER ANDEN VIL BESTILLE EN PIZZA. SÅ VIL BUDDET SPØRGE EFTER DET STED, HVOR DEN SKAL LEVERES.

EKSEMPEL:

GODDAG, DET ER J - JEG VIL GERNE

BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR J?

DIT SVAR SKAL VÆRE '2,3'.

HAR DU FORSTÅET DET ?J

FINT! SÅ ER DU PARAT TIL AT TAGE IMOD BESTILLINGER.

HELD OG LYKKE!!

GODDAG,

ER DET CHRISTIAN'S PIZZERIA?

DU TALER MED C.

JEG VIL GERNE BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR C ?3,1

GODDAG CHRISTIAN DET ER C HER.

TAK FOR PIZZAEN, DEN SMAGTE GODT.

GODDAG,

ER DET CHRISTIAN'S PIZZERIA?

DU TALER MED A.

JEG VIL GERNE BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR A ?1,4

MIN NAVN ER M.

JEG HAR IKKE BESTILT PIZZA.

PIZZA. JEG BOR PÅ 1, 4

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR A ?1,1

GODDAG CHRISTIAN DET ER A HER.

TAK FOR PIZZAEN, DEN SMAGTE GODT.

GODDAG,

ER DET CHRISTIAN'S PIZZERIA?

DU TALER MED I.

JEG VIL GERNE BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR I ?1,3

GODDAG CHRISTIAN DET ER I HER.

TAK FOR PIZZAEN, DEN SMAGTE GODT.

GODDAG,

ER DET CHRISTIAN'S PIZZERIA?

DU TALER MED B.

JEG VIL GERNE BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR B ?2,1

GODDAG CHRISTIAN DET ER B HER.

TAK FOR PIZZAEN, DEN SMAGTE GODT.

GODDAG,

ER DET CHRISTIAN'S PIZZERIA?

DU TALER MED N.

JEG VIL GERNE BESTILLE EN PIZZA.

BUDDET SPØRGER CHRISTIAN:

SIG MIG, HVOR BOR N ?2,4

GODDAG CHRISTIAN DET ER N HER.

TAK FOR PIZZAEN, DEN SMAGTE GODT.

VIL DU LEVERE FLERE PIZZAER ?NEJ

DET ER I ORDEN CHRISTIAN VI SES!.....

```


100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                PIZZA"
130 PRINT "-----"
140 PRINT
150 DIM S$(16),M$(4)
160 PRINT "PIZZABESTILLINGSSPIL"
170 PRINT
180 PRINT "HVAD HEDDER DU TIL FORNAVN ";
190 INPUT N$
200 PRINT
210 PRINT "DAVS, ";N$. I DETTE"
220 PRINT "SPIL TAGER DU IMOD BESTILLINGER PÅ"
230 PRINT "PIZZAER. NÅR ALLE BESTILLINGER ER"
240 PRINT "INDLØBET SKAL DU FORTÆLLE BYDRENGEN"
250 PRINT "HVOR PIZZAERNE SKAL LEVERES."
260 PRINT
270 PRINT
280 FOR I = 1 TO 16
290 READ S$(I)
300 NEXT I
310 FOR I = 1 TO 4
320 READ M$(I)
330 NEXT I
340 DATA "A","B","C","D","E","F","G","H","I"
350 DATA "J","K","L","M","N","O","P"
360 DATA "1","2","3","4"
370 PRINT "KORT OVER RØDEBY:"
380 PRINT
390 PRINT "  ----1----2----3----4----"
400 K = 4
410 FOR I = 1 TO 4
420 PRINT "--"
430 PRINT "--"
440 PRINT "--"
450 PRINT "--"
460 PRINT M$(K);
470 S1 = 16 - 4 * I + 1
480 PRINT "    ";S$(S1);"    ";S$(S1 + 1);
490 PRINT "    ";S$(S1 + 2);"    ";S$(S1 + 3);
500 PRINT "    ";M$(K)
510 K = K - 1
520 NEXT I
530 PRINT "--"
540 PRINT "--"
550 PRINT "--"
560 PRINT "--"
570 PRINT "  ----1----2----3----4----"
580 PRINT
590 PRINT
600 PRINT "OVENSTRÆNDE ER ET KORT OVER DE HUSE"
610 PRINT "HVOR DU SKAL LEVERE PIZZAER."
620 PRINT
630 PRINT "DET ER DIN OPGAVER AT FORTÆLLE BY-"
640 PRINT "DRENGEN HVILKE STEDER, DVS. HVILKE"
650 PRINT "KOORDINATER DE HUSE HAR, HVOR DER ER"
660 PRINT "BESTILT PIZZAER."
670 PRINT
680 PRINT "HAR DU BRUG FOR MERE INSTRUKTION ";

```

```

690 INPUT A$
700 IF LEFT$(A$,1) = "J" THEN 740
710 IF LEFT$(A$,1) = "N" THEN 990
720 PRINT "SKRIV 'JA' ELLER 'NEJ' - IGEN..."
730 GOTO 670
740 PRINT
750 PRINT "EN ELLER ANDEN VIL BESTILLE EN PIZZA."
760 PRINT "SR VIL BUDDET SPØRGE EFTER DET STED,"
770 PRINT "HVOR DEN SKAL LEVERES."
780 PRINT
790 PRINT "EKSEMPEL:"
800 PRINT "GODDAG, DET ER J - JEG VIL GERNE"
810 PRINT "BESTILLE EN PIZZA."
820 PRINT "BUDDET SPØRGER ";N$;": "
830 PRINT "SIG MIG, HVOR BOR J?"
840 PRINT "DIT SVAR SKAL VÆRE '2,3'."
850 PRINT
860 PRINT "HAR DU FORSTRET DET ";
870 INPUT A$
880 PRINT
890 IF LEFT$(A$,1) = "J" THEN 940
900 PRINT
910 PRINT "NÅR IKKE, SR ER DETTE SPIL ALT FOR"
920 PRINT "VANSKELIGT FOR DIG. FARVEL!"
930 GOTO 1270
940 PRINT "FINT! SR ER DU PARAT TIL AT TAGE IMOD"
950 PRINT "BESTILLINGER."
960 PRINT
970 PRINT "HELD OG LYKKE!!"
980 PRINT
990 FOR I = 1 TO 5
1000 S = INT ( RND (1) * 16 + 1)
1010 PRINT
1020 PRINT "GODDAG,"
1030 PRINT "ER DET ";N$;"S PIZZERIA?"
1040 PRINT "DU TALER MED ";S$(S);"."
1050 PRINT "JEG VIL GERNE BESTILLE EN PIZZA."
1060 PRINT
1070 PRINT "BUDDET SPØRGER ";N$;": "
1080 PRINT "SIG MIG, HVOR BOR ";S$(S);" ";
1090 INPUT A(1),A(2)
1100 PRINT
1110 T = A(1) + (A(2) - 1) * 4
1120 IF T = S THEN 1180
1130 PRINT "MIN NAVN ER ";S$(T);"."
1140 PRINT "JEG HAR IKKE BESTILT PIZZA."
1150 PRINT "PIZZA. JEG BOR PÅ ";A(1);",";A(2)
1160 PRINT
1170 GOTO 1070
1180 PRINT "GODDAG ";N$;" DET ER ";S$(S);" HER."
1190 PRINT "TAK FOR PIZZAEN, DEN SMAGTE GODT."
1200 NEXT I
1210 PRINT
1220 PRINT "VIL DU LEVERE FLERE PIZZAER ";
1230 INPUT A$
1240 IF LEFT$(A$,1) = "J" THEN 990
1250 PRINT
1260 PRINT "DET ER I ORDEN ";N$;" VI SES!...."
1270 END

```



# Plot i 3-D

Dette program plottes en hel familie af kurver for enhver funktion. Funktionen Z plottes, som om den kommer op af x-y planet, med x og y inden for en cirkel med radius 30. Den deraf resulterende udskrift ser næsten ud til at være tredimensionel.

Funktionen, som skal tegnes, indføres i programlinie 190. Ligesom det er tilfældet med alle matematiske funktioner, er nogle af dem »kønnere« end andre. Her er nogle eksempler på funktioner, der tager sig godt ud:

190 DEF FNA(Z) = 30\*EXP((÷ 1)\*Z\*Z/100)

190 DEF FNA(Z) = SQR(900.01 - Z\*Z)\*0.9 ÷ 2

190 DEF FNA(Z) = 30\*(COS(Z/16) ÷ 2

190 DEF FNA(Z) = 30 - 30\*SIN(Z/18)

190 DEF FNA(Z) = 30\*EXP(-COS(Z/16)) ÷ 30


(Bessel funktionen - Summerfelds integral)

190 DEF FNA(Z) = 30\*SIN(Z/10)

Forfatteren til dette lille elegante program er Mark Bramhall fra DEC.


MORRISTOWN CREATIVE COMPUTING NEW JERSEY

PLOT I 3-D



MORRISTOWN CREATIVE COMPUTING NEW JERSEY

PLOT I 3-D



```

0 REM      P093
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                          PLOT I 3-D"
130 PRINT "-----"
140 PRINT
150 PRINT
160 REM
170 REM      Indfør den ønskede formel i nedenstående programlinie
180 REM
190 DEF FN A(Z) = 30 * EXP ( - Z * Z / 100)
200 REM
210 PRINT
220 PRINT
230 FOR X = - 30 TO 30 STEP 1.5
240 L = 0
250 Y1 = 5 * INT ( SQR (900 - X * X) / 5)
260 FOR Y = Y1 TO - Y1 STEP - 5
270 Z = INT (25 + FN A( SQR (X * X + Y * Y)) - .7 * Y)
280 IF Z < = L THEN GOTO 310
290 L = Z
300 PRINT TAB( Z);"*";
310 NEXT Y
320 PRINT
330 NEXT X
340 END
    
```



# Poesi

Dette program udskriver tilfældige poetiske sætninger, der på sin vis kan siges at ligne den japanske Haiku-lyrik. Programmet bruger tyve forskellige sætninger i fire grupper, der hver består af fem sætninger, og programmet arbejder sig generelt gennem disse grupper, i den rækkefølge de optræder i programmet. Programmet placerer tilfældigt komma (i 19% af tilfældene), indrykning af linier (i 22% af tilfældene) og begynder nye afsnit (i 18% af tilfældene, men altid efter hver 20'ende sætning).

De ord, som findes i denne version, minder noget om ordforrådet i Edgar Allan Poes lyrik. Prøv at indføre ord og betegnelser fra edb-verdenen, fra kærlighed og romantik, fra fireårige børns ordforråd eller fra et helt fjerde emneområde.

Her følger nogle naturlyriske fraser, som du kunne forsøge dig med:

Bregnebeklædte vidder

Morgendagen

Gryets kulde

Vigende pinjer

Fortryller mig

Sprøde og smukke

Talende tavshed

Raslende blade

Naturens stemme

Überørt af mennesker

Udstrålende fred

Den oprindelige forfatter til programmet er ukendt. Programmet er modificeret og bearbejdet af Him Bailey, Peggy Ewing og David Ahl fra DEC.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                                POESI
-----
```

```
SORGFULDE MIDNAT FYRIGE ØJNE LOKKER MIG MØRKET DERUNDER
OG DOG IGEN
```

```
FYRIGE ØJNE BRÆNDEDE
SOM VISMANDEN SAGDE ALDRIG MERE
NYFØDT PROFET LOKKER MIG VIL BLIVE LØFTET ,
LANGSOMT KRYBENDE
FYRIGE ØJNE BRÆNDEDE , TEGN PÅ AFSKED , LANGSOMT KRYBENDE
```

```
BARN AF ONDSKAB STADIG SIDDENDE...MØRKET DERUNDER ,
ALDRIG MERE
```

```
HENFØRER MIG NYFØDT PROFET , ALDRIG VANDRENDE TEGN PÅ AFSKED ,
...EVIGHED
```

```
VEN ELLER FJENDE
BRÆNDEDE OG DU MIN SJÆL OG DOG IGEN
SORGFULDE MIDNAT HENFØRER MIG MØRKET DERUNDER LANGSOMT KRYBENDE
VEN ELLER FJENDE
HENFØRER MIG TEGN PÅ AFSKED INTET MERE
NYFØDT PROFET
STADIG SIDDENDE...VIL BLIVE LØFTET , INTET MERE
```

```
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                                POESI"
130 PRINT "-----"
140 PRINT
150 PRINT
160 ON I GOTO 170,180,190,200,210
170 PRINT "SORGFULDE MIDNAT "; GOTO 410
180 PRINT "FYRIGE ØJNE "; GOTO 410
190 PRINT "VEN ELLER FJENDE "; GOTO 410
200 PRINT "BARN AF ONDSKAB "; GOTO 410
210 PRINT "NYFØDT PROFET "; GOTO 410
220 ON I GOTO 230,240,250,260,270
230 PRINT "LOKKER MIG "; GOTO 410
240 PRINT "HENFØRER MIG "; GOTO 410
250 PRINT "STADIG SIDDENDE..."; GOTO 410
260 PRINT "ALDRIG VANDRENDE "; GOTO 410
270 PRINT "BRÆNDEDE "; GOTO 410
280 ON I GOTO 290,300,310,320,330
290 PRINT "OG DU MIN SJÆL "; GOTO 410
300 PRINT "MØRKET DERUNDER "; GOTO 410
310 PRINT "VIL BLIVE LØFTET "; GOTO 410
320 PRINT "SOM VISMANDEN SAGDE "; GOTO 410
330 IF U = 0 THEN 410
340 PRINT "TEGN PÅ AFSKED "; GOTO 410
350 ON I GOTO 360,370,380,390,400
360 PRINT "INTET MERE "; GOTO 410
370 PRINT "OG DOG IGEN "; GOTO 410
380 PRINT "LANGSOMT KRYBENDE "; GOTO 410
390 PRINT "...EVIGHED"; GOTO 410
400 PRINT "ALDRIG MERE ";
410 IF U = 0 OR RND (1) > .19 THEN 430
420 PRINT ", "; U = 2
430 IF RND (1) > .65 THEN 450
440 U = U + 1: GOTO 460
450 PRINT :U = 0
460 I = INT ( INT (10 * RND (1)) / 2) + 1
470 J = J + 1
480 K = K + 1
490 IF U > 0 OR INT ( J / 2) < > J / 2 THEN 510
500 PRINT "      ";
510 ON J GOTO 160,220,280,350,520
520 J = 0
530 PRINT
540 IF K > 20 THEN 560
550 GOTO 460
560 PRINT
570 U = 0
580 K = 0
590 GOTO 220
600 END
```

# Poker

Du spiller mod datamaten i dette spil. Når spillet begynder, får du og datamaten hver 200 kroner. Spillet er slut, når en af spillerne ikke længere har penge tilbage, men hvis det er dig, der ikke har flere penge, så tilbyder datamaten at købe dit ur eller din diamantslipsenål.

Datamaten gør sin indsats, inden den giver kort, og du gør det bagefter. Hvis du får en hånd, som ikke er noget værd og ønsker at lægge dig, så gør en indsats på 0. Hvis du vil se, kan du sætte 0.5 kroner. Når datamaten har gjort en indsats, skal du selvfølgelig gøre en indsats, der er lige så stor, men hvis du mener, at du har en god hånd, kan du selvfølgelig forhøje indsatsen på et hvilket som helst tidspunkt i spillet.

Programmet er skrevet af A. Christopher på Hall of Trinity College, Hartford, Connecticut.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
POKER  
-----

VELKOMMEN TIL KASIONO'ET. VI HAR HVER  
200 KRONER. JEG SATSER ALTID FØR  
NYE KORT GIVES, DU SATSER BAGEFTER.

NÅR DU VIL LÆGGE DIG, SÅ SKAL DU BLOT  
SATSE 0 KRONER. HVIS DU VIL SE, TASTES  
.5 KRONER

OG SÅ ER DER VIST SNAKKET NOK -----

LAD OS SÅ KOMME IGANG.

GEBYRET ER 5 KR. JEG GIVER.

DIN HÅND SER SÅLEDES UD:

-----  
1 -- 5 I RUDER  
2 -- 6 I RUDER  
3 -- 3 I KLØR  
4 -- 7 I KLØR  
5 -- 4 I HJERTER

JEG SER.

HVAD ER DIN INDSATS ?10

JEG SER.

HVOR MANGE KORT VIL DU KØBE ?3

INDTAST NUMRENE PÅ DE KORT DU VIL BYTTE

-----  
BYT KORT NUMMER ?1  
BYT KORT NUMMER ?3  
BYT KORT NUMMER ?5

DIN NYE HÅND SER SÅDAN UD:

-----  
1 -- KONGE I HJERTER  
2 -- 6 I RUDER  
3 -- 7 I HJERTER  
4 -- 7 I KLØR  
5 -- 2 I RUDER

JEG TAGER 3 KORT

HVAD ER DIN INDSATS ?10

JEG LÆGGER KORTENE.

DU VINDER.

JEG HAR NU 185 KR.  
DU HAR 215 KR.

ØNSKER DU AT FORTSÆTTE ?J

GEBYRET ER 5 KR. JEG GIVER.

DIN HÅND SER SÅLEDES UD:

-----  
1 -- 3 I HJERTER  
2 -- 8 I HJERTER  
3 -- 2 I RUDER  
4 -- ES I RUDER  
5 -- 9 I HJERTER

JEG RØBNER MED 29

HVAD ER DIN INDSATS ?30

JEG SER, OG FORHØJER MED 2

HVAD ER DIN INDSATS ?32

JEG SER, OG FORHØJER MED 38

HVAD ER DIN INDSATS ?100

JEG SER.

HVOR MANGE KORT VIL DU KØBE ?3

INDTAST NUMRENE PÅ DE KORT DU VIL BYTTE

-----  
BYT KORT NUMMER ?1  
BYT KORT NUMMER ?3  
BYT KORT NUMMER ?5

DIN NYE HÅND SER SÅDAN UD:

-----  
1 -- KONGE I SPAR  
2 -- 8 I HJERTER  
3 -- 3 I KLØR  
4 -- ES I RUDER  
5 -- 10 I RUDER

JEG TAGER 2 KORT

HVAD ER DIN INDSATS ?10

JEG SER.

JEG SER, OG FORHØJER MED 14

HVAD ER DIN INDSATS ?14

NU SAMMENLIGNER VI VORE TO HÅNDER.

MIN HÅND:

-----  
6 -- 4 I KLØR  
7 -- 4 I SPAR  
8 -- 7 I SPAR  
9 -- DAME I SPAR  
10 -- ES I SPAR

DU HAR SKRALD ES ER HØJESTE KORT.

OG JEG HAR ET PAR I 4

JEG VINDER.

JEG HAR NU 386 KR.  
DU HAR 14 KR.

ØNSKER DU AT FORTSÆTTE ?J

GEBYRET ER 5 KR. JEG GIVER.

DIN HÅND SER SÅLEDES UD:

-----  
1 -- 8 I HJERTER  
2 -- 2 I SPAR  
3 -- ES I SPAR  
4 -- 2 I HJERTER  
5 -- KNÆGT I KLØR

JEG RØBNER MED 32

HVAD ER DIN INDSATS ?32

DU KAN IKKE SATSE DET DU IKKE HAR!

KUNNE DU TÆNKE DIG AT SÆLGE DIT UR ?J

JEG VIL GIVE DIG 75 KRONER FOR DET.

HVAD ER DIN INDSATS ?32

HVOR MANGE KORT VIL DU KØBE ?2

INDTAST NUMRENE PÅ DE KORT DU VIL BYTTE

-----  
BYT KORT NUMMER ?1  
BYT KORT NUMMER ?5

DIN NYE HÅND SER SÅDAN UD:

-----  
1 -- ES I KLØR  
2 -- 2 I SPAR  
3 -- ES I SPAR  
4 -- 2 I HJERTER  
5 -- KONGE I RUDER

JEG TAGER 1 KORT

HVAD ER DIN INDSATS ?.5

JEG SATSER 28

HVAD ER DIN INDSATS ?28

NU SAMMENLIGNER VI VORE TO HÅNDER.

MIN HÅND:

-----  
6 -- 4 I RUDER  
7 -- 5 I KLØR  
8 -- 9 I RUDER  
9 -- 10 I KLØR  
10 -- 10 I RUDER

DU HAR TO PAR I ES

OG JEG HAR ET PAR I 10

DU VINDER.

JEG HAR NU 321 KR.  
DU HAR 154 KR.

ØNSKER DU AT FORTSÆTTE ?J

GEBYRET ER 5 KR. JEG GIVER.

DIN HÅND SER SÅLEDES UD:

-----  
1 -- KONGE I KLØR  
2 -- 7 I HJERTER  
3 -- 3 I RUDER  
4 -- 2 I HJERTER  
5 -- ES I KLØR

JEG RØBNER MED 27

HVAD ER DIN INDSATS ?20

HVIS DU IKKE KAN SVARE PÅ MIN  
INDSATS, MÅ DU LÆGGE KORTENE.

```

HVAD ER DIN INDSATS ?2
HVOR MANGE KORT VIL DU KØBE ?3
INDTAST NUMRENE PÅ DE KORT DU VIL BYTTE
-----
BYT KORT NUMMER ?2
BYT KORT NUMMER ?3
BYT KORT NUMMER ?4

DIN NYE HÅND SER SÅDAN UD:
-----
 1 -- KONGE I KLØR
 2 --      2 I RUDER
 3 --      2 I KLØR
 4 --      ES I HJERTER
 5 --      ES I KLØR

JEG TAGER 2 KORT
HVAD ER DIN INDSATS 100
JEG SER.
NU SAMMENLIGNER VI VORE TO HÅNDER.
MIN HÅND:
-----
 6 --      5 I KLØR
 7 --      6 I RUDER
 8 --      7 I RUDER
 9 -- KNÆGT I SPAR
10 -- KONGE I HJERTER

      DU HAR TO PAR I      ES
OG JEG HAR SKRALD KONGE ER HØJESTE KORT.
DU VINDER.
JEG HAR NU 189 KR.
DU HAR 286 KR.
ØNSKER DU AT FORTSÆTTE ?N

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          POKER"
130 PRINT "-----"
140 PRINT
150 DIM A(50),B(15)
160 DEF FN A(X) = INT (10 * RND (1))
170 DEF FN B(X) = X - 100 * INT (X / 100)
180 PRINT "VELKOMMEN TIL KASIONO'ET. VI HAR HVER"
190 PRINT "200 KRONER. JEG SATSER ALTID FØR"
200 PRINT "NYE KORT GIVES, DU SATSER BAGEFTER."
210 PRINT
220 PRINT "NÅR DU VIL LÆGGE DIG, SÅ SKAL DU BLOT"
230 PRINT "SATSE 0 KRONER. HVIS DU VIL SE, TASTES"
240 PRINT ".5 KRONER"
250 PRINT
260 PRINT "OG SÅ ER DER VIST SNAKKET NOK ----"
270 PRINT
280 PRINT "LAD OS SÅ KOMME IGANG."
290 O = 1
300 C = 200
310 S = 200
320 P = 0
330 PRINT
340 IF C < = 5 THEN 4140
350 PRINT "GEBYRET ER 5 KR.  JEG GIVER."
360 PRINT
370 IF S > 5 THEN 390
380 GOSUB 4300
390 P = P + 10
400 S = S - 5
410 C = C - 5
420 FOR Z = 1 TO 10
430 GOSUB 2120
440 NEXT Z
450 PRINT "DIN HÅND SER SÅLEDES UD:"
460 PRINT "-----"
470 N = 1
480 GOSUB 2240
490 N = 6
500 I = 2

510 GOSUB 2550
520 PRINT
530 IF I < > 6 THEN 670
540 IF FN A(O) < = 7 THEN 570
550 X = 11100
560 GOTO 620
570 IF FN A(O) < = 7 THEN 600
580 X = 11110
590 GOTO 620
600 IF FN A(O) > = 1 THEN 650
610 X = 11111
620 I = 7
630 Z = 23
640 GOTO 790
650 Z = 1
660 GOTO 710
670 IF U > = 13 THEN 750
680 IF FN A(O) > = 2 THEN 700
690 GOTO 620
700 Z = 0
710 K = 0
720 PRINT "JEG SER."
730 PRINT
740 GOTO 840
750 IF U < = 16 THEN 780
760 Z = 2
770 IF FN A(O) > = 1 THEN 790
780 Z = 35
790 V = Z + FN A(O)
800 GOSUB 3910
810 PRINT "JEG ÅBNER MED";V
820 PRINT
830 K = V
840 GOSUB 3430
850 GOSUB 870
860 GOTO 1100
870 IF I < > 3 THEN 1030
880 PRINT
890 PRINT "JEG VINDER."
900 PRINT
910 C = C + P
920 PRINT "JEG HAR NU";C;"KR."
930 PRINT "      DU HAR";S;"KR."
940 PRINT
950 PRINT "ØNSKER DU AT FORTSÆTTE ";
960 INPUT H$
970 IF LEFT$(H$,1) = "J" THEN 320
980 IF LEFT$(H$,1) = "N" THEN 4640
990 PRINT
1000 PRINT "SVAR VENLIGST 'JA' ELLER 'NEJ'."
1010 PRINT
1020 GOTO 950
1030 IF I < > 4 THEN 1090
1040 PRINT
1050 PRINT "DU VINDER."
1060 PRINT
1070 S = S + P
1080 GOTO 920
1090 RETURN
1100 PRINT
1110 PRINT "HVOR MANGE KORT VIL DU KØBE ";
1120 INPUT T
1130 PRINT
1140 IF T = 0 THEN 1310
1150 Z = 10
1160 IF T < 4 THEN 1190
1170 PRINT "DU KAN IKKE KØBE FLERE END 3 KORT."
1180 GOTO 1100
1190 PRINT "INDTAST NUMRENE PÅ DE KORT DU VIL BYTTE"
1200 PRINT "-----"
1210 FOR Q = 1 TO T
1220 PRINT "BYT KORT NUMMER ";
1230 INPUT U
1240 GOSUB 2110
1250 NEXT Q
1260 PRINT
1270 PRINT "DIN NYE HÅND SER SÅDAN UD:"
1280 PRINT "-----"
1290 N = 1
1300 GOSUB 2240
1310 Z = 10 + T
1320 FOR U = 6 TO 10
1330 IF INT (X / 10 ^ (U - 6)) < > 10 * INT (X /
10 ^ (U - 5)) THEN 1350
1340 GOSUB 2110
1350 NEXT U
1360 PRINT
1370 PRINT "JEG TAGER";Z - 10 - T;"KORT"

```

```

1380 IF Z = 11 + T THEN 1410
1390 PRINT
1400 GOTO 1420
1410 PRINT
1420 N = 6
1430 V = I
1440 I = 1
1450 GOSUB 2550
1460 B = U
1470 M = D
1480 IF V < > 7 THEN 1510
1490 Z = 28
1500 GOTO 1650
1510 IF I < > 6 THEN 1540
1520 Z = 1
1530 GOTO 1650
1540 IF U > = 13 THEN 1590
1550 Z = 2
1560 IF FN A(O) < > 6 THEN 1580
1570 Z = 19
1580 GOTO 1650
1590 IF U > = 16 THEN 1640
1600 Z = 19
1610 IF FN A(O) < > 8 THEN 1630
1620 Z = 11
1630 GOTO 1650
1640 Z = 2
1650 K = 0
1660 GOSUB 3430
1670 IF T < > .5 THEN 1770
1680 IF V = 7 THEN 1720
1690 IF I < > 6 THEN 1720
1700 PRINT "JEG SER."
1710 GOTO 1780
1720 V = Z + FN A(O)
1730 GOSUB 3910
1740 PRINT "JEG SATSER";V
1750 K = V
1760 GOSUB 3440
1770 GOSUB 870
1780 PRINT
1790 PRINT "NU SAMMENLIGNER VI VORE TO HÆNDER."
1800 PRINT
1810 J$ = H$
1820 K$ = I$
1830 PRINT "MIN HÅND:"
1840 PRINT "-----"
1850 N = 6
1860 GOSUB 2240
1870 N = 1
1880 GOSUB 2550
1890 PRINT
1900 PRINT "    DU HAR ";
1910 K = D
1920 GOSUB 4160
1930 H$ = J$
1940 I$ = K$
1950 K = M
1960 PRINT
1970 PRINT "OG JEG HAR ";
1980 GOSUB 4160
1990 IF B > U THEN 880
2000 IF U > B THEN 1040
2010 IF H$ = "EN FLU" THEN 2080
2020 IF FN B(M) < FN B(D) THEN 1040
2030 IF FN B(M) > FN B(D) THEN 880
2040 PRINT
2050 PRINT "HÆNDERNE ER LIGE STÆRKE."
2060 PRINT "DE";P;"KR. FORBLIVER I PULJEN."
2070 GOTO 330
2080 IF FN B(M) > FN B(D) THEN 880
2090 IF FN B(D) > FN B(M) THEN 1040
2100 GOTO 2040
2110 Z = Z + 1
2120 A(Z) = 100 * INT (4 * RND (1)) + INT (100 *
RND (1))
2130 IF INT (A(Z) / 100) > 3 THEN 2120
2140 IF A(Z) - 100 * INT (A(Z) / 100) > 12 THEN
2120
2150 IF Z = 1 THEN 2230
2160 FOR K = 1 TO Z - 1
2170 IF A(Z) = A(K) THEN 2120
2180 NEXT K
2190 IF Z < = 10 THEN 2230
2200 N = A(U)
2210 A(U) = A(Z)
2220 A(Z) = N
2230 RETURN
2240 FOR Z = N TO N + 4
2250 IF Z < 10 THEN PRINT " ";
2260 PRINT Z;"-- ";
2270 GOSUB 2320
2280 PRINT "I";
2290 GOSUB 2450
2300 NEXT Z
2310 RETURN
2320 K = FN B(A(Z))
2330 IF K < > 9 THEN 2350
2340 PRINT "KNÆGT ";
2350 IF K < > 10 THEN 2370
2360 PRINT " DAME ";
2370 IF K < > 11 THEN 2390
2380 PRINT "KONGE ";
2390 IF K < > 12 THEN 2410
2400 PRINT " ES ";
2410 IF K > = 9 THEN 2440
2420 IF K + 2 < > 10 THEN PRINT " ";
2430 PRINT " ";K + 2;
2440 RETURN
2450 K = INT (A(Z) / 100)
2460 IF K < > 0 THEN 2480
2470 PRINT " KLØR"
2480 IF K < > 1 THEN 2500
2490 PRINT " RUDER"
2500 IF K < > 2 THEN 2520
2510 PRINT " HJERTER"
2520 IF K < > 3 THEN 2540
2530 PRINT " SPAR"
2540 RETURN
2550 U = 0
2560 FOR Z = N TO N + 4
2570 B(Z) = FN B(A(Z))
2580 IF Z = N + 4 THEN 2610
2590 IF INT (A(Z) / 100) < > INT (A(Z + 1) / 100)
THEN 2610
2600 U = U + 1
2610 NEXT Z
2620 IF U < > 4 THEN 2690
2630 X = 11111
2640 D = A(N)
2650 H$ = "EN FLU"
2660 I$ = "HSH I"
2670 U = 15
2680 RETURN
2690 FOR Z = N TO N + 3
2700 FOR K = Z + 1 TO N + 4
2710 IF B(Z) < = B(K) THEN 2770
2720 X = A(Z)
2730 A(Z) = A(K)
2740 B(Z) = B(K)
2750 A(K) = X
2760 B(K) = A(K) - 100 * INT (A(K) / 100)
2770 NEXT K
2780 NEXT Z
2790 X = 0
2800 FOR Z = N TO N + 3
2810 IF B(Z) < > B(Z + 1) THEN 2850
2820 X = X + 11 * 10 ^ (Z - N)
2830 D = A(Z)
2840 GOSUB 3140
2850 NEXT Z
2860 IF X < > 0 THEN 3000
2870 IF B(N) + 3 < > B(N + 3) THEN 2900
2880 X = 1111
2890 U = 10
2900 IF B(N + 1) + 3 < > B(N + 4) THEN 3000
2910 IF U < > 10 THEN 2980
2920 U = 14
2930 H$ = "STRAIG"
2940 I$ = "HT"
2950 X = 11111
2960 D = A(N + 4)
2970 RETURN
2980 U = 10
2990 X = 11110
3000 IF U > = 10 THEN 3070
3010 D = A(N + 4)
3020 H$ = "SKRALD"
3030 I$ = " "
3040 U = 9
3050 X = 11000
3060 GOTO 3120
3070 IF U < > 10 THEN 3100
3080 IF I = 1 THEN 3120
3090 GOTO 3130
3100 IF U > 12 THEN 3130

```

```

3110 IF FN B(D) > 6 THEN 3130
3120 I = 6
3130 RETURN
3140 IF U > = 11 THEN 3190
3150 U = 11
3160 H$ = "ET PA"
3170 I$ = "R I "
3180 RETURN
3190 IF U < > 11 THEN 3290
3200 IF B(Z) < > B(Z - 1) THEN 3250
3210 H$ = "TRE EN"
3220 I$ = "S I "
3230 U = 13
3240 RETURN
3250 H$ = "TO PAR"
3260 I$ = " I "
3270 U = 12
3280 RETURN
3290 IF U > 12 THEN 3340
3300 U = 16
3310 H$ = "FULL H"
3320 I$ = "OUSE I "
3330 RETURN
3340 IF B(Z) < > B(Z - 1) THEN 3390
3350 U = 17
3360 H$ = "FIRE E"
3370 H$ = "NS I "
3380 RETURN
3390 U = 16
3400 H$ = "FULDT "
3410 I$ = "HUS "
3420 RETURN
3430 G = 0
3440 PRINT "HVAD ER DIN INDSATS ";
3450 INPUT T
3460 PRINT
3470 IF T - INT (T) = 0 THEN 3540
3480 IF K < > 0 THEN 3510
3490 IF G < > 0 THEN 3510
3500 IF T = .5 THEN 3830
3510 PRINT "INGEN SMÅPENGE....TAK!"
3520 PRINT
3530 GOTO 3440
3540 IF S - G - T > = 0 THEN 3570
3550 GOSUB 4300
3560 GOTO 3440
3570 IF T < > 0 THEN 3600
3580 I = 3
3590 GOTO 3800
3600 IF G + T > = K THEN 3650
3610 PRINT "HVIS DU IKKE KAN SVARE PÅ MIN"
3620 PRINT "INDSATS, MÅ DU LÆGGE KORTENE."
3630 PRINT
3640 GOTO 3440
3650 G = G + T
3660 IF G = K THEN 3800
3670 IF Z < > 1 THEN 3840
3680 IF G > 5 THEN 3720
3690 IF Z > = 2 THEN 3770
3700 V = 5
3710 GOTO 3840
3720 IF Z = 1 THEN 3740
3730 IF T < = 25 THEN 3770
3740 I = 4
3750 PRINT "JEG LÆGGER KORTENE."
3760 RETURN
3770 IF Z = 2 THEN 3850
3780 PRINT "JEG SER."
3790 K = G
3800 S = S - G
3810 C = C - K
3820 P = P + G + K
3830 RETURN
3840 IF G > 3 * Z THEN 3770
3850 V = G - K + FN A(O)
3860 GOSUB 3910
3870 PRINT "JEG SER, OG FORHØJER MED";V
3880 PRINT
3890 K = G + V
3900 GOTO 3440
3910 IF C - G > = 0 THEN 4130
3920 IF G < > 0 THEN 3950
3930 V = C
3940 RETURN
3950 IF C - G > = 0 THEN 3780
3960 IF (O / 2) < > INT (O / 2) THEN 4050
3970 PRINT "VIL DU KØBE DIT UR TILBAGE FOR"
3980 PRINT "50 KRONER ";
3990 INPUT J$
4000 PRINT
4010 IF LEFT$ (J$,1) = "N" THEN 4050
4020 C = C + 50
4030 O = O / 2
4040 RETURN
4050 IF (O / 3) < > INT (O / 3) THEN 4140
4060 PRINT "VIL DU KØBE DIN SLIPSENÅL"
4070 PRINT "TILBAGE FOR 50 KRONER ";
4080 INPUT J$
4090 PRINT
4100 IF LEFT$ (J$,1) = "N" THEN 4140
4110 C = C + 50
4120 O = O / 3
4130 RETURN
4140 PRINT "JEG ER FLAD. TIL LYKKE. DU VANDT!"
4150 GOTO 4640
4160 PRINT H$;I$;
4170 IF H$ < > "EN FLU" THEN 4220
4180 K = INT (K / 100)
4190 GOSUB 2460
4200 PRINT
4210 RETURN
4220 K = FN B(K)
4230 GOSUB 2330
4240 IF H$ = "SKRALD" THEN 4260
4250 IF H$ < > "STRAIG" THEN 4280
4260 PRINT " ER HØJESTE KORT."
4270 RETURN
4280 PRINT
4290 RETURN
4300 PRINT "DU KAN IKKE SATSE DET DU IKKE HAR!"
4310 PRINT
4320 IF O / 2 = INT (O / 2) THEN 4470
4330 PRINT "KUNNE DU TÆNKE DIG AT SÆLGE DIT UR ";
4340 INPUT J$
4350 PRINT
4360 IF LEFT$ (J$,1) = "N" THEN 4470
4370 IF FN A(O) > = 7 THEN 4410
4380 PRINT "JEG VIL GIVE DIG 75 KRONER FOR DET."
4390 S = S + 75
4400 GOTO 4440
4410 PRINT "SIKKE' ET ELENDIGT UR DU HAR - JEG"
4420 PRINT "KAN KUN GIVE DIG 25 KRONER FOR DET."
4430 S = S + 25
4440 O = O * 2
4450 PRINT
4460 RETURN
4470 IF O / 3 = INT (O / 3) THEN 4620
4480 PRINT "HAR DU LYST TIL AT KOMME AF MED DIN"
4490 PRINT "DIAMANT-SLIPSENÅL ";
4500 INPUT J$
4510 PRINT
4520 IF LEFT$ (J$,1) = "N" THEN 4610
4530 IF FN A(O) > = 6 THEN 4570
4540 PRINT "DU ER NU 100 KRONER RIGERE!"
4550 S = S + 100
4560 GOTO 4590
4570 PRINT "DET ER JO SIMILI, DU FÅR 25 KRONER"
4580 S = S + 25
4590 O = O * 3
4600 PRINT
4610 RETURN
4620 PRINT "DU ER FLAD SOM EN FREGNE....."
4630 PRINT "FARVEL MED DIG KAMMERAT."
4640 END

```

# Roulette

Dette spil simulerer en amerikansk roulette. Det amerikanske roulett hjul har 38 tal (1 til 36 samt 0 og 00). Det europæiske hjul har kun 37 tal (1 til 36 og 0). På Bahama, i Puerto Rico og i de syd-amerikanske lande er man ved at gå over til den amerikanske roulette, fordi denne giver kasinoet bedre procenter. Lige og ulige tal ligger skiftevis rundt om hjulet, og det gør rød og sort også. Selve hjulets konstruktion giver en høj grad af tilfældighed i de talrækker, der opnås. Faktum er, at roulett hjul ofte anvendes til at fremstille tabeller med tilfældige tal.

I dette spil kan indsatsen ligge mellem 5 og 500 kroner, og du kan satse på rødt, sort, lige, ulige, de første 18 tal, de sidste 18 tal, en lodret kolonne tal eller et enkelt tal. Du kan gøre lige så mange indsatser, du vil, hver gang hjulet drejes rundt.

Der findes ingen langsigtede vinderstrategier for roulettspillet. En god metode er dog »fordobling«. Her sætter du den laveste indsats (her 5 kr.) på et væddemål med lige odds (dvs. lige, ulige, sort eller rødt). Hvis du taber, fordobler du indsatsen (til 10 kr.) og sætter igen. Hvis du taber indsatsen igen, fordobler du indsatsen endnu en gang (til 20 kr.) og satser igen. Fortsæt med at fordoble indsatserne, indtil du vinder (dvs. når du har fået de indsatser, du har sat i serien tilbage). Så snart du har vundet, starter du forfra igen med at sætte 5 kr. Lad være med at sætte mere end 5 kr., hvis du ikke bruger »fordoblings-princippet«. Sæt aldrig på andre væddemål end dem, der giver lige odds. Held og lykke!

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
ROULETTE  
-----

INDTAST DATO (DAG,MÅNED,ÅR) ?1,12,1986

VELKOMMEN TIL ROULETTE I LAS VEGAS

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

SPILLEBORDET SER SÅLEDES UD:

(\* BETYDER RØD)

1*	2	3*
4	5*	6
7*	8	9*
10	11	12*

13	14*	15
16*	17	18*
19*	20	21*
22	23*	24

25*	26	27*
28	29	30*
31	32*	33
34*	35	36*

-----  
00 0  
-----

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

DE FORSKELLIGE FORMER FOR INDSATS:

TALLENE FRA 1 TO 36 ER EN INDSATS  
PÅ DET TAL DER INDTASTES. DISSE  
INDSATSER GIVER 35:1

2:1 VÆDDEMÅL ER:

37) 1-12 40) FØRSTE RÆKKE  
38) 13-24 41) ANDEN RÆKKE  
39) 25-36 42) TREDIE RÆKKE

1:1 VÆDDEMÅL ER:  
43) 1-18 46) ULIGE  
44) 19-36 47) RØD  
45) LIGE 48) SORT

49) 0 OG 50) 00 GIVER 35:1

BEMÆRK, AT 0 OG 00 IKKE TÆLLER MED  
I DE ANDRE KOMBINATIONSVÆDDEMÅL

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

NÅR DU BLIVER SPURGT OM DIN INDSATS,  
SKAL DU INDTASTE VÆDDEMÅLETS NUMMER OG  
INDSATSEN ADSKILDT AF ET KOMMA.  
HVIS DU F.EKS. VIL SPILLE 500 KR. PÅ  
SORT, SÅ SKAL DU INDTASTE: '48,500'.

MINDSTE VÆDDEMÅL ER 5, STØRSTE 500 KR.

HVOR MANGE INDSATSER ?2

INDSATS NUMMER 1 ?39,500

INDSATS NUMMER 2 ?48,200

HJULET KØRER.....

27 RØD

DU VINDER 1000 KR. PÅ INDSATS NR. 1  
DU TABER 200 KR. PÅ INDSATS NR. 2

TOTALT

-----  
KASINO: 99200  
DIG: 1800

VIL DU SPILLE IGEN ('J' ELLER 'N') ?J

HVOR MANGE INDSATSER ?1

INDSATS NUMMER 1 ?48,200

HJULET KØRER.....

29 SORT

DU VINDER 200 KR. PÅ INDSATS NR. 1

TOTALT

-----  
KASINO: 99000  
DIG: 2000

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

HVEM SKAL CHECKEN  
UDSTEDES TIL ?CHRISTIAN BORUP

-----  
\*\*BANK OF LAS VEGAS\*\* CHECK NO. 5403

LAS VEGAS D.1/12 1986

UDBETAL TIL: CHRISTIAN BORUP KR. 2000

D.A. TAMAT

-----XX-----  
-----

KOM SNART IGEN!

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          ROULETTE"
130 PRINT "-----"
140 PRINT
150 PRINT "INDTAST DATO (DAG,MÅNED,ÅR) ";
160 INPUT US$,VS$,WS$
170 PRINT
180 REM Roulette
190 REM David Joslin
200 PRINT "VELKOMMEN TIL ROULETTE I LAS VEGAS"
210 PRINT
220 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
230 INPUT Y$
240 PRINT
250 IF LEFT$(Y$,1) = "N" THEN 820
260 PRINT "SPILLEBORDET SER SÅLEDES UD:"
270 PRINT
280 PRINT "      (* BETYDER RØD)"
290 PRINT
300 PRINT "      1*   2   3*"
310 PRINT "      4     5*  6  "
320 PRINT "      7*   8   9*"
330 PRINT "      10  11  12*"
340 PRINT "      -----"
350 PRINT "      13  14*  15  "
360 PRINT "      16*  17  18*"
370 PRINT "      19*  20  21*"
380 PRINT "      22  23*  24  "
390 PRINT "      -----"
400 PRINT "      25*  26  27*"
410 PRINT "      28   29  30*"
420 PRINT "      31  32*  33  "
430 PRINT "      34*  35  36*"
440 PRINT "      -----"
450 PRINT "          00  0  "
460 PRINT
470 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
480 INPUT C$
490 PRINT
500 PRINT "DE FORSKELLIGE FORMER FOR INDSATS:"
510 PRINT
520 PRINT "TALLENE FRA 1 TO 36 ER EN INDSATS"
530 PRINT "PR DET TAL DER INDTASTES. DISSE"
540 PRINT "INDSATSER GIVER 35:1"
550 PRINT
560 PRINT "2:1 VÆDDEMÅL ER:"
570 PRINT "37) 1-12  40) FØRSTE RÆKKE"
580 PRINT "38) 13-24  41) ANDEN RÆKKE"
590 PRINT "39) 25-36  42) TREDIE RÆKKE"
600 PRINT
610 PRINT "1:1 VÆDDEMÅL ER:"
620 PRINT "43) 1-18  46) ULIGE"
630 PRINT "44) 19-36  47) RØD"
640 PRINT "45) LIGE  48) SORT"
650 PRINT
660 PRINT "49) 0  OG  50) 00 GIVER 35:1"
670 PRINT
680 PRINT "BEMÆRK, AT 0 OG 00 IKKE TÆLLER MED"
690 PRINT "I DE ANDRE KOMBINATIONSVÆDDEMÅL"
700 PRINT
710 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
720 INPUT C$
730 PRINT
740 PRINT "NÅR DU BLIVER SPURGT OM DIN INDSATS,"
750 PRINT "SKAL DU INDTASTE VÆDDEMÅLETS NUMMER OG"
760 PRINT "INDSATSEN ADSKILDT AF ET KOMMA."
770 PRINT "HVIS DU F.EKS. VIL SPILLE 500 KR. PR"
780 PRINT "SORT, SÅ SKAL DU INDTASTE: '48,500'."
790 PRINT
800 PRINT "MINDSTE VÆDDEMÅL ER 5, STØRSTE 500 KR."
810 PRINT
820 REM Program begynder her
830 DIM B(100),C(100),T(100),X(38)
840 DIM A(50)
850 FOR I = 1 TO 38
860 X(I) = 0
870 NEXT I
880 P = 1000
890 D = 100000
900 PRINT "HVOR MANGE INDSATSER ";
910 INPUT Y
920 PRINT
930 IF Y < 1 OR Y > INT(Y) THEN 900
940 FOR I = 1 TO 50
950 A(I) = 0
960 NEXT I
970 FOR C = 1 TO Y

```

```

980 PRINT "INDSAT NUMMER";C;
990 INPUT X,Z
1000 PRINT
1010 B(C) = Z
1020 T(C) = X
1030 IF X < 1 OR X > 50 OR X > INT(X) THEN 980
1040 IF Z < 1 OR Z > INT(Z) THEN 980
1050 IF Z < 5 OR Z > 500 THEN 980
1060 IF A(X) = 0 THEN 1100
1070 PRINT "DET TAL HAR DU ALLEREDE SPILLET PR."
1080 PRINT
1090 GOTO 980
1100 A(X) = 1
1110 NEXT C
1120 PRINT "HJULET KØRER....."
1130 PRINT
1140 S = INT(100 * RND(1))
1150 IF S = 0 OR S > 38 THEN 1140
1160 X(S) = X(S) + 1
1170 IF S < 37 THEN 1230
1180 IF S = 37 THEN 1210
1190 PRINT "00"
1200 GOTO 1330
1210 PRINT "0"
1220 GOTO 1330
1230 RESTORE
1240 FOR I = 1 TO 18
1250 READ R
1260 IF R = S THEN 1310
1270 NEXT I
1280 A$ = "SORT"
1290 PRINT S;A$
1300 GOTO 1330
1310 A$ = "RØD"
1320 GOTO 1290
1330 PRINT
1340 C = 0
1350 C = C + 1
1360 IF T(C) < 37 THEN 2050
1370 ON T(C) - 36 GOTO
1400,1520,1550,1580,1630,1680,1730,1810,1840,1870,1900,
1970,2050
1380 GOTO 2050
1390 GOTO 2690
1400 REM 1-12 - (37) - 2:1
1410 IF S < = 12 THEN 1470
1420 PRINT "DU TABER ";B(C);"KR. ";
1430 PRINT "PR INDSATS NR.";C
1440 D = D + B(C)
1450 P = P - B(C)
1460 GOTO 1510
1470 PRINT "DU VINDER";B(C) * 2;"KR. ";
1480 PRINT "PR INDSATS NR.";C
1490 D = D - B(C) * 2
1500 P = P + B(C) * 2
1510 GOTO 2160
1520 REM 13-24 - (38) - 2:1
1530 IF S > 12 AND S < 25 THEN 1470
1540 GOTO 1420
1550 REM 25-36 - (39) - 2:1
1560 IF S > 24 AND S < 37 THEN 1470
1570 GOTO 1420
1580 REM FØRSTE KØL. - (40) - 2:1
1590 FOR I = 1 TO 34 STEP 3
1600 IF S = I THEN 1470
1610 NEXT I
1620 GOTO 1420
1630 REM ANDEN KØL. - (41) - 2:1
1640 FOR I = 2 TO 35 STEP 3
1650 IF S = I THEN 1470
1660 NEXT I
1670 GOTO 1420
1680 REM TREDIE KØL. - (42) - 2:1
1690 FOR I = 3 TO 36 STEP 3
1700 IF S = I THEN 1470
1710 NEXT I
1720 GOTO 1420
1730 REM 1-18 - (43) - 1:1
1740 IF S < 19 THEN 1760
1750 GOTO 1420
1760 PRINT "DU VINDER";B(C);"KR. ";
1770 PRINT "PR INDSATS NR.";C
1780 D = D - B(C)
1790 P = P + B(C)
1800 GOTO 2160
1810 REM 19-36 - (44) - 1:1
1820 IF S < 37 AND S > 18 THEN 1760
1830 GOTO 1420
1840 REM LIGE - (45) - 1:1

```

```


1850 IF S / 2 = INT ( S / 2 ) AND S < 37 THEN 1760
1860 GOTO 1420
1870 REM ULIGE - (46) - 1:1
1880 IF S / 2 < > INT ( S / 2 ) AND S < 37 THEN 1760
1890 GOTO 1420
1900 REM RØD - (47) - 1:1
1910 RESTORE
1920 FOR I = 1 TO 18
1930 READ R
1940 IF S = R THEN 1760
1950 NEXT I
1960 GOTO 1420
1970 REM SORT - (48) - 1:1
1980 RESTORE
1990 FOR I = 1 TO 18
2000 READ R
2010 IF S = R THEN 1420
2020 NEXT I
2030 IF S > 36 THEN 1420
2040 GOTO 1760
2050 REM 00-36 - (49)+(50) - 35:1
2060 IF T(C) < 49 THEN 2100
2070 IF T(C) = 49 AND S = 37 THEN 2120
2080 IF T(C) = 50 AND S = 38 THEN 2120
2090 GOTO 1420
2100 IF T(C) = S THEN 2120
2110 GOTO 1420
2120 PRINT "DU VINDER";B(C) * 35;"KR. ";
2130 PRINT "PR INDSATS NR.";C
2140 D = D - B(C) * 35
2150 P = P + B(C) * 35
2160 IF C < Y THEN 1350
2170 PRINT
2180 PRINT "TOTALT"
2190 PRINT "-----"
2200 PRINT "KASINO:";D
2210 PRINT "  DIG:";P
2220 PRINT
2230 IF P > 0 THEN 2260
2240 PRINT "HOVSA, DU HAR BRUGT DIN SIDSTE KRONE!"
2250 GOTO 2640
2260 IF D > 0 THEN 2310
2270 PRINT "DU HAR SPRÆNGT BANKEN!"

```

```

2280 PRINT
2290 P = 101000
2300 GOTO 2360
2310 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
2320 INPUT Y$
2330 PRINT
2340 IF LEFT$(Y$,1) = "J" THEN 900
2350 DATA 1,3,5,7,9,12,14,16,18,19,21,23,25,27,30,32,34,36
2360 IF P < 1 THEN 2640
2370 PRINT "HVEM SKAL CHECKEN"
2380 PRINT "UDSTEDES TIL ";
2390 INPUT B$
2400 PRINT
2410 PRINT
2420 FOR I = 1 TO 39
2430 PRINT "-";
2440 NEXT I
2450 PRINT
2460 PRINT "***BANK OF LAS VEGAS**      "; "CHECK NO.:"; " " INT (10000 * RND (1))
2470 PRINT
2480 PRINT "LAS VEGAS D.:";US$;" / ";VS$;" " ;WS
2490 PRINT
2500 PRINT "UDBETAL TIL: ";B$;" KR.:";
2510 PRINT P
2520 PRINT
2530 PRINT
2540 PRINT "
2550 PRINT "
2560 FOR I = 1 TO 39
2570 PRINT "-";
2580 NEXT
2590 PRINT
2600 PRINT
2610 PRINT
2620 PRINT "KOM SNART IGEN!"
2630 GOTO 2680
2640 PRINT "TAK FOR ALLE DINE PENGE."
2650 PRINT
2660 PRINT "JEG VIL BRUGE DEM TIL AT KØBE ET NYT"
2670 PRINT "ROULETTE-HJUL AF ÆGTE GULD...MANGE TAK"
2680 PRINT
2690 END

```









# Saks-sten-papir

Kan du huske den gamle leg saks-sten-papir? Du og din modspiller slår i luften med knyttede hænder tre gange og viser derefter enten håndfladen (papir), en knyttet hånd (sten) eller to fingre (saks). Afhængigt af hvad spillerne viser, er omgangen enten uafgjort (begge viser det samme), eller også vinder den ene. Papiret kan pakke stenen ind, så derfor vinder papir over sten. Saks kan klippe papir, så derfor vinder saks. Sten kan ødelægge saksen, så derfor vinder sten.

I denne datamatudgave af spillet kan du spille op til ti gange mod datamaten.

Charles Lund skrev spillet, mens han gik på The American School i Haag, Holland.

```
-----  
SAKS-STEN-PAPIR  
-----
```

HVOR MANGE SPIL VIL DU SPILLE ?3

SPIL NUMMER 1

3=STEN 2=SAKS 1=PAPIR

1..2..3 HVAD VÆLGER DU ?1

HER ER MIT VALG ..... PAPIR

DET STÅR LIGE....INGEN VINDER!

SPIL NUMMER 2

3=STEN 2=SAKS 1=PAPIR

1..2..3 HVAD VÆLGER DU ?2

HER ER MIT VALG ..... PAPIR

DU VINDER!!!!

SPIL NUMMER 3

3=STEN 2=SAKS 1=PAPIR

1..2..3 HVAD VÆLGER DU ?1

HER ER MIT VALG ..... PAPIR

DET STÅR LIGE....INGEN VINDER!

HER ER DET ENDELIGE RESULTAT:

JEG HAR VUNDET 0 SPIL.  
DU HAR VUNDET 1 SPIL.

MENS 2 SPIL ENDTE UAFGJORT.

TAK FOR SPILLET!

```
100 PRINT "-----"  
110 PRINT "          SAKS-STEN-PAPIR"  
120 PRINT "-----"  
130 PRINT  
140 PRINT "HVOR MANGE SPIL VIL DU SPILLE ";  
150 INPUT Q  
160 IF Q < 11 THEN 200  
170 PRINT "DESVÆRRE KAN DU IKKE SPILLE SÅ MANGE"  
180 PRINT "SPIL - INDFAST EN LAVERE VÆRDI."  
190 GOTO 130  
200 FOR G = 1 TO Q  
210 PRINT  
220 PRINT "SPIL NUMMER";G  
230 PRINT  
240 X = INT ( RND (1) * 3 + 1)  
250 PRINT "3=STEN 2=SAKS 1=PAPIR"  
260 PRINT  
270 PRINT "1..2..3 HVAD VÆLGER DU ";  
280 INPUT K  
290 PRINT  
300 IF (K - 1) * (K - 2) * (K - 3) < > 0 THEN  
PRINT "UGYLDIG INDFAST.": GOTO 230  
310 PRINT "HER ER MIT VALG ..... ";  
320 ON X GOTO 330,350,370  
330 PRINT "PAPIR"  
340 GOTO 390  
350 PRINT "SAKS"  
360 GOTO 390  
370 PRINT "STEN"  
380 GOTO 390  
390 PRINT  
400 IF X = K THEN 520  
410 IF X > K THEN 500  
420 IF X = 1 THEN 460  
430 PRINT "DU VINDER!!!!!"  
440 H = H + 1  
450 GOTO 530  
460 IF K < > 3 THEN 430  
470 PRINT "HOVSA!! JEG VINDER!!!"  
480 C = C + 1  
490 GOTO 530  
500 IF K < > 1 OR X < > 3 THEN 470  
510 GOTO 430  
520 PRINT "DET STÅR LIGE....INGEN VINDER!"  
530 NEXT G  
540 PRINT  
550 PRINT "HER ER DET ENDELIGE RESULTAT:"  
560 PRINT  
570 PRINT "JEG HAR VUNDET";C;"SPIL."  
580 PRINT " DU HAR VUNDET";H;"SPIL."  
590 PRINT  
600 PRINT "MENS";Q - (C + H);"SPIL ENDTE UAFGJORT."  
610 PRINT  
620 PRINT "TAK FOR SPILLET!"  
630 END
```



# Skytte

I dette spil kan du styre artilleriilden mod et stillestående mål. Du skal angive elevationsvinklen (i grader). Elevationsvinklen er det antal grader, kanonen danner med jorden. En elevationsvinkel på 45 grader giver den længste skudlængde, mens værdier over og under 45 grader giver kortere skudlængder.

Du får fem skud til at ramme modstanderen, inden han rammer dig. Skudlængderne varierer mellem 6.000 og 2.000 meter, og destruktionsradius er 30 meter. Elevationsvinklen skal ligge indenfor 0.2 grader for at ramme.

Tom Kloos ved Oregon Museum of Science and Industry i Portland skrev det oprindelige program. David Ahl foretog omfattende ændringer.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

-----  
 SKYTTE  
 -----

DU HAR BEFALINGEN OG GIVER ORDRER TIL  
 ET KANONMANDSKAB. DU INDTASTER DEN  
 ELEVATIONSVINKLEN SOM KANONEN SKAL  
 HAVE FOR AT RAMME MÅLET. ET NEDSLAG  
 INDEN FOR EN CIRKEL PÅ 100 METER,  
 ØDELÆGGER MÅLET HELT.

KANONENS RÆKKEVIDDE ER 34761 METER.

AFSTANDEN TIL MÅLET ER 12686 METER.

ELEVATION ?80  
 790 METER FOR KORT.

ELEVATION ?79.5  
 222 METER FOR KORT.

ELEVATION ?79.2  
 117 METER FORBI MÅLET.

ELEVATION ?79.3

\*\*\*\*\* MÅLET ØDELAGT \*\*\*\*\*

DU HAR BRUGT 4 GRANATER.

EN SPEJDER HAR OBSERVERET MERE  
 FJENDTLIG AKTIVITET.....

AFSTANDEN TIL MÅLET ER 11314 METER.

ELEVATION ?79.3  
 1377 METER FORBI MÅLET.

ELEVATION ?81  
 565 METER FOR KORT.

ELEVATION ?80  
 582 METER FORBI MÅLET.

ELEVATION ?80.5

\*\*\*\*\* MÅLET ØDELAGT \*\*\*\*\*

DU HAR BRUGT 4 GRANATER.

EN SPEJDER HAR OBSERVERET MERE  
 FJENDTLIG AKTIVITET.....

AFSTANDEN TIL MÅLET ER 29151 METER.

ELEVATION ?30  
 952 METER FORBI MÅLET.

ELEVATION ?40  
 5082 METER FORBI MÅLET.

ELEVATION ?20  
 6808 METER FOR KORT.

ELEVATION ?50  
 5083 METER FORBI MÅLET.

ELEVATION ?70  
 6802 METER FOR KORT.

ELEVATION ?60

\*\*BANGG! \*\*

FJENDEN HAR UDRADERET DIG HELT!!!!

DU MÅ VIST HELLERE GÅ PÅ BEGYNDERKURSET  
 IGEN..... DU TRÆNGER VIRKELIG TIL DET!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

OK. SÅ GÅR VI HJEM TIL LEJREN.

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          SKYTTE"
130 PRINT "-----"
140 PRINT
150 PRINT "DU HAR BEFALINGEN OG GIVER ORDRER TIL"
160 PRINT "ET KANONMANDSKAB. DU INDTASTER DEN"
170 PRINT "ELEVATIONSVINKLEN SOM KANONEN SKAL"
180 PRINT "HAVE FOR AT RAMME MÅLET. ET NEDSLAG"
190 PRINT "INDEN FOR EN CIRKEL PÅ 100 METER,"
200 PRINT "ØDELÆGGER MÅLET HELT."
210 PRINT
220 R = INT (40000 * RND (1) + 20000)
230 PRINT "KANONENS RÆKKEVIDDE ER";R;"METER."
240 PRINT
250 Z = 0
260 S1 = 0
270 T = INT (R * (.1 + .8 * RND (1)))
280 S = 0
290 GOTO 500
300 PRINT "MINDSTE ELEVATION ER 1 GRAD."
310 GOTO 510
320 PRINT "STØRSTE ELEVATION ER 89 GRADER."
330 GOTO 510
340 PRINT "          "; ABS (E);"METER FORBI
MÅLET."
350 GOTO 510
360 PRINT "          "; ABS (E);"METER FOR
KORT."
370 GOTO 510
380 PRINT
390 PRINT "***** MÅLET ØDELAGT *****"
400 PRINT
410 PRINT "DU HAR BRUGT";S;"GRANATER."
420 S1 = S1 + S
430 IF Z = 4 THEN 710
440 Z = Z + 1
450 PRINT
460 PRINT "EN SPEJDER HAR OBSERVERET MERE"
470 PRINT "FJENDTLIG AKTIVITET....."
480 PRINT
490 GOTO 270
500 PRINT "AFSTANDEN TIL MÅLET ER";T;"METER."
510 PRINT
520 PRINT "ELEVATION ";
530 INPUT B
540 IF B > 89 THEN 320
550 IF B < 1 THEN 300
560 S = S + 1
570 IF S < 6 THEN 640
580 PRINT
590 PRINT "**BANGG! **"
600 PRINT
610 PRINT "FJENDEN HAR UDRADERET DIG HELT!!!!"
620 PRINT
630 GOTO 770
640 B2 = 2 * B / 57.3
650 I = R * SIN (B2)
660 X = T - I
670 E = INT (X)
680 IF ABS (E) < 100 THEN 380
690 IF E > 100 THEN 360
700 GOTO 340
710 PRINT
720 PRINT "DU HAR BRUGT";S1;"GRANATER IALT."
730 PRINT
740 IF S1 > 18 THEN 770
750 PRINT "GODT SKUDT !!"
760 GOTO 790
770 PRINT "DU MÅ VIST HELLERE GÅ PÅ BEGYNDERKURSET"
780 PRINT "IGEN..... DU TRÆNGER VIRKELIG TIL DET!"
790 PRINT
800 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
810 INPUT Z$
820 PRINT
830 IF LEFT$ (Z$,1) = "J" THEN 220
840 PRINT "OK. SÅ GÅR VI HJEM TIL LEJREN."
850 END

```

# Slagskib

Slagskib er baseret på det populære spil »Sænke slagskibe«, der hovedsagelig spilles for at gøre spillerne bekendt med placeringen og betegnelsen for punkter i et koordinatsystem.


I spillet opstiller datamaten først fjendens flåde i et net, der måler 6 gange 6 felter. Flåden består af seks skibe: 2 destroyere (skibe nummer 1 og 2), der er 2 felter lange, 2 krydsere (skibe nummer 3 og 4), der er 3 felter lange, og 2 slagskibe (nummer 5 og seks), der er 4 felter lange. Programmet udskriver derefter flådens stilling i et kodet eller forvansket diagram (se kørslen af programmet nedenfor). Så er det din tur til at prøve at sænke de forskellige skibe ved at indtaste koordinaterne for det sted, hvor du ønsker at kaste dine bomber. De koordinater, du indtaster, skal bestå af to tal, hver mellem 1 og 6, adskilt af et komma. Datamaten forsyner dig med svar (plask, fuldtræffer, etc.), hvilket du selv bør optegne på et net, der måler 6 gange 6 felter. På den måde opbygger du et diagram over den fjendtlige flådes virkelige placering. Dette diagram vil forhåbentlig være dig til stor hjælp, når du skal dechifrere det forvanskede diagram, som datamaten skriver ud, når spillet starter. Hver gang et skib sænkes, udskriver datamaten, hvilke skibe du har sænket og angiver også »FORBIER/TRÆFFER FORHOLDET«.

Det første, du bør lære, er at lokalisere og beskrive positionerne på nettet og især at skelne mellem f.eks. koordinaterne »3,4« og »4,3«. Den metode, vi anvender, svarer mere til den betegnelse for punkter, der anvendes i et koordinatsystem, end den der anvendes ved algebraiske matricer. Det første koordinatciffer angiver, hvor mange kolonner fra venstre punktet er beliggende, mens det andet koordinatciffer angiver rækkenummeret nedefra og op.

Den anden ting, du bør vide noget om, er »FORBIER/TRÆFFER FORHOLDET«. »FORBIER/TRÆFFER FORHOLDET« er antallet af træffere. Hvis du har haft 9 forbiere og 15 træffere, vil forholdet blive 9/15 eller 3/5 (de er begge korrekte). Datamaten vil dog altid udskrive FORBIER/TRÆFFER FORHOLDET som et kommatal, i dette tilfælde 0.6.

Hovedformålet med dette spil og dets undervisningsværdi ligger i spillerens forsøg på at dechifrere kortet over fjendens flåde. For at gøre dette skal du sammenligne det forvanskede net med det virkelige net, som du danner efterhånden, som du spiller spillet.

Den oprindelige forfatter af både programmet og spillevejledningen er Ray Westergard fra Lawrence Hall of Science, Berkeley, Californien.



MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

SLAGSKIB

FØLGENDE KORT OVER FJENDENS FLÅDE ER BLEVET AFFOTOGRAFERET, MEN ER ENDNU IKKE BLEVET DECHIFRERET.

```
0 0 1 4 5 0
0 1 4 3 5 0
0 4 3 6 5 0
0 3 0 6 5 2
0 0 0 6 2 0
0 0 0 6 0 0
```

DU KAN GØRE BRUG AF KORTET, HVIS DU KAN DECHIFRERE DET, MEN SØRG FOR, AT DEN METODE DU BRUGER HOLDES HEMMELIG.

LAD SPILLET BEGYNDE!!

DIT SKUD ?5,5  
PLASK!!!! DU RAMTE VANDET!!! PRØV IGEN.

DIT SKUD ?5,2  
FULDTRÆFFER I SKIB NUMMER: 2  
.....PRØV IGEN.

DIT SKUD ?6,3  
FULDTRÆFFER I SKIB NUMMER: 6  
.....PRØV IGEN.

DIT SKUD ?5,2  
DU HAR ALLEREDE RAMT SKIB NUMMER 2  
PÅ DENNE POSITION!! PLASK!!! PRØV IGEN!

DIT SKUD ?4,1  
FULDTRÆFFER I SKIB NUMMER: 2  
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

```
1 DESTROYERE
0 KRYDSERE
0 HANGARSKIBE
```

DIT FORBIER/TRÆFFER FORHOLD ER .66

DIT SKUD ?5,3  
FULDTRÆFFER I SKIB NUMMER: 6  
.....PRØV IGEN.

DIT SKUD ?4,3  
FULDTRÆFFER I SKIB NUMMER: 6  
.....PRØV IGEN.

DIT SKUD ?3,3  
FULDTRÆFFER I SKIB NUMMER: 6  
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

```
1 DESTROYERE
0 KRYDSERE
1 HANGARSKIBE
```

DIT FORBIER/TRÆFFER FORHOLD ER .33

DIT SKUD ?1,2  
FULDTRÆFFER I SKIB NUMMER: 5  
.....PRØV IGEN.

DIT SKUD ?1,3  
FULDTRÆFFER I SKIB NUMMER: 4  
.....PRØV IGEN.

DIT SKUD ?1,4  
FULDTRÆFFER I SKIB NUMMER: 1  
.....PRØV IGEN.

DIT SKUD ?2,2  
FULDTRÆFFER I SKIB NUMMER: 5  
.....PRØV IGEN.

```

DIT SKUD ?3,2
FULDTRÆFFER I SKIB NUMMER: 5
.....PRØV IGEN.

DIT SKUD ?4,2
FULDTRÆFFER I SKIB NUMMER: 5
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

  1 DESTROYERE
  0 KRYDSERE
  2 HANGARSKIBE

DIT FORBIER/TRÆFFER FORHOLD ER .16

DIT SKUD ?2,3
FULDTRÆFFER I SKIB NUMMER: 3
.....PRØV IGEN.

DIT SKUD ?2,4
FULDTRÆFFER I SKIB NUMMER: 4
.....PRØV IGEN.

DIT SKUD ?3,5
FULDTRÆFFER I SKIB NUMMER: 4
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

  1 DESTROYERE
  1 KRYDSERE
  2 HANGARSKIBE

DIT FORBIER/TRÆFFER FORHOLD ER .13

DIT SKUD ?2,5
FULDTRÆFFER I SKIB NUMMER: 1
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

  2 DESTROYERE
  1 KRYDSERE
  2 HANGARSKIBE

DIT FORBIER/TRÆFFER FORHOLD ER .12

DIT SKUD ?3,4
FULDTRÆFFER I SKIB NUMMER: 3
.....PRØV IGEN.

DIT SKUD ?4,5
FULDTRÆFFER I SKIB NUMMER: 3
OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!

INDTIL NU HAR FJENDEN MISTET:

  2 DESTROYERE
  2 KRYDSERE
  2 HANGARSKIBE

DIT FORBIER/TRÆFFER FORHOLD ER .11

DU HAR UDSLETTET FJENDES FLÅDE TOTALT!

SLUT FORBIER/TRÆFFER FORHOLD ER .11

*****
FØLGENDE KORT OVER FJENDENS FLÅDE ER
BLEVET AFFOTOGRAFERET, MEN ER ENDNU
IKKE BLEVET DECHIFRERET.

  0 0 5 5 5 5
  0 0 3 0 0 0
  0 3 0 2 0 0
  3 0 0 2 0 4
  0 0 1 1 0 4
  6 6 6 6 0 4

DU KAN GØRE BRUG AF KORTET, HVIS DU KAN
DECHIFRERE DET, MEN SØRG FOR, AT DEN
METODE DU BRUGER HOLDES HEMMELIG.

LAD SPILLET BEGYNDE!!

DIT SKUD ?

```

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                SLAGSKIB"
130 PRINT "-----"
140 REM
150 REM Programmet er skrevet af Ray Westergard
160 REM Copyright 1971 The University of California
170 REM Fremstillet p8 The Lawrence Hall of Science
180 REM Berkely.
190 DIM F(6,6),H(6,6),A(4),B(4),C(6),L(3)
200 FOR X = 1 TO 6
210 FOR Y = 1 TO 6
220 F(X,Y) = 0
230 NEXT Y
240 NEXT X
250 FOR I = 1 TO 3
260 N = 4 - I
270 FOR J = 1 TO 2
280 A = INT (6 * RND (1) + 1)
290 B = INT (6 * RND (1) + 1)
300 D = INT (4 * RND (1) + 1)
310 IF F(A,B) > 0 THEN 280
320 M = 0
330 ON D GOTO 340,560,820,1040
340 B(1) = B
350 B(2) = 7
360 B(3) = 7
370 FOR K = 1 TO N
380 IF M > 1 THEN 440
390 IF B(K) = 6 THEN 430
400 IF F(A,B(K) + 1) > 0 THEN 430
410 B(K + 1) = B(K) + 1
420 GOTO 500
430 M = 2
440 IF B(1) < B(2) AND B(1) < B(3) THEN Z = B(1)
450 IF B(2) < B(1) AND B(2) < B(3) THEN Z = B(2)
460 IF B(3) < B(1) AND B(3) < B(2) THEN Z = B(3)
470 IF Z = 1 THEN 280
480 IF F(A,Z - 1) > 0 THEN 280
490 B(K + 1) = Z - 1
500 NEXT K
510 F(A,B) = 9 - 2 * I - J
520 FOR K = 1 TO N
530 F(A,B(K + 1)) = F(A,B)
540 NEXT K
550 GOTO 1330
560 A(1) = A
570 B(1) = B
580 A(2) = 0:A(3) = 0
590 B(2) = 0:B(3) = 0
600 FOR K = 1 TO N
610 IF M > 1 THEN 690
620 IF A(K) = 1 OR B(K) = 1 THEN 680
630 IF F(A(K) - 1,B(K) - 1) > 0 THEN 680
640 IF F(A(K) - 1,B(K)) > 0 AND F(A(K) - 1,B(K)) =
F(A(K),B(K) - 1) THEN 680
650 A(K + 1) = A(K) - 1
660 B(K + 1) = B(K) - 1
670 GOTO 800
680 M = 2
690 IF A(1) > A(2) AND A(1) > A(3) THEN Z1 = A(1)
700 IF A(2) > A(1) AND A(2) > A(3) THEN Z1 = A(2)
710 IF A(3) > A(1) AND A(3) > A(2) THEN Z1 = A(3)
720 IF B(1) > B(2) AND B(1) > B(3) THEN Z2 = B(1)
730 IF B(2) > B(1) AND B(2) > B(3) THEN Z2 = B(2)
740 IF B(3) > B(1) AND B(3) > B(2) THEN Z2 = B(3)
750 IF Z1 = 6 OR Z2 = 6 THEN 280
760 IF F(Z1 + 1,Z2 + 1) > 0 THEN 280
770 IF F(Z1,Z2 + 1) > 0 AND F(Z1,Z2 + 1) = F(Z1 +
1,Z2) THEN 280
780 A(K + 1) = Z1 + 1
790 B(K + 1) = Z2 + 1
800 NEXT K
810 GOTO 1290
820 A(1) = A
830 A(2) = 7
840 A(3) = 7
850 FOR K = 1 TO N
860 IF M > 1 THEN 920
870 IF A(K) = 6 THEN 910
880 IF F(A(K) + 1,B) > 0 THEN 910
890 A(K + 1) = A(K) + 1
900 GOTO 980
910 M = 2
920 IF A(1) < A(2) AND A(1) < A(3) THEN Z = A(1)
930 IF A(2) < A(1) AND A(2) < A(3) THEN Z = A(2)
940 IF A(3) < A(1) AND A(3) < A(2) THEN Z = A(3)
950 IF Z = 1 THEN 280

```

```

960 IF F(Z - 1,B) > 0 THEN 280
970 A(K + 1) = Z - 1
980 NEXT K
990 F(A,B) = 9 - 2 * I - J
1000 FOR K = 1 TO N
1010 F(A(K + 1),B) = F(A,B)
1020 NEXT K
1030 GOTO 1330
1040 A(1) = A
1050 B(1) = B
1060 A(2) = 7:A(3) = 7
1070 B(2) = 0:B(3) = 0
1080 FOR K = 1 TO N
1090 IF M > 1 THEN 1170
1100 IF A(K) = 6 OR B(K) = 1 THEN 1160
1110 IF F(A(K) + 1,B(K) - 1) > 0 THEN 1160
1120 IF F(A(K) + 1,B(K)) > 0 AND F(A(K) + 1,B(K)) =
F(A(K),B(K) - 1) THEN 1160
1130 A(K + 1) = A(K) + 1
1140 B(K + 1) = B(K) - 1
1150 GOTO 1280
1160 M = 2
1170 IF A(1) < A(2) AND A(1) < A(3) THEN Z1 = A(1)
1180 IF A(2) < A(1) AND A(2) < A(3) THEN Z1 = A(2)
1190 IF A(3) < A(1) AND A(3) < A(2) THEN Z1 = A(3)
1200 IF B(1) > B(2) AND B(1) > B(3) THEN Z2 = B(1)
1210 IF B(2) > B(1) AND B(2) > B(3) THEN Z2 = B(2)
1220 IF B(3) > B(1) AND B(3) > B(2) THEN Z2 = B(3)
1230 IF Z1 = 1 OR Z2 = 6 THEN 280
1240 IF F(Z1 - 1,Z2 + 1) > 0 THEN 280
1250 IF F(Z1,Z2 + 1) > 0 AND F(Z1,Z2 + 1) = F(Z1 -
1,Z2) THEN 280
1260 A(K + 1) = Z1 - 1
1270 B(K + 1) = Z2 + 1
1280 NEXT K
1290 F(A,B) = 9 - 2 * I - J
1300 FOR K = 1 TO N
1310 F(A(K + 1),B(K + 1)) = F(A,B)
1320 NEXT K
1330 NEXT J
1340 NEXT I
1350 PRINT
1360 PRINT "FØLGENDE KORT OVER FJENDENS FLÅDE ER"
1370 PRINT "BLEVET AFFOTOGRAFERET, MEN ER ENDNU"
1380 PRINT "IKKE BLEVET DECHIFRERET."
1390 PRINT
1400 FOR I = 1 TO 6
1410 FOR J = 1 TO 6
1420 H(I,J) = F(J,I)
1430 NEXT J
1440 NEXT I
1450 FOR I = 1 TO 6
1460 FOR J = 1 TO 6
1470 PRINT H(I,J);
1480 NEXT J
1490 PRINT
1500 NEXT I
1510 PRINT
1520 PRINT "DU KAN GØRE BRUG AF KORTET, HVIS DU KAN"
1530 PRINT "DECHIFRERE DET, MEN SØRG FOR, AT DEN"
1540 PRINT "METODE DU BRUGER HOLDES HEMMELIG."
1550 PRINT
1560 FOR I = 1 TO 6
1570 FOR J = 1 TO 6
1580 H(I,J) = 0
1590 NEXT J
1600 NEXT I
1610 FOR I = 1 TO 3
1620 L(I) = 0
1630 NEXT I
1640 C(1) = 2:C(2) = 2
1650 C(3) = 1:C(4) = 1
1660 C(5) = 0:C(6) = 0
1670 S = 0:H = 0
1680 PRINT "LAD SPILLET BEGYNDE!!"
1690 PRINT
1700 PRINT "DIT SKUD ";
1710 INPUT X,Y
1720 IF X < 1 OR X > 6 OR INT(X) < > ABS(X)
THEN 1740
1730 IF Y > 0 AND Y < 7 AND INT(Y) = ABS(Y) THEN
1760
1740 PRINT "UGYLDIGT INDFAST. PRØV IGEN"
1750 GOTO 1690
1760 R = 7 - Y
1770 C = X
1780 IF F(R,C) > 0 THEN 1820
1790 S = S + 1
1800 PRINT "PLASK!!!! DU RAMTE VANDET!!! PRØV IGEN."
1810 GOTO 1690
1820 IF C(F(R,C)) < 4 THEN 1880
1830 PRINT "DER HAR VÆRET ET SKIB HER, MEN DU HAR"
1840 PRINT "ALLEREDE SÆNKET DET.....FORBIER!"
1850 PRINT "PLASK!!!.....PRØV IGEN!"
1860 S = S + 1
1870 GOTO 1690
1880 IF H(R,C) > 0 THEN 1960
1890 H = H + 1
1900 H(R,C) = F(R,C)
1910 PRINT "FULDTRÆFFER I SKIB NUMMER:";F(R,C)
1920 C(F(R,C)) = C(F(R,C)) + 1
1930 IF C(F(R,C)) > = 4 THEN 2000
1940 PRINT ".....PRØV IGEN."
1950 GOTO 1690
1960 PRINT "DU HAR ALLEREDE RAMT SKIB NUMMER";F(R,C)
1970 PRINT "PR DENNE POSITION!! PLASK!!! PRØV IGEN!"
1980 S = S + 1
1990 GOTO 1690
2000 L((INT(F(R,C) - 1) / 2) + 1) = L((INT(F(R,C)
- 1) / 2) + 1) + 1
2010 PRINT "OG DU HAR SÆNKET DET! HURRA!!! HURRA!!!"
2020 PRINT
2030 PRINT "INDTIL NU HAR FJENDEN MISTET:"
2040 PRINT
2050 PRINT L(1);"DESTROYERE"
2060 PRINT L(2);"KRYDSERE"
2070 PRINT L(3);"HANGARSKIBE"
2080 PRINT
2090 PRINT "DIT FORBIER/TRÆFFER FORHOLD ER"; INT((S
* 100) / H) / 100
2100 IF (L(1) + L(2) + L(3)) < 6 THEN 1690
2110 PRINT
2120 PRINT "DU HAR UDSLETTET FJENDES FLÅDE TOTALT!"
2130 PRINT
2140 PRINT "SLUT FORBIER/TRÆFFER FORHOLD ER"; INT
((S * 100) / H) / 100
2150 IF S / H > 0 THEN 2170
2160 PRINT "TILLYKKE - DU HAR RAMT PLET HVER GANG!"
2170 PRINT
2180 PRINT
2190 PRINT "*****"
2200 PRINT
2210 GOTO 200
2220 END

```



# Slalom

Dette spil simulerer turen ned ad en slalomløjpe, der indeholder fra 1 til 25 porte. Spilleren vælger selv, hvor mange porte der skal være og har en vis kontrol over hastigheden ned ad løjpen.

Selv om du ikke er vant til at køre slalom, så har du hermed en enestående lejlighed til at prøve det uden fare for at komme til skade. Hvis du er vant til at stå på ski, så kan du nyde spillet, mens dit ben er i gips.

Slalom er skrevet af J. Panek, da han studerede ved Dartmouth College.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                        SLALOM
-----
```

HVOR MANGE PORTE HAR DENNE BANE (1 TIL 25) ?4

TAST 'INS' HVIS DU ØNSKER INSTRUKTIONER.  
TAST 'MAX' HVIS DU UD SKRIFT AF MAKSIMALHASTIGHEDER.  
TAST 'RUN' HVIS DU VIL STARTE LØBET.

HVAD ØNSKES ?INS

VI ER MIDT I STORSLALOM VED VINTEROLYMPIADEN 1984  
KUN DET DANSKE HOLD HAR EN CHANCE FOR VINDE GULD.

INDTAST

```
0 -- HVIS DU VIL VIDE HVOR LÆNGE DU HAR VÆRET OM DET
1 -- HVIS DU VIL ØGE FARTEN MEGET
2 -- HVIS DU VIL ØGE FARTEN NOGET
3 -- HVIS DU VIL ØGE FARTEN LIDT
4 -- HVIS DU VIL FORTSÆTTE MED SAMME HASTIGHED
5 -- HVIS DU VIL BREMSE LIDT
6 -- HVIS DU VIL BREMSE NOGET
7 -- HVIS DU VIL BREMSE MEGET
8 -- HVIS DU VIL SNYDE OG SPRINGE PORTEN OVER
```

BRUG DISSE ALTERNATIVER NÅR DATAMATEN SPØRGER:

VÆLG ALTERNATIV ?

HELD OG LYKKE

HVAD ØNSKES ?MAX

MAKSIMALHASTIGHED GENNEM PORTENE ER:

PORT NR.	KM/T
1	14
2	18
3	26
4	29

HVAD ØNSKES ?RUN

HVORDAN ER DU SOM SKILØBER, (1=DÅRLIG, 3=BEDST) ?3

STARTEREN TÆLLER: 5...4...3...2...1...START!

HER KOMMER PORT 1  
HASTIGHED: 15 KM/T.  
VÆLG ALTERNATIV ?3  
HASTIGHED: 27.2 KM/T.

HER KOMMER PORT 2  
HASTIGHED: 16 KM/T.  
VÆLG ALTERNATIV ?0  
DET TOG .32 SEKUNDER.  
VÆLG ALTERNATIV ?1  
HASTIGHED: 39.1 KM/T.

HER KOMMER PORT 3  
HASTIGHED: 23 KM/T.  
VÆLG ALTERNATIV ?8  
!!!! SNYD  
DU KLAREDE DET!!

HER KOMMER PORT 4  
HASTIGHED: 23 KM/T.  
VÆLG ALTERNATIV ?8  
!!!! SNYD  
DET BLEV OPDAGET!  
DET TOG .2 SEKUNDER.

VIL DU PRØVE EN TUR TIL ?N  
TAK FOR SPILLET.

```
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                        SLALOM"
130 PRINT "-----"
140 PRINT
150 PRINT "HVOR MANGE PORTE HAR DENNE BANE (1 TIL 25) ";
160 INPUT V
170 PRINT
180 IF V > 25 THEN 210
190 IF V < 1 THEN 250
200 GOTO 1440
210 PRINT "MAKSIMALE ANTAL PORTE ER 25."
220 PRINT
230 V = 25
240 GOTO 1440
250 PRINT "PRØV IGEN,"
260 PRINT
270 GOTO 150
280 PRINT "HVORDAN ER DU SOM SKILØBER, (1=DÅRLIG, 3=BEDST) ";
290 INPUT A
300 PRINT
310 IF A < 1 THEN 340
320 IF A > 3 THEN 340
330 GOTO 370
340 PRINT "SKAL LIGGE MELLEML 1 OG 3."
350 PRINT
360 GOTO 280
370 PRINT "STARTEREN TÆLLER: 5...4...3...2...1...START!"
380 T = 0
390 S = INT ( RND (1) * (18 - 9) + 9)
400 PRINT
410 FOR O = 1 TO V
420 READ Q
430 PRINT
440 PRINT "HER KOMMER PORT";O
450 PRINT "      HASTIGHED:";S;"KM/T."
460 S1 = S
470 PRINT "VÆLG ALTERNATIV ";
480 INPUT O1
490 IF O1 = 0 THEN 920
500 IF O1 > 8 THEN 1400
510 IF O1 < 1 THEN 1400
520 GOSUB 940
530 IF S < 7 THEN 1360
540 T = T + (Q - S + 1)
550 IF S > Q THEN 1690
560 NEXT O
570 PRINT "DET TOG"; INT (T + RND (1) * 100) / 100;"SEKUNDER."
580 M = T
590 M = M / V
600 IF M < 1.5 - (A * 0.1) THEN 1710
610 IF M < 2.9 - (A * 0.1) THEN 1740
620 IF M < 4.4 - (A * 0.1) THEN 1770
630 PRINT
640 PRINT "VIL DU PRØVE EN TUR TIL ";
650 INPUT B$
660 IF LEFT$ (B$,1) = "N" THEN 1800
670 IF LEFT$ (B$,1) = "J" THEN 370
680 PRINT "TAST VENLIGST ENTEN 'JA' ELLER 'NEJ'"
690 GOTO 630
700 GOTO 1920
710 PRINT
720 PRINT "      VI ER MIDT I STORSLALOM VED VINTEROLYMPIADEN 1984"
730 PRINT "      KUN DET DANSKE HOLD HAR EN CHANCE FOR VINDE GULD."
740 PRINT
750 PRINT "INDTAST"
760 PRINT " 0 -- HVIS DU VIL VIDE HVOR LÆNGE DU HAR VÆRET OM DET"
770 PRINT " 1 -- HVIS DU VIL ØGE FARTEN MEGET"
780 PRINT " 2 -- HVIS DU VIL ØGE FARTEN NOGET"
790 PRINT " 3 -- HVIS DU VIL ØGE FARTEN LIDT"
800 PRINT " 4 -- HVIS DU VIL FORTSÆTTE MED SAMME HASTIGHED"
810 PRINT " 5 -- HVIS DU VIL BREMSE LIDT"
820 PRINT " 6 -- HVIS DU VIL BREMSE NOGET"
830 PRINT " 7 -- HVIS DU VIL BREMSE MEGET"
840 PRINT " 8 -- HVIS DU VIL SNYDE OG SPRINGE PORTEN OVER"
850 PRINT
860 PRINT "BRUG DISSE ALTERNATIVER NÅR DATAMATEN SPØRGER:"
870 PRINT
880 PRINT "VÆLG ALTERNATIV ?"
890 PRINT
900 PRINT "HELD OG LYKKE"
910 GOTO 1470
920 PRINT "DET TOG"; INT (T + RND (1) * 100) / 100;"SEKUNDER."
930 GOTO 470
940 ON O1 GOTO 1080,960,1120,1030,1140,1050,1100,1160
950 GOTO 1920
960 S = S + INT ( RND (1) * 2 + 3)
970 PRINT "      HASTIGHED:";S * 1.7;"KM/T."
```

```

980 IF S > Q THEN 1240
990 IF S > Q - 1 THEN 1010
1000 RETURN
1010 PRINT "DET VAR TÆT PÅ!!"
1020 RETURN
1030 PRINT " HASTIGHED: ";S * 1.7;"KM/T."
1040 GOTO 980
1050 S = S - INT ( RND (1) * 2 + 3)
1060 PRINT " HASTIGHED:";S * 1.7;"KM/T."
1070 GOTO 980
1080 S = S + INT ( RND (1) * 5 + 5)
1090 GOTO 1030
1100 S = S - INT ( RND (1) * 5 + 5)
1110 GOTO 1060
1120 S = S + INT ( RND (1) * 3 + 1)
1130 GOTO 1060
1140 S = S - INT ( RND (1) * 3 + 1)
1150 GOTO 1060
1160 PRINT "!!!! SNYD"
1170 IF RND (1) < .7 THEN 1210
1180 PRINT "DU KLAREDE DET!!"
1190 T = T + 1.5
1200 RETURN
1210 PRINT "DET BLEV OPDAGET!"
1220 PRINT "DET TOG"; INT (T + RND (1) * 100) / 100;"SEKUNDER."
1230 GOTO 630
1240 IF RND (1) < ((S - Q) * .1) + .2 THEN 1280
1250 PRINT "DU VAR HURTIGERE END MAKSIMALHASTIGHEDEN,"
1260 PRINT "OG DU KLAREDE DET!!!"
1270 RETURN
1280 PRINT "DU VAR HURTIGERE END MAKSIMALHASTIGHEDEN,"
1290 PRINT "OG DU ";
1300 IF RND (1) < .5 THEN 1340
1310 PRINT "BLEV SMADRET HELT!"
1320 PRINT "DET TOG"; INT (T + RND (1) * 100) / 100;"SEKUNDER."
1330 GOTO 630
1340 PRINT "VÆLTEDE ET FLAG."
1350 GOTO 1320
1360 PRINT "LAD OS NU VÆRE REALISTISKE, IKKE???"
1370 PRINT "LAD OS GRÅ TILBAGE OG PRØVE IGEN"
1380 S = S1
1390 GOTO 430
1400 PRINT
1410 PRINT "HVAFFOR NO'ET?"
1420 PRINT
1430 GOTO 470
1440 PRINT "TAST 'INS' HVIS DU ØNSKER INSTRUKTIONER."
1450 PRINT "TAST 'MAX' HVIS DU UD SKRIFT AF MAKSIMALHASTIGHEDER."
1460 PRINT "TAST 'RUN' HVIS DU VIL STARTE LØBET."
1470 PRINT
1480 PRINT "HVAD ØNSKES ";
1490 INPUT A$
1500 PRINT
1510 REM
1520 IF A$ = "INS" THEN 710
1530 IF A$ = "MAX" THEN 1580
1540 IF A$ = "RUN" THEN 280
1550 PRINT "";"A$;" ER ET ULOVLIGT IND TAST -- PRØV IGEN."
1560 PRINT
1570 GOTO 1490
1580 PRINT
1590 PRINT "MAKSIMALHASTIGHED Gennem PORTENE ER:
1600 PRINT
1610 PRINT "PORT NR.      KM/T"
1620 PRINT "-----"
1630 RESTORE
1640 FOR B = 1 TO V
1650 READ Q
1660 PRINT B;"      ";Q
1670 NEXT B
1680 GOTO 1470
1690 T = T + .5
1700 GOTO 560
1710 PRINT "DU VANDT GULD-MEDALJE!"
1720 G(1) = G(1) + 1
1730 GOTO 1790
1740 PRINT "DU VANDT SØLV-MEDALJE."
1750 S(1) = S(1) + 1
1760 GOTO 1790
1770 PRINT "DU VANDT BRONZE."
1780 B(1) = B(1) + 1
1790 GOTO 630
1800 PRINT
1810 PRINT "TAK FOR SPILLET."
1820 PRINT
1830 IF G(1) < 1 THEN 1850
1840 PRINT " GULD MEDALJER:";G(1)
1850 IF S(1) < 1 THEN 1870
1860 PRINT " SØLV MEDALJER:";S(1)
1870 IF B(1) < 1 THEN 1920
1880 PRINT "BRONZE MEDALJER:";B(1)
1890 DATA 14,18,26,29,18,25,28,32,29,20
1900 DATA 29,29,25,21,26,29,20,21,20,18
1910 DATA 26,25,33,31,22
1920 END

```



# Smugorm

I dette spil drejer det sig om at finde de fire smugorm, der har gemt sig i felterne på et net, der måler 10 gange 10 felter. Hjemmebasen (nederste venstre hjørne) har koordinaterne (0,0). Alle gæt skal angives som to heltal (0 til 9) adskilt af et komma. Det første ciffer er antallet af felter til højre for hjemmebasen, mens det andet ciffer er antallet af felter over denne.

Du har ti gæt til at finde smugormene, og efter hvert gæt fortæller datamaten dig, hvor langt du er fra ormen. Hvis du spiller spillet ved hjælp af et ternet stykke papir og en lineal, så bør du kunne finde alle fire smugorme ved at anvende trekants-beregninger, sådan som det gøres ved radio-pejling.

Hvis du vil gøre spillet sværere, kan du udskrive afstandene til smugormen enten afrundet eller forkortet til nærmeste underliggende heltal. Programlinie 480 skal i så fald indeholde funktionen  $\text{INT}(D+.5)$  og  $\text{INT}(D)$ .

Dette program er modificeret noget af Bob Albrecht fra People's Computer Company. Det blev oprindeligt skrevet af Bud Valenti fra SOLO-projektet i Pittsburgh, Pennsylvania.



MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

SMUGORM

I DETTE SPIL SKAL DU FINDE 4 SMUGORM, DER GEMMER SIG PÅ ET NET DER ER 10 GANGE 10 FELTER STORT. BASEN ER I POSITION '0,0'.

HVER AF DINE GÆT SKAL BESTÅ AF TO TAL, DER BEGGE SKAL LIGGE MELLEM 0 OG 9.

DET FØRSTE TAL ANGIVER AFSTANDEN TIL HØJRE FOR BASEN, MENS DET ANDET TAL VISER AFSTANDEN OVER BASEN.

DU HAR 10 FØRSØG. EFTER HVERT FØRSØG VIL JEG FORTÆLLE DIG HVOR LANGT DU ER FRA SMUGORMEN.

OMGANG 1 DIT GÆT ?5,5

DU BEFINDER DIG 51  
 ENHEDER FRA SMUGORM 1  
 DU BEFINDER DIG 86.6  
 ENHEDER FRA SMUGORM 2  
 DU BEFINDER DIG 76  
 ENHEDER FRA SMUGORM 3  
 DU BEFINDER DIG 38  
 ENHEDER FRA SMUGORM 4

OMGANG 2 DIT GÆT ?3,5

DU BEFINDER DIG 17  
 ENHEDER FRA SMUGORM 1  
 DU BEFINDER DIG 99.1  
 ENHEDER FRA SMUGORM 2  
 DU BEFINDER DIG 68  
 ENHEDER FRA SMUGORM 3  
 DU BEFINDER DIG 17  
 ENHEDER FRA SMUGORM 4

OMGANG 3 DIT GÆT ?2,5

DU HAR FUNDET SMUGORM NUMMER 1

DU BEFINDER DIG 108.8  
 ENHEDER FRA SMUGORM 2  
 DU BEFINDER DIG 70  
 ENHEDER FRA SMUGORM 3  
 DU BEFINDER DIG 24  
 ENHEDER FRA SMUGORM 4

OMGANG 4 DIT GÆT ?2,9

DU BEFINDER DIG 167.4  
 ENHEDER FRA SMUGORM 2  
 DU BEFINDER DIG 17  
 ENHEDER FRA SMUGORM 3  
 DU BEFINDER DIG 86.6  
 ENHEDER FRA SMUGORM 4

OMGANG 5 DIT GÆT ?3,9

DU BEFINDER DIG 161.2  
 ENHEDER FRA SMUGORM 2  
 DU HAR FUNDET SMUGORM NUMMER 3

DU BEFINDER DIG 85  
 ENHEDER FRA SMUGORM 4

OMGANG 6 DIT GÆT ?6,3

DU BEFINDER DIG 51  
 ENHEDER FRA SMUGORM 2  
 DU BEFINDER DIG 53.7  
 ENHEDER FRA SMUGORM 4

OMGANG 7 DIT GÆT ?6,0

DU HAR FUNDET SMUGORM NUMMER 2

DU BEFINDER DIG 85  
 ENHEDER FRA SMUGORM 4

OMGANG 8 DIT GÆT ?3,5

DU BEFINDER DIG 17  
 ENHEDER FRA SMUGORM 4

OMGANG 9 DIT GÆT ?3,4

DU HAR FUNDET SMUGORM NUMMER 4

DU FANDT DEM ALLE PÅ 9 FØRSØG.

DET VAR ET MORSOMT SPIL.

LAD OS SPILLE IGEN.....

NU GEMMER DE FIRE SMUGORM SIG.

OMGANG 1 DIT GÆT ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          SMUGORM"
130 PRINT "-----"
140 PRINT
150 REM Med tilladelse fra:
160 REM PEOPLE'S COMPUTER COMPANY
170 DIM P(4,2)
180 PRINT "I DETTE SPIL SKAL DU FINDE 4 SMUGORM,"
190 PRINT "DER GEMMER SIG PÅ ET NET DER ER 10 GAN-"
200 PRINT "GE 10 FELTER STORT. BASEN ER I POSITION"
210 PRINT "'0,0'."
220 PRINT
230 PRINT "HVER AF DINE GÅT SKAL BESTÅ AF TO TAL,"
240 PRINT "DER BEGGE SKAL LIGGE MELLEM 0 OG 9."
250 PRINT
260 PRINT "DET FØRSTE TAL ANGIVER AFSTANDEN TIL"
270 PRINT "HØJRE FOR BASEN, MENS DET ANDET TAL VI-"
280 PRINT "SER AFSTANDEN OVER BASEN."
290 PRINT
300 PRINT "DU HAR 10 FØRSØG."
310 PRINT "EFTER HVERT FØRSØG VIL JEG FORTÆLLE DIG"
320 PRINT "HVOR LANGT DU ER FRA SMUGORMEN."
330 PRINT
340 GOSUB 770
350 T = 0
360 T = T + 1
370 PRINT
380 PRINT "OMGANG";T;"DIT GÅT ";
390 INPUT M,N
400 PRINT
410 FOR I = 1 TO 4
420 IF P(I,1) = - 1 THEN GOTO 520
430 IF P(I,1) < > M THEN GOTO 490
440 IF P(I,2) < > N THEN GOTO 490
450 P(I,1) = - 1
460 PRINT "DU HAR FUNDET SMUGORM NUMMER";I
470 PRINT
480 GOTO 520
490 D = SQR ((P(I,1) - M) ^ 2 + (P(I,2) - N) ^ 2)
500 PRINT "DU BEFINDER DIG";(INT (D * 170)) / 10;
510 PRINT "ENHEDER FRA SMUGORM";I
520 NEXT I
530 FOR J = 1 TO 4
540 IF P(J,1) < > - 1 THEN GOTO 590
550 NEXT J
560 PRINT
570 PRINT "DU FANDT DEM ALLE PÅ";T;"FØRSØG."
580 GOTO 690
590 IF T < 10 THEN GOTO 360
600 PRINT
610 PRINT "DESVÆRRE, DET VAR DINE 10 FØRSØG."
620 PRINT "NU SKAL JEG FORTÆLLE HVOR DE ER."
630 PRINT
640 FOR I = 1 TO 4
650 IF P(I,1) = - 1 THEN GOTO 680
660 PRINT "SMUGORM";I;"ER PÅ (";
670 PRINT P(I,1);",";P(I,2);")"
680 NEXT I
690 PRINT
700 PRINT "DET VAR ET MORSOMT SPIL."
710 PRINT
720 PRINT "LAD OS SPILLE IGEN....."
730 PRINT
740 PRINT "NU GEMMER DE FIRE SMUGORM SIG."
750 PRINT
760 GOTO 340
770 FOR J = 1 TO 2
780 FOR I = 1 TO 4
790 P(I,J) = INT (10 * RND (1))
800 NEXT I
810 NEXT J
820 RETURN
830 END

```

# Solitaire

Dette er en datamatudgave af det gamle europæiske spil Solitaire, et logik-spil for én person. Spillebrættet består af huller, der danner et kors. I alle huller – undtagen det midterste – sidder der pinde. Det gælder om at fjerne alle 32 pinde – eller så mange som muligt – ved at hoppe over en anden pind og ned i et tomt hul. Den pind, der blev hoppet over, fjernes fra brættet.

Der findes mange forskellige strategier for at vinde dette spil, og hver strategi kan selvfølgelig spilles på otte forskellige måder. Kan du finde på en metode, så man altid vinder?

Charles Lund skrev programmet, mens han gik på den amerikanske skole i Haag, Holland.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

SOLITAIRE

HER ER SPILLEBRÆTTET:

```

      !   !   !
      13  14  15

      !   !   !
      22  23  24

!   !   !   !   !   !   !
29  30  31  32  33  34  35

!   !   !   !   !   !   !
38  39  40  41  42  43  44

!   !   !   !   !   !   !
47  48  49  50  51  52  53

      !   !   !
      58  59  60

      !   !   !
      67  68  69
    
```

FOR AT SPARE TID, VIL DER BLIVE ANVENDT ET FORENKLET SPILLEBRÆT UNDER SPILLET. DU KAN BRUGE OVENSTÅENDE UDSKRIFT SOM REFERENCE, NÅR DU SKAL INDTASTE NUMMERET PÅ DEN PIND DU ØNSKER AT FLYTTE. OG LAD OS SR FR BEGYNDT.....

```

      ! ! !
      ! ! !
! ! ! ! ! ! !
! ! ! 0 ! ! !
! ! ! ! ! ! !
      ! ! !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?59  
HVORTIL ?41

```

      ! ! !
      ! ! !
! ! ! ! ! ! !
! ! ! ! ! ! !
! ! ! 0 ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?32  
HVORTIL ?50

```

      ! ! !
      ! ! !
! ! ! 0 ! ! !
! ! ! 0 ! ! !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?43  
HVORTIL ?41

```

      ! ! !
      ! ! !
! ! ! 0 ! ! !
! ! ! ! 0 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?34  
HVORTIL ?32

```

      ! ! !
      ! ! !
! ! ! ! 0 0 !
! ! ! ! 0 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?31  
HVORTIL ?33

```

      ! ! !
      ! ! !
! ! 0 0 ! 0 !
! ! ! ! 0 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?40  
HVORTIL ?42

```

      ! ! !
      ! ! !
! ! 0 0 ! 0 !
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?14  
HVORTIL ?32

```

      ! 0 !
      ! 0 !
! ! 0 ! ! 0 !
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?13  
HVORTIL ?31

```

      0 0 !
      0 0 !
! ! ! ! ! 0 !
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?32  
HVORTIL ?34

```

      0 0 !
      0 0 !
! ! ! 0 0 ! !
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?35  
HVORTIL ?33

```

      0 0 !
      0 0 !
! ! ! 0 ! 0 0
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?30  
HVORTIL ?32

```

      0 0 !
      0 0 !
! 0 0 ! ! 0 0
! ! 0 0 ! 0 !
! ! ! ! ! ! !
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?53  
HVORTIL ?35

```

      0 0 !
      0 0 !
! 0 0 ! ! 0 !
! ! 0 0 ! 0 0
! ! ! ! ! ! 0
      ! 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?58  
HVORTIL ?40

```

      0 0 !
      0 0 !
! 0 0 ! ! 0 !
! ! ! 0 ! 0 0
! ! 0 ! ! ! 0
      0 0 !
      ! ! !
    
```

HVILKEN PIND SKAL FLYTTES ?47  
HVORTIL ?49

```
  0 0 !
  0 0 !
! 0 0 ! ! 0 !
! ! ! 0 ! 0 0
0 0 ! ! ! ! 0
  0 0 !
  ! ! !
```

HVILKEN PIND SKAL FLYTTES ?29  
HVORTIL ?31

```
  0 0 !
  0 0 !
0 0 ! 0 ! 0 !
0 0 0 0 ! 0 0
0 0 0 ! ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?49  
HVORTIL ?51

```
  0 0 0
  0 0 0
0 0 ! 0 0 0 !
0 0 0 0 0 0 0
0 0 0 0 ! 0 0
  0 0 0
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?40  
HVORTIL ?58

```
  0 0 !
  0 0 !
! 0 0 ! ! 0 !
! ! 0 0 ! 0 0
0 0 0 ! ! ! 0
  ! 0 !
  ! ! !
```

HVILKEN PIND SKAL FLYTTES ?51  
HVORTIL ?49

```
  0 0 !
  0 0 !
0 0 ! 0 ! 0 !
0 0 0 0 ! 0 0
0 0 ! 0 0 ! 0
  0 0 !
  ! 0 0
```

SPILLET ER FORBI.

DER VAR 4 PINDE TILBAGE.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

FARVEL FOR NU.....

HVILKEN PIND SKAL FLYTTES ?67  
HVORTIL ?49

```
  0 0 !
  0 0 !
! 0 0 ! ! 0 !
! ! 0 0 ! 0 0
0 0 ! ! ! ! 0
  0 0 !
  0 ! !
```

HVILKEN PIND SKAL FLYTTES ?33  
HVORTIL ?51

```
  0 0 !
  0 0 !
0 0 ! 0 0 0 !
0 0 0 0 0 0 0
0 0 ! 0 ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?69  
HVORTIL ?67

```
  0 0 !
  0 0 !
! 0 0 ! ! 0 !
! ! 0 0 ! 0 0
0 0 ! ! ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?15  
HVORTIL ?33

```
  0 0 0
  0 0 0
0 0 ! 0 ! 0 !
0 0 0 0 0 0 0
0 0 ! 0 ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?38  
HVORTIL ?40

```
  0 0 !
  0 0 !
! 0 0 ! ! 0 !
0 0 ! 0 ! 0 0
0 0 ! ! ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?60  
HVORTIL ?42

```
  0 0 0
  0 0 0
0 0 ! 0 ! 0 !
0 0 0 0 ! 0 0
0 0 ! 0 0 ! 0
  0 0 0
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?49  
HVORTIL ?31

```
  0 0 !
  0 0 !
! 0 ! ! ! 0 !
0 0 0 0 ! 0 0
0 0 0 ! ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?33  
HVORTIL ?51

```
  0 0 0
  0 0 0
0 0 ! 0 0 0 !
0 0 0 0 0 0 0
0 0 ! 0 ! ! 0
  0 0 0
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?32  
HVORTIL ?30

```
  0 0 !
  0 0 !
! ! 0 0 ! 0 !
0 0 0 0 ! 0 0
0 0 0 ! ! ! 0
  0 0 !
  ! 0 0
```

HVILKEN PIND SKAL FLYTTES ?52  
HVORTIL ?50

```
  0 0 0
  0 0 0
0 0 ! 0 0 0 !
0 0 0 0 0 0 0
0 0 ! ! 0 0 0
  0 0 0
  ! 0 0
```

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          SOLITAIRE"
130 PRINT "-----"
140 PRINT
150 DATA 13, 14, 15, 22, 23, 24, 29, 30, 31, 32, 33,
34, 35, 38, 39, 40, 41
160 DATA 42, 43, 44, 47, 48, 49, 50, 51, 52, 53, 58,
59, 60, 67, 68, 69
170 DIM B(70), T(9,9)
180 PRINT
190 PRINT "HER ER SPILLEBRÆTTET:"
200 PRINT
210 PRINT
220 PRINT "      !  !  !"
230 PRINT "      13 14 15"
240 PRINT
250 PRINT "      !  !  !"
260 PRINT "      22 23 24"
270 PRINT
280 PRINT " !  !  !  !  !  !  !"
290 PRINT " 29 30 31 32 33 34 35"
300 PRINT
310 PRINT " !  !  !  !  !  !  !"
320 PRINT " 38 39 40 41 +42 43 44"
330 PRINT
340 PRINT " !  !  !  !  !  !  !"
350 PRINT " 47 48 49 50 51 52 53"
360 PRINT
370 PRINT "      !  !  !"
380 PRINT "      58 59 60"
390 PRINT
400 PRINT "      !  !  !"
410 PRINT "      67 68 69"
420 PRINT
430 PRINT
440 PRINT "FOR AT SPARE TID, VIL DER BLIVE ANVENDT"
450 PRINT "ET FORENKLET SPILLEBRÆT UNDER SPILLET."
460 PRINT "DU KAN BRUGE OVENSTÅENDE UDSKRIFT SOM"
470 PRINT "REFERENCE, NÅR DU SKAL INDTASTE NUMME-"
480 PRINT "RET PÅ DEN PIND DU ØNSKER AT FLYTTE."
490 PRINT "OG LAD OS SÅ FÅ BEGYNDT....."
500 FOR R = 1 TO 9
510 FOR C = 1 TO 9
520 IF (R - 4) * (R - 5) * (R - 6) = 0 THEN 560
530 IF (C - 4) * (C - 5) * (C - 6) = 0 THEN 560
540 T(R,C) = - 5
550 GOTO 580
560 IF (R - 1) * (C - 1) * (R - 9) * (C - 9) = 0
THEN 540
570 T(R,C) = 5
580 NEXT C
590 NEXT R
600 T(5,5) = 0
610 GOSUB 860
620 REM Indtast træk og test om OK
630 FOR W = 1 TO 33
640 READ M
650 B(M) = - 7
660 NEXT W
670 B(41) = - 3
680 PRINT
690 PRINT "HVILKEN PIND SKAL FLYTTES ";
700 INPUT Z
710 IF B(Z) = - 7 THEN 740
720 PRINT "ULOVLIGT TRÆK, PRØV IGEN..."
730 GOTO 680
740 PRINT "          HVORTIL ";
750 INPUT P
760 IF B(P) = 0 THEN 720
770 IF B(P) = - 7 THEN 720
780 IF Z = P THEN 680
790 IF ((Z + P) / 2) = INT ((Z + P) / 2) THEN 810
800 GOTO 720
810 IF ( ABS (Z - P) - 2) * ( ABS (Z - P) - 18) < >
0 THEN 720
820 GOSUB 1030
830 GOSUB 860
840 GOSUB 1390
850 GOTO 680
860 REM Udskriv brættet
870 FOR X = 1 TO 9
880 IF X < 4 THEN PRINT "  ";
890 IF X > 6 THEN PRINT "  ";
900 FOR Y = 1 TO 9
910 IF (X - 1) * (X - 9) * (Y - 1) * (Y - 9) = 0
THEN 940
920 IF (X - 4) * (X - 5) * (X - 6) = 0 THEN 950
930 IF (Y - 4) * (Y - 5) * (Y - 6) = 0 THEN 950
940 GOTO 990
950 IF T(X,Y) < > 5 THEN 980
960 PRINT " !";
970 GOTO 990
980 PRINT " 0";
990 NEXT Y
1000 PRINT
1010 NEXT X
1020 RETURN
1030 REM Flyt pinde på brættet
1040 C = 1
1050 FOR X = 1 TO 9
1060 FOR Y = 1 TO 9
1070 IF C < > Z THEN 1350
1080 IF C + 2 < > P THEN 1140
1090 IF T(X,Y + 1) = 0 THEN 720
1100 T(X,Y + 2) = 5
1110 T(X,Y + 1) = 0
1120 B(C + 1) = - 3
1130 GOTO 1310
1140 IF C + 18 < > P THEN 1200
1150 IF T(X + 1,Y) = 0 THEN 720
1160 T(X + 2,Y) = 5
1170 T(X + 1,Y) = 0
1180 B(C + 9) = - 3
1190 GOTO 1310
1200 IF C - 2 < > P THEN 1260
1210 IF T(X,Y - 1) = 0 THEN 720
1220 T(X,Y - 2) = 5
1230 T(X,Y - 1) = 0
1240 B(C - 1) = - 3
1250 GOTO 1310
1260 IF C - 18 < > P THEN 1350
1270 IF T(X - 1,Y) = 0 THEN 720
1280 T(X - 2,Y) = 5
1290 T(X - 1,Y) = 0
1300 B(C - 9) = - 3
1310 B(Z) = - 3
1320 B(P) = - 7
1330 T(X,Y) = 0
1340 GOTO 1380
1350 C = C + 1
1360 NEXT Y
1370 NEXT X
1380 RETURN
1390 REM Undersøg om spillet er forbi
1400 F = 0
1410 FOR R = 2 TO 8
1420 FOR C = 2 TO 8
1430 IF T(R,C) < > 5 THEN 1610
1440 F = F + 1
1450 FOR A = R - 1 TO R + 1
1460 T = 0
1470 FOR B = C - 1 TO C + 1
1480 T = T + T(A,B)
1490 NEXT B
1500 IF T < > 10 THEN 1520
1510 IF T(A,C) < > 0 THEN 1790
1520 NEXT A
1530 FOR X = C - 1 TO C + 1
1540 T = 0
1550 FOR Y = R - 1 TO R + 1
1560 T = T + T(Y,X)
1570 NEXT Y
1580 IF T < > 10 THEN 1600
1590 IF T(R,X) < > 0 THEN 1790
1600 NEXT X
1610 NEXT C
1620 NEXT R
1630 REM Spillet er forbi
1640 PRINT "SPILLET ER FORBI."
1650 PRINT
1660 PRINT "DER VAR";F;"PINDE TILBAGE."
1670 PRINT
1680 IF F < > 1 THEN 1720
1690 PRINT "BRAVO!  DU KLAREDE DET TIL UG!"
1700 PRINT "DENNE UDSKRIFT BEVISER DIN GENIALITET."
1710 PRINT
1720 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1730 INPUT AS
1740 IF LEFT$(AS,1) = "N" THEN 1800
1750 RESTORE
1760 PRINT
1770 GOTO 500
1780 GOTO 1800
1790 RETURN
1800 PRINT
1810 PRINT "FARVEL FOR NU....."
1820 PRINT
1830 END

```

# Splat

Splat simulerer et faldskærmsudspring, hvor du skal forsøge at åbne faldskærmen så sent som muligt uden at ende som splat! Du kan enten selv vælge din sluthastighed, eller du kan lade datamaten gøre det for dig. Du kan selv vælge din acceleration, der er afhængig af tyngdekraften, eller du kan lade datamaten vælge for dig, men i så fald risikerer du at blive tvunget til at foretage dit udspring på en af de otte planeter (til og med Neptun), eller på Månen eller Solen.

Derefter fortæller datamaten dig, fra hvilken højde du springer og beder dig angive, hvor mange sekunders frit fald du ønsker. Derefter deles den tid, som det frie fald varer, i otte perioder, og du får en løbende rapport hele vejen ned. Datamaten holder rede på dine tidligere hop (i den indicerede variabel A) og fortæller dig, hvor godt springet var i forhold til tidligere heldige faldskærmsudspring. Hvis du gerne vil sammenligne dine resultater med tidligere spil, kan du lagre den indicerede variabel A på bånd eller diskette og indlæse den på ny, hver gang du begynder at spille.

John Yegge skrev dette program, mens han gik på Oak Ridge Associated Universities.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

SPLAT

VELKOMMEN TIL 'SPLAT', ET SPIL DER  
SIMULERER ET FALDSKÆRMSUDSPRING. PRØV  
AT ÅBNE FALDSKÆRMEN I SIDSTE ØJEBLIK,  
UDEN AT 'SPLATTE' UD.....

ØNSKER DU AT VÆLGE  
DIN SLUTHASTIGHED ('J' ELLER 'N') ?N

OK. SLUTHASTIGHEDEN ER 272 KM/T.

ØNSKER DU AT VÆLGE ACCELERATION MED  
HENSYN TIL TYNGDEKRAFT (J/N) ?N

SÅ ER DU PÅ PLANETEN NEPTUN.  
ACCELERATIONEN ER: 12.08 M/S/S.

HØJDE = 7644 M  
SLUTHASTIGHED = 398.933 M/S +-5%  
ACCELERATION = 12.08 M/S/S +-5%

STIL URET TIL TIDEN FOR FRIT FALD.

HVOR MANGE SEKUNDER ?20

SÅ HOPPER VI!

TID (S)	HØJDE AT FALDE (M)
0	7644
2.5	7606.27
5	7493.11
7.5	7304.5
10	7040.44
12.5	6700.94
15	6286
17.5	5795.61
20	5229.78

FALDSKÆRMEN ER ÅBEN.

FORBLØFFENDE!!!! SLET IKKE DÆRLIGT  
FOR ET FØRSTE UDSPRINGSFORSØG!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?J

ØNSKER DU AT VÆLGE  
DIN SLUTHASTIGHED ('J' ELLER 'N') ?J

HVILKEN SLUTHASTIGHED (KM/T) ?100

ØNSKER DU AT VÆLGE ACCELERATION MED  
HENSYN TIL TYNGDEKRAFT (J/N) ?N

DU ER PÅ JUPITER.  
ACCELERATIONEN ER: 25.99 M/S/S.

HØJDE = 9043 M  
SLUTHASTIGHED = 146.666 M/S +-5%  
ACCELERATION = 25.989 M/S/S +-5%

STIL URET TIL TIDEN FOR FRIT FALD.

HVOR MANGE SEKUNDER ?200

SÅ HOPPER VI!

TID (S)	HØJDE AT FALDE (M)
0	9043
5.484	----- SLUTHASTIGHED NÆRT
25	5715.47299
50	1977.98632
63.23	SPLAT

FOR S....%\$£&?<>~%\$`£\$, ETC.

JEG VIL GIVE DIG ENDNU EN CHANCE.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

ÅRHR, TAG NU OG PRØV IGEN, IKKE ?N

ZZZZZZZZZZZZZZZZZZZZZZ, SNORK

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                SPLAT"
130 PRINT "-----"
140 PRINT
150 DIM A(42)
160 PRINT "VELKOMMEN TIL 'SPLAT', ET SPIL DER"
170 PRINT "SIMULERER ET FALDSKÆRMSUDSPRING. PRØV"
180 PRINT "AT ÅBNE FALDSKÆRMEN I SIDSTE ØJEBLIK,"
190 PRINT "UDEN AT 'SPLATTE' UD....."
200 PRINT
210 D1 = 0:V = 0:A = 0:N = 0:M = 0:D1 = INT (9001 *
RND (1) + 1000)
220 PRINT "ØNSKER DU AT VÆLGE"
230 PRINT "DIN SLUTHASTIGHED ('J' ELLER 'N') ";
240 INPUT A1$
250 PRINT
260 IF LEFT$ (A1$,1) = "N" THEN GOTO 330
270 IF LEFT$ (A1$,1) < > "J" THEN PRINT "INDTAST
'J' ELLER 'N' ";: GOTO 240
280 PRINT "HVILKEN SLUTHASTIGHED (KM/T) ";
290 INPUT V1
300 V1 = V1 * (5280 / 3600)
310 V = V1 + ((V1 * RND (1)) / 20) - ((V1 * RND
(1)) / 20)
320 GOTO 370
330 V1 = INT (1000 * RND (1))
340 PRINT "OK. SLUTHASTIGHEDEN ER "; INT (V1 *
1000) / 1000;"KM/T."
350 V1 = V1 * (5280 / 3600)
360 V = V1 + ((V1 * RND (1)) / 20) - ((V1 * RND
(1)) / 20)
370 PRINT
380 PRINT "ØNSKER DU AT VÆLGE ACCELERATION MED"
390 PRINT "HENSYN TIL TYNGDEKRAFT (J/N) ";
400 INPUT B1$
410 PRINT
420 IF LEFT$ (B1$,1) = "N" THEN GOTO 480
430 IF LEFT$ (B1$,1) < > "J" THEN PRINT "'JA'
ELLER 'NEJ' ";: INPUT B1$: GOTO 420
440 PRINT "HVILKEN ACCELERATION (M/SEK/SEK) ";
450 INPUT A2
460 A = A2 + ((A2 * RND (1)) / 20) - ((A2 * RND
(1)) / 20)
470 GOTO 980
480 ON INT (1 + (10 * RND (1))) GOTO
490,520,550,580,610,640,670,700,730,760
490 PRINT "OK. DU ER PÅ PLANETEN MERKUR."
500 PRINT "ACCELERATIONEN ER: 3.72 M/S/S"
510 GOTO 780
520 PRINT "FINT. DU ER PÅ PLANETEN VENUS."
530 PRINT "ACCELERATIONEN ER: 8.63 M/S/S."
540 GOTO 800
550 PRINT "SÅ ER DU PÅ JORDEN."
560 PRINT "HER ER ACCELERATIONEN: 9.81 M/S/S."
570 GOTO 820
580 PRINT "UDEMÆRKET, DU ER PÅ MÅNEN."
590 PRINT "ACCELERATIONEN ER: 1.71 M/S/S."
600 GOTO 840
610 PRINT "ALRIGHT! SÅ ER DU PÅ MARS."
620 PRINT "ACCELERATIONEN ER: 3.81 M/S/S."
630 GOTO 860
640 PRINT "DU ER PÅ JUPITER."
650 PRINT "ACCELERATIONEN ER: 25.99 M/S/S."
660 GOTO 880
670 PRINT "GODT...DU ER PÅ PLANETEN SATURN."
680 PRINT "ACCELERATIONEN ER: 11.47 M/S/S."
690 GOTO 900
700 PRINT "DET ER I ORDEN. DU ER PÅ URANUS."
710 PRINT "ACCELERATIONEN ER: 10.31 M/S/S."
720 GOTO 920
730 PRINT "SÅ ER DU PÅ PLANETEN NEPTUN."
740 PRINT "ACCELERATIONEN ER: 12.08 M/S/S."
750 GOTO 940
760 PRINT "DU ER PÅ SOLEN."
770 PRINT "ACCELERATIONEN ER: 273.28 M/S/S."
780 A2 = 3.72
790 GOTO 460
800 A2 = 8.63
810 GOTO 460
820 A2 = 9.81
830 GOTO 460
840 A2 = 1.71
850 GOTO 460
860 A2 = 3.81
870 GOTO 460
880 A2 = 25.99
890 GOTO 460

```



```

900 A2 = 11.47
910 GOTO 460
920 A2 = 10.31
930 GOTO 460
940 A2 = 12.08
950 GOTO 460
960 A2 = 273.28
970 GOTO 460
980 PRINT
990 PRINT "          HØJDE =" ; D1 ; "M"
1000 PRINT "SLUTHASTIGHED =" ; INT (V1 * 1000) /
1000 ; "M/S +-5%"
1010 PRINT "ACCELERATION =" ; INT (A2 * 1000) /
1000 ; "M/S/S +-5%"
1020 PRINT
1030 PRINT "STIL URET TIL TIDEN FOR FRIT FALD."
1040 PRINT
1050 PRINT "HVOR MANGE SEKUNDER " ;
1060 INPUT T
1070 PRINT
1080 PRINT "SR HOPPER VI!"
1090 PRINT
1100 PRINT "TID (S) " ; "HØJDE AT FALDE (M)"
1110 PRINT "===== " ; "===== "
1120 FOR I = 0 TO T STEP (T / 8)
1130 IF I > V / (A) THEN GOTO 1190
1140 D = D1 - ((A / 2) * I ^ 2)
1150 IF D < = 0 THEN GOTO 2250
1160 PRINT INT (I * 100) / 100, INT (D * 100) / 100
1170 NEXT I
1180 GOTO 1260
1190 PRINT ; INT (V / A * 1000) / 1000 ;
1200 PRINT "---- SLUTHASTIGHED NÆT"
1210 FOR I = I TO T STEP (T / 8)
1220 D = D1 - ((V ^ 2 / (2 * A)) + (V * (I - (V /
A))))
1230 IF D < = 0 THEN GOTO 2270
1240 PRINT I, D
1250 NEXT I
1260 PRINT
1270 PRINT "FALDSKÆRMEN ER RBEN."
1280 PRINT
1290 K = 0
1300 K1 = 0
1310 FOR J = 0 TO 42
1320 IF A(J) = 0 THEN GOTO 1380
1330 K = K + 1
1340 IF D > = A(J) THEN GOTO 1360
1350 K1 = K1 + 1
1360 NEXT J
1370 GOTO 1310
1380 A(J) = D
1390 IF J > 2 THEN GOTO 1470
1400 PRINT "FORBLØFFENDE!!!! SLET IKKE DÆRLIGT"
1410 PRINT "FOR ET " ;
1420 IF J = 0 THEN PRINT "FØRSTE " ;
1430 IF J = 1 THEN PRINT "ANDET " ;
1440 IF J = 2 THEN PRINT "TREDIE " ;
1450 PRINT "UDSPRINGSFORSØG!"
1460 GOTO 2310
1470 IF K - K1 < = .1 * K THEN GOTO 1530
1480 IF K - K1 < = .25 * K THEN GOTO 1590
1490 IF K - K1 < = .5 * K THEN GOTO 1650
1500 IF K - K1 < = .75 * K THEN GOTO 1700
1510 IF K - K1 < = .9 * K THEN GOTO 1750
1520 GOTO 1820
1530 PRINT "WAW! DET VAR ALLETIDERS SPRING. AF DE"
1540 PRINT K ; "VELLYKKEDE SPRING, DER VAR FØR"
1550 PRINT "DIT, VAR DET KUN" ; K - K1 ; "DER RB--"
1560 PRINT "NEDE FALDSKÆRMEN SENERE END DU."
1570 PRINT
1580 GOTO 2310
1590 PRINT "UDEMÆRKET! DER VAR" ; K ; "SPRING FØR"
1600 PRINT "DIT, HVORAF" ; K - K1 ; "KOM LAVERE END"
1610 PRINT "DU, FØR DERES FALDSKÆRME FOLDEDE SIG"
1620 PRINT "UD."
1630 PRINT
1640 GOTO 2310
1650 PRINT "SLET IKKE DÆRLIGT. AF DE" ; K
1660 PRINT "SPRING DER GIK FORUD FOR DIT, VAR DET"
1670 PRINT "KUN" ; K - K1 ; "DER VAR BEDRE."
1680 PRINT
1690 GOTO 2310
1700 PRINT "DU ER LIDT FORSIGTIG AF DIG, HVA???"
1710 PRINT "DU BLEV KUN NUMMER" ; K - K1 ; "AF DE"
1720 PRINT K ; "SPRING DER GIK FORUD FOR DIT."
1730 PRINT
1740 GOTO 2310
1750 PRINT "HM... DER ER VIST IKKE MEGET SPORTS--"
1760 PRINT "MAND I DIG, VEL? DER VAR" ; K
1770 PRINT "SPRING FØR DIT, OG DU VAR KUN" ; K1
1780 PRINT "SPRING BEDRE END DE VÆRSTE!!!!!!"
1790 PRINT "TAG DIG NU SAMMEN!"
1800 PRINT
1810 GOTO 2310
1820 PRINT "HOVSA! DU TRAK I UDLØSNINGSLINEN ALT"
1830 PRINT "FOR TIDLIGT. DER ER FORETAGET " ; K
1840 PRINT "SPRING FØR DIT, OG DU BLEV KUN"
1850 PRINT "NUMMER" ; K - K1 ; "!!! GØR NOGET VED"
1860 PRINT "DET."
1870 PRINT
1880 GOTO 2310
1890 PRINT "HVIL I FRED."
1900 PRINT
1910 GOTO 2290
1920 PRINT "GID EN ENGEL VIL FØRE DIG TIL PARADIS."
1930 PRINT
1940 GOTO 2290
1950 PRINT "FARVEL. PAX TECUM."
1960 PRINT
1970 GOTO 2290
1980 PRINT "DIN SKURK."
1990 PRINT
2000 GOTO 2290
2010 PRINT
2020 PRINT "FOR S...%$&£?<>^%$`£$, ETC."
2030 PRINT
2040 GOTO 2290
2050 PRINT
2060 PRINT "ET VELRETTET SPARK DET RETTE STED"
2070 PRINT "VIL HJÆLPE GEVALDIGT!!!"
2080 PRINT
2090 GOTO 2290
2100 PRINT
2110 PRINT "HMMM. DU SKULLE HAVE VALGT ET KORTERE"
2120 PRINT "TIDSINTERVAL."
2130 PRINT
2140 GOTO 2290
2150 PRINT "MUMLE, MUMLE, MUMLE...."
2160 PRINT
2170 GOTO 2290
2180 PRINT "MAN SKULLE TRO DU VAR UDE OG PLUKKE"
2190 PRINT "BLOMSTER. TROR DU DET ER EN SKOVTUR?"
2200 PRINT
2210 GOTO 2290
2220 PRINT "HVAD DER KOMMER LET, GØR LET."
2230 PRINT
2240 GOTO 2290
2250 PRINT INT (SQR (2 * D1 / A) * 1000) /
1000 ; "SPLAT"
2260 ON INT (1 + (10 * RND (1))) GOTO
1890,1920,1950,1980,2010,2050,2100,2150,2180,2220
2270 PRINT INT (((V / A) + ((D1 - (V ^ 2 / (2 *
A))) / V) * 1000) / 1000 ; "SPLAT"
2280 GOTO 2260
2290 PRINT "JEG VIL GIVE DIG ENDNU EN CHANCE."
2300 GOTO 2310
2310 PRINT
2320 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') " ;
2330 INPUT Z$
2340 PRINT
2350 IF LEFT$ (Z$,1) = "J" THEN GOTO 210
2360 IF LEFT$ (Z$,1) = "N" THEN GOTO 2390
2370 PRINT "SVAR 'JA' ELLER 'NEJ'"
2380 GOTO 2310
2390 PRINT "ÅÅÅR, TAG NU OG PRØV IGEN, IKKE " ;
2400 INPUT Z$
2410 PRINT
2420 IF LEFT$ (Z$,1) = "J" THEN GOTO 210
2430 IF LEFT$ (Z$,1) < > "N" THEN PRINT "'JA'
ELLER 'NEJ'" : GOTO 2390
2440 PRINT "ZZZZZZZZZZZZZZZZZZZZZZZZZZZZZZ, SNORK"
2450 END

```

# Stjerner

I dette spil vælger datamaten et tilfældigt tal mellem 1 og 100 (eller en anden maksimumværdi, som du selv kan indføre i linienummer 170). Du skal forsøge at gætte tallet, mens datamaten giver dig svar om, hvor tæt du er. En stjerne (\*) betyder, at du er ret langt fra tallet, mens 7 stjerner (\*\*\*\*\*) betyder, at du er meget nær. Du får 7 gæt.

Overfladisk betragtet ser dette spil ud til at ligne Gæt meget, men selve gættestrategien er anderledes i Stjerner. Du kan jo forsøge at finde ud af en eller flere strategier, der gør det let at finde det rette tal.

Bob Albrecht fra People's Computer Company har konstrueret dette spil.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

STJERNER

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

JEG TÆNKER PÅ ET HELTAL MELLEML 1 OG 100  
DU SKAL PRØVE AT GÆTTE TALLEL. NÅR DU  
HAR GÆTTET UDSKRIVER JEG EN ELLER FLE-  
RE STJERNER (\*). JO FLERE STJERNER JEG  
SKRIVER, JO NÆRMERE ER DU PÅ DET TAL  
JEG TÆNKER PÅ. DU FÅR 7 GÆT.

1 STJERNE (\*) BETYDER LANGT VÆK!  
7 STJERNER (\*\*\*\*\*) BETYDER MEGET  
TÆT PÅ.

OK...JEG TÆNKER NU PÅ ET TAL,  
DU KAN BEGYNDE AT GÆTTE.

DIT GÆT ?33  
\*\*

DIT GÆT ?80  
\*\*\*\*\*

DIT GÆT ?70  
\*\*\*\*\*

DIT GÆT ?74

\*\*\*\*\*!!!

DU GÆTTEDE DET PÅ 4 GÆT!!!

LAD OS SPILLE IGEN...

OK...JEG TÆNKER NU PÅ ET TAL,  
DU KAN BEGYNDE AT GÆTTE.

DIT GÆT ?30  
\*\*\*\*\*

DIT GÆT ?25  
\*\*\*\*\*

DIT GÆT ?27  
\*\*\*\*\*

DIT GÆT ?28

\*\*\*\*\*!!!

DU GÆTTEDE DET PÅ 4 GÆT!!!

LAD OS SPILLE IGEN...

OK...JEG TÆNKER NU PÅ ET TAL,  
DU KAN BEGYNDE AT GÆTTE.

DIT GÆT ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                STJERNER"
130 PRINT "-----"
140 PRINT
150 REM  People's Computer Center, Menlo Park, CA.
160 REM  A er lig højeste værdi, M er lig antal gæt
170 A = 100
180 M = 7
190 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
200 INPUT A$
210 PRINT
220 IF LEFT$(A$,1) = "N" THEN 350
230 REM  Spilleinstruktioner
240 PRINT "JEG TÆNKER PÅ ET HELTAL MELLEML 1 OG";A
250 PRINT "DU SKAL PRØVE AT GÆTTE TALLEL. NÅR DU"
260 PRINT "HAR GÆTTET UDSKRIVER JEG EN ELLER FLE-"
270 PRINT "RE STJERNER (*). JO FLERE STJERNER JEG"
280 PRINT "SKRIVER, JO NÆRMERE ER DU PÅ DET TAL"
290 PRINT "JEG TÆNKER PÅ. DU FÅR";M;"GÆT."
300 PRINT
310 PRINT
320 PRINT " 1 STJERNE (*) BETYDER LANGT VÆK!"
330 PRINT " 7 STJERNER (*****) BETYDER MEGET"
340 PRINT "                TÆT PÅ."
350 PRINT
360 X = INT (A * RND (1) + 1)
370 PRINT "OK...JEG TÆNKER NU PÅ ET TAL,"
380 PRINT "DU KAN BEGYNDE AT GÆTTE."
390 REM  Begynd at gætte. Du har M gæt.
400 FOR K = 1 TO M
410 PRINT
420 PRINT "DIT GÆT ";
430 INPUT G
440 IF G = X THEN 680
450 D = ABS (G - X)
460 IF D > = 64 THEN 580
470 IF D > = 32 THEN 570
480 IF D > = 16 THEN 560
490 IF D > = 8 THEN 550
500 IF D > = 4 THEN 540
510 IF D > = 2 THEN 530
520 PRINT "*";
530 PRINT "*";
540 PRINT "*";
550 PRINT "*";
560 PRINT "*";
570 PRINT "*";
580 PRINT "*";
590 PRINT
600 NEXT K
610 REM  Gættede det ikke på M gæt.
620 PRINT
630 PRINT "DES VÆRRE, DET VAR DINE";M;"GÆT."
640 PRINT
650 PRINT "TALLEL VAR:";X
660 GOTO 350
670 REM  Vi har en vidner
680 PRINT
690 FOR N = 1 TO 35
700 PRINT "*";
710 NEXT N
720 PRINT "!!!!"
730 PRINT
740 PRINT "DU GÆTTEDE DET PÅ";K;"GÆT!!!"
750 PRINT
760 PRINT "LAD OS SPILLE IGEN..."
770 GOTO 350
780 END

```

# Super star trek

Mange forskellige versioner af Star trek har eksisteret på forskellige universiteter siden slutningen af tresserne. Jeg kan huske at have spillet en variant på Carnegie-Mellon-universitetet i 1967 eller 1968 og en helt anden version på Berkeley. Disse spil var imidlertid meget langt fra den omfattende version, der blev skrevet af Mike Mayfield fra Centerline Engineering og/eller Custom Data. Programmet blev skrevet på en HP2000C og var færdigprogrammeret i oktober 1972. Denne version blev den »standardiserede version« i februar 1973, da den blev optaget i HP's programbibliotek og indlæst i en masse datamaskiner på HP Data Center.

I sommeren 1973 oversatte jeg HP-versionen til BASIC-PLUS for DEC's RSTS-11 compiler og tilføjede et par programstumper hist og her, nu da jeg var i gang. Mary Cole fra DEC bidrog virkelig meget til dette arbejde. Senere det år udgav jeg versionen under navnet Spacewar (Spacewar – et lidt fejlagtigt navn) i min bog »101 BASIC COMPUTER GAMES«.

I dag er det vanskeligt at finde et større dataanlæg i USA, hvor en af disse versioner ikke spilles.

## Kvadrantnomenklatur

For nyligt er der blevet rejst kritik over for oprindelsen af den »kvadrantnomenklatur«, der anvendes på alle standard CG (Cartesianske Galaktiske) kort. For den, der blot er en lille smule interesseret i historie, er forklaring unødvendig. Dog vil følgende udredning være tilstrækkelig til at få kritikerne til at tie:

Som ethvert skolebarn ved, så har de fleste intelligente civilisationer i Mælkevejen indført betegnelserne på de fleste planeter lang tid før den tredje Magellanske konference, hvor den såkaldt »26'ende overenskomst« blev vedtaget. I dette historiske dokument vedkendte de deltagende civilisationer, at alle to-dimensionale beskrivelser af galaksen skulle indeholde 64 områder, inddelt i en matrix på 8 gange 8 positioner. Dette blev delvis gjort for at hædre befolkningen på Jorden (som havde gjort en stor indsats, da planet-føderationen blev skabt). På Jorden havde de galaktiske kort i mange hundrede år altid markeret 16 hovedområder, som hver var opkaldt efter velkendte formationer på den del af himlen, man kunne se fra Jorden. Hvert af disse områder blev inddelt i fire »kvadranter«, som blev angivet med romerske tal (hvis oprindelse er gået tabt).

Endnu i dag angiver logbøgerne på rumskibe, der kommer fra baser, der ligger tæt ved Jorden, de galaktiske områder som »kvadranter«.

De forhold der hersker mellem den historiske og den standardiserede nomenklatur vises på nedenstående forenklede CG-kort.

	1	2	3	4	5	6	7	8
1		ANTARES				SIRIUS		
	I	II III IV				II III IV		
2		RIGEL				DENEK		
	I	II III IV				II III IV		
3		PROCYON				CAPELLA		
	I	II III IV				II III IV		
4		VEGA				BETELGEUSE		
	I	II III IV				II III IV		
5		CANOPUS				ALDEBARAN		
	I	II III IV				II III IV		
6		ALTAIR				REGULUS		
	I	II III IV				II III IV		
7		SAGITTARIUS				ARCTURUS		
	I	II III IV				II III IV		
8		POLLUX				SPIKA		
	I	II III IV				II III IV		

## Spilleregler og bemærkninger til Super Star Trek Af Robert Leedom og David Ahl

1. Målsætning. Du er kaptajn på rumskibet »Enterprise«, og det er din opgave at finde frem til og tilintetgøre en flåde fjendtlige Klingon-rumskibe (sædvanligvis 17 stk.), der truer den Forenede Planet Føderation. Du skal udføre denne mission inden for et bestemt antal stjernerdatoer. Du har også to eller tre af Planet Føderationens rumbaser til rådighed, hvor du kan få proviant, brændstof og våben.

2. Du vil ved spillets begyndelse blive placeret et sted i galaksen. Denne er inddelt i et kvadrantnet, der måler 8 gange 8 kvadranter. Det astronomiske navn, som en kvadrant har, angives altid, når du kommer ind i den (se ovenstående afsnit om kvadrantnomenklaturen). Hver kvadrant er inddelt i et net, hvis sektioner måler 8 gange 8 felter.

3. I sektions-diagrammerne bruges følgende symboler:

- (\*) Enterprise
- +K+ Klingon
- »|« Rumbase
- \* Stjerne

4. Du har 8 kommandoer til din rådighed (En detaljeret beskrivelse af de enkelte kommandoer gives i programmets instruktioner):

- NAV: Styrer Enterprises' kurs og afstandsfaktor
- NSS: Nær-sensor scanner (rækkevidde: en kvadrant)
- FSS: Fjern-sensor scanner (rækkevidden: 9 kvadranter)
- PHA: Phaser-kanon (energi-kanon)
- TOR: Photon-torpedo
- BES: Beskyttelsesskjold (beskytter mod fjendtlig phaser-beskydning)
- SKA: Skadesrapport
- COM: Kald styrecomputer

5. Styredatamaten kan anvendes til at foretage følgende (udførlig beskrivelse i programmets instruktioner):

- 0 Sammenfattende galaktisk rapport
- 1 Statusrapport
- 2 Photon-torpedo data
- 3 Navigationsdata til rumbase
- 4 Kurs/afstands-beregner
- 5 Kort over galaksen med områdenavne

6. Visse rapporteringer om rumskibets situation aflægges af officererne i Enterprise. Dvs. de figurer, som var med i den oprindelige TV-version: Spock, Scott, Uhura, Chekov, etc.

7. Klingonerne kan bevæge sig inden for deres respektive kvadrant. Hvis du forsøger at styre tæt på dem, vil de flytte sig og skyde mod dig.

8. Bemærkninger om affyring af skud og skader:

A: Phaser-kanonernes virkning bliver mindre, jo længere væk målet er.

B: Hvis en Klingon rammer dig med tilstrækkelig stor kraft (i forhold til beskyttelsesskjoldets styrke), vil dit rumskib kunne lide skade ved dette. Disse skader vil fremgå af »Skadesrapporten«.



```

'BES' KOMMANDO TIL STYRING AF BESKYTTESESSKJOLD.
ANGIVER DET ANTAL ENERGIENHEDER DER ANVENDES TIL
OPBYGGELSE AF BESKYTTESESSKJOLDDET. ENERGIEN TAGES
FRA SKIBETS TOTALE ENERGIRESERVE.
BEMÆRK, AT ENERGIANGIVELSEN I STATUS-RAPPORTEN
INDEHOLDER ENERGIEN DER BRUGES PÅ BESKYTTESES-
SKJOLDDET.

'SKA' KOMMANDO DER UDSKRIVER SKADESRAPPORT.
UDSKRIVER OPLYSNINGER OM ALLE ENHEDERS
REPARATIONSTILSTAND. HVIS 'REP-TILST' HAR EN
NEGATIV VÆRDI, SÅ ER ENHEDEN MIDLERTIDIGT UDE AF
FUNKTION.

'COM' KOMMANDO DER FREMKALDER SKIBETS STYRECOMPUTER.
STYRECOMPUTEREN HAR FØLGENDE 6 VALGMULIGHEDER:

0: SAMMENFATTENDE GALAKTISK RAPPORT
VISER DE OPLYSNINGER OM TIDLIGERE NÆR- OG FJERN-
SENSOR-SCANNINGER SOM ER LAGRET I DATAMATENS
HUKOMMELSE.

1: STATUS RAPPORT
VISER ANTALLET AF KLINGONER, STJERNEDATA SAMT
HVOR MANGE RUMBASER DER ER TILBAGE I SPILLET.

2: PHOTON-TORPEDO DATA
ANGIVER KURS OG AFSTAND TIL ALLE KLINGON-
SKIBE I KVADRANTEN (BEREGNET FRA 'ENTERPRISE').

3: NAVIGATIONSDATA TIL RUMBASE
ANGIVER KURS OG AFSTAND TIL RUMSTATIONER I
KVADRANTEN.

4: KURS/AFSTANDS BEREGNER
GIVER MULIGHED FOR INDTASTNING AF KOORDINATER,
HVOREFTER KURS OG AFSTAND FRA 'ENTERPRISE' TIL
DISSE BEREGNES.

5: KORT OVER GALAKSEN MED OMRÅDENAVNE
UDSKRIVER ET KORT MED NAVNENE PÅ DE 16 HOVED-
OMRÅDER DER ANVENDES I SPILLET.

NU KAN DU INDLÆSE SELVE PROGRAMMET: 'SUPER STAR TREK'

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " SUPER STAR TREK INSTRUKTIONER"
130 PRINT "-----"
140 REM Instruktioner for 'SUPER STAR TREK' - 5/3 1978
150 FOR I = 1 TO 4
160 PRINT
170 NEXT I
180 PRINT TAB( 12);"*****"
190 PRINT TAB( 12);"*" "*"
200 PRINT TAB( 12);"*" "*"
210 PRINT TAB( 12);"*" "*** SUPER STAR TREK ***" "*"
220 PRINT TAB( 12);"*" "*"
230 PRINT TAB( 12);"*" "*"
240 PRINT TAB( 12);"*****"
250 FOR I = 1 TO 4
260 PRINT
270 NEXT I
280 PRINT " INSTRUKTIONER FOR SUPER STAR TREK"
290 PRINT "-----"
300 PRINT
310 PRINT "1. NÅR 'KOMMANDO ?' UDSKRIVES SKAL DU INDASTE EN AF"
320 PRINT " FØLGENDE LOVLIGE KOMMANDOER (NAV,NSS,FSS,PHA,TOR,BES,")
330 PRINT " SKA,COM ELLER XXX).""
340 PRINT
350 PRINT "2. HVIS DU INDTASTER EN ULOVLIG KOMMANDO VIL DER BLIVE"
360 PRINT " UDSKREVET EN KORT LISTE MED DE LOVLIGE KOMMANDOER.""
370 PRINT
380 PRINT "3. NOGLE KOMMANDOER KRÆVER, AT DER OGSÅ INDTASTES"
390 PRINT " DATA (F.EKS. SVARER KOMMANDOEN 'NAV' MED SPØRGSMALET"
400 PRINT " 'KURS (1-9) ?'). HVIS DU INDTASTER ULOVLIGE DATA"
410 PRINT " F.EKS. NEGATIVE TAL, VIL KOMMANDOSEKVENSEN BLIVE"
420 PRINT " AFBRUDT.""
430 PRINT
440 PRINT " GALAKSEN ER INDDELT I ET NET MED 8 GANGE 8 KVADRANTER.""
450 PRINT "HVER AF KVADRANTERNE ER YDERLIGERE INDDELT I 8 GANGE 8 SEK-"
460 PRINT "TORER.""

```

```

470 PRINT
480 PRINT " NÅR SPILLET BEGYNDER BLIVER DU PLACERET ET STED I"
490 PRINT "GALAKSEN OG HERFRA BEGYNDER DU DIN LØBEBANE SOM KAPTAJN PÅ"
500 PRINT "RUMSKIBET 'ENTERPRISE'. DIN MISSION BESTÅR I AT FINDE OG"
510 PRINT "ØDELÆGGE DE FJENDTLIGE KLINGON-KRIGSKIBE, DER TRUER"
520 PRINT "'PLANET-FØDERATIONENS' EKISTENS.""
530 PRINT
540 PRINT " SOM KAPTAJN HAR DU FØLGENDE KOMMANDOER TIL DIN RÅDIG-"
550 PRINT "HED TIL AT STYRE RUMSKIBET 'ENTERPRISE':"
560 PRINT
570 PRINT "'NAV' KOMMANDOEN STYRER MOTORERNE.""
580 PRINT " KURSEN ANGIVES VED HJÆLP AF 4 3 2"
590 PRINT " VEKTORER, SOM VIST PÅ . . ."
600 PRINT " FIGUREN TIL HØJRE -----> . . ."
610 PRINT " DER KAN ANVENDES BÅDE 5 ----+---- 1"
620 PRINT " HELTAL OG KOMMATAL. . . ."
630 PRINT " SÅLEDES VIL KURSEN 1.5 . . ."
640 PRINT " LIGGE MIDTVEJS MELLE 1 OG 2. 6 7 8"
650 PRINT " KURS"
660 PRINT " KURSVÆRDIERNE KAN NÆRME SIG 9,"
670 PRINT " DER SVARER TIL 1.0. EN KURS PÅ"
680 PRINT " F.EKS. 8.5 LIGGER DERFOR MELLE"
690 PRINT " 8 OG 1.""
700 PRINT
710 PRINT " EN RETNINGSFAKTOR HAR SAMME STØRRELSE SOM EN"
720 PRINT " KVADRANT. HVIS DU ØNSKER AT KOMME FRA KVADRANT"
730 PRINT " 6,5 TIL 5,5, SÅ SKAL DU BRUGE KURS 3, MED"
740 PRINT " RETNINGSFAKTOR 1.""
750 PRINT
760 PRINT
770 PRINT "'NSS' KOMMANDO FOR RAPPORT FRA NÆR-SENSOR SCANNER.""
780 PRINT " VISER DIG ET SCANNING-BILLEDE AF DEN KVADRANT DU"
790 PRINT " I ØJEBLIKKET BEFINDER DIG I.""
800 PRINT
810 PRINT " DIN NÆR-SENSOR SCANNER VISER FØLGENDE SYMBOLER:"
820 PRINT " <> = DIT RUMSKIBS POSITION"
830 PRINT " +K+ = KLINGON KRIGSRUMSKIB"
840 PRINT " >!< = RUMBASE, DER TILHØRER PLANET-FØDERATIONEN"
850 PRINT " HER KAN DU FR BRÆNDSTOF, REPARATION OG VÅBEN.""
860 PRINT " * = STJERNE.""
870 PRINT
880 PRINT " EN SAMMENFATTENDE 'STATUS' VIL OGSÅ BLIVE VIST.""
890 PRINT
900 PRINT
910 PRINT "'FSS' KOMMANDO FOR RAPPORT FRA FJERN-SENSOR SCANNER.""
920 PRINT " VISER FORHOLDENE I ALLE DE KVADRANTER SOM OMGIVER"
930 PRINT " 'ENTERPRISE' (DER VISES MIDT I SCANNER-BILLEDET).""
940 PRINT " FJERN-SENSOR SCANNERENS OPLYSNINGER ER KODET PÅ"
950 PRINT " FORMEN 'XYZ', HVOR TALLET X ANGIVER ANTAL KLINGON"
960 PRINT " KRIGSRUMSKIBE, Y ANTALLET AF RUMBASER OG Z ANTAL"
970 PRINT " STJERNER I KVADRANTEN.""
980 PRINT " SÅLEDES BETYDER '207', AT DER ER 2 KLINGONER, INGEN"
990 PRINT " RUMBASER OG 7 STJERNER I DEN PÅGÆLDENDE KVADRANT.""
1000 PRINT
1010 PRINT
1020 PRINT "'PHA' KOMMANDO DER STYRER PHASER-KANONER"
1030 PRINT " GIVER DIG MULIGHED FOR AT RAMME DE FJENDTLIGE"
1040 PRINT " KLINGON-SKIBE MED SÅ MEGEN ENERGI, AT DU KAN"
1050 PRINT " TRÆNGE IGENNEM DERES BESKYTTESESSKJOLD.""
1060 PRINT " (HUSK DOG, AT OGSÅ KLINGON-SKIBENE HAR PHASERE!).""
1070 PRINT
1080 PRINT
1090 PRINT "'TOR' KOMMANDO DER STYRER PHOTON-TORPEDOER.""
1100 PRINT " TIL STYRING AF PHOTON-TORPEDOERNES KURS, ANVENDES"
1110 PRINT " DE SAMME VEKTORER SOM VED STYRING AF MOTORERNE"
1120 PRINT " ('NAV').""
1130 PRINT " HVIS DU RAMMER KLINGON-SKIBET, VIL DET BLIVE"
1140 PRINT " ØDELAGT, OG DET VIL IKKE KUNNE SKYDE TILBAGE PÅ DIG.""
1150 PRINT " HVIS DU RAMMER VED SIDEN AF, SÅ BLIVER DU SELV"
1160 PRINT " BESKUDT AF ALLE KLINGON-SKIBE I KVADRANTEN.""
1170 PRINT
1180 PRINT " STYRE DATAMATEN ('COM'-KOMMANDOEN) GIVER MULIGHED"
1190 PRINT " FOR AT BEREGNE PHOTON-TORPEDOERNES KURS"
1200 PRINT " (ALTERNATIV 2).""
1210 PRINT
1220 PRINT
1230 PRINT "'BES' KOMMANDO TIL STYRING AF BESKYTTESESSKJOLD.""
1240 PRINT " ANGIVER DET ANTAL ENERGIENHEDER DER ANVENDES TIL"
1250 PRINT " OPBYGGELSE AF BESKYTTESESSKJOLDDET. ENERGIEN TAGES"
1260 PRINT " FRA SKIBETS TOTALE ENERGIRESERVE.""
1270 PRINT " BEMÆRK, AT ENERGIANGIVELSEN I STATUS-RAPPORTEN"
1280 PRINT " INDEHOLDER ENERGIEN DER BRUGES PÅ BESKYTTESES-"
1290 PRINT " SKJOLDDET.""
1300 PRINT
1310 PRINT
1320 PRINT "'SKA' KOMMANDO DER UDSKRIVER SKADESRAPPORT.""
1330 PRINT " UDSKRIVER OPLYSNINGER OM ALLE ENHEDERS"
1340 PRINT " REPARATIONSTILSTAND. HVIS 'REP-TILST' HAR EN"

```

```

1350 PRINT "      NEGATIV VÆRDI, SÅ ER ENHEDEN MIDLERTIDIGT UDE AF"
1360 PRINT "      FUNKTION."
1370 PRINT
1380 PRINT
1390 PRINT "'COM' KOMMANDO DER FREMKALDER SKIBETS STYRECOMPUTER."
1400 PRINT "      STYRECOMPUTEREN HAR FØLGENDE 6 VALGMULIGHEDER:"
1410 PRINT
1420 PRINT "      0: SAMMENFATTENDE GALAKTISK RAPPORT"
1430 PRINT "      VISER DE OPLYSNINGER OM TIDLIGERE NÆR- OG FJERN-"
1440 PRINT "      SENSOR-SCANNINGER SOM ER LAGRET I DATAMATENS"
1450 PRINT "      HUKOMMELSE."
1460 PRINT
1470 PRINT "      1: STATUS RAPPORT"
1480 PRINT "      VISER ANTALLET AF KLINGONER, STJERNEDATA SAMT"
1490 PRINT "      HVOR MANGE RUMBASER DER ER TILBAGE I SPILLET."
1500 PRINT
1510 PRINT "      2: PHOTON-TORPEDO DATA"
1520 PRINT "      ANGIVER KURS OG AFSTAND TIL ALLE KLINGON-"
1530 PRINT "      SKIBE I KVADRANTEN (BEREGNET FRA 'ENTERPRISE')."
1540 PRINT
1550 PRINT "      3: NAVIGATIONSDATA TIL RUMBASE"
1560 PRINT "      ANGIVER KURS OG AFSTAND TIL RUMSTATIONER I"
1570 PRINT "      KVADRANTEN."
1580 PRINT
1590 PRINT "      4: KURS/AFSTANDS BEREGER"
1600 PRINT "      GIVER MULIGHED FOR INDFASTNING AF KOORDINATER,"
1610 PRINT "      HVOREFTER KURS OG AFSTAND FRA 'ENTERPRISE' TIL"
1620 PRINT "      DISSE BEREGNES."
1630 PRINT
1640 PRINT "      5: KORT OVER GALAKSEN MED OMRÅDENAVNE"
1650 PRINT "      UDSKRIVER ET KORT MED NAVNENE PÅ DE 16 HOVED-"
1660 PRINT "      OMRÅDER DER ANVENDES I SPILLET."
1670 PRINT
1680 PRINT
1690 PRINT "NU KAN DU INDLÆSE SELVE PROGRAMMET: 'SUPER STAR TREK'"
1700 END

```

MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY

-----  
SUPER STAR TREK  
-----

```

      *
     / \
    /   \
   /     \
  /       \
 /         \
/           \

```

RUMSKIBET USS ENTERPRISE ... NCC-1701

DU HAR FÆT FØLGENDE ORDRER:

ØDELÆG DE 16 FJENDTLIGE KLINGON RUMSKIBE, DER HAR INVADERET GALAKSEN FØR DISSE NÅR AT ANGRIBE PLANET-FØDERATIONENS HOVEDKVARTER.

ANGREBET VIL SKE PÅ STJERNEDATO: 3627, HVILKET GIVER DIG 27 DAGE TIL AT UDFØRE ORDREN INDEN DET ER FOR SENT.

DER ER 1 RUMBASER I GALAKSEN, DER KAN FORSYNE DIT SKIB MED BRÆNDSTOF OG VÆBEN, OG HVOR DU KAN FR DIT RUMSKIB REPARERET.

TAST <RETURN> NÅR DU ER PARAT TIL AT BEGYNDE ?

DIN MISSION BEGYNDER MED DIT RUMSKIB PLACERET I DEN GALAKTISKE KVADRANT 'REGULUS I'.

```

-----
*
STJERNEDATO: 3600
TILSTAND: GRØN
KVADRANT: 6, 5
SEKTOR: 7, 7
PHOTON-TORPEDOER: 10
TOTAL ENERGI: 3000
<*> BESKYTTELSESSKJOLD: 0
* * KLINGONER TILBAGE: 16
-----

```

KOMMANDO ?FSS

FJERN-SENSOR SCANNER RAPPORT FOR KVADRANT 6, 5

```

-----
: 004 : 005 : 005 :
-----
: 004 : 003 : 006 :
-----
: 108 : 005 : 102 :
-----

```

KOMMANDO ?NAV

KURS (0-9) ?8

AFSTANDSFAKTOR (0-8) ?1

DU BEVÆGER DIG NU IND I KVADRANT 'ARCTURUS II'.

KAMP-OMRÅDE      RØDT BEREDSKAB

\*\*\* BESKYTTELSESSKJOLD HAR FARLIGT LAVT NIVEAU

```

-----
STJERNEDATO: 3601
TILSTAND: *RØD*
KVADRANT: 7, 6
SEKTOR: 7, 7
PHOTON-TORPEDOER: 10
TOTAL ENERGI: 2982
BESKYTTELSESSKJOLD: 0
KLINGONER TILBAGE: 16
-----

```

KOMMANDO ?COM

STYREDATAMAT VENTER PÅ DINE INSTRUKTIONER ?2

FRA ENTERPRISE TIL KLINGON KRIGSRUMSKIB

KURS = 4.2  
AFSTAND = 6.40312424

KOMMANDO ?TOR

PHOTON-TORPEDO KURS (1-9) ?4.2

```

TORPEDOBANE:
6, 6
5, 5
5, 4
4, 3
3, 2

```

\*\*\* KLINGON TILINTETGJORT \*\*\*

KOMMANDO ?NAV

KURS (0-9) ?5

AFSTANDSFAKTOR (0-8) ?2

SKADESRAPPORT:  
STYRECOMPUTER SKADET.

DU BEVÆGER DIG NU IND I KVADRANT 'SAGITTARIUS IV'.

KAMP-OMRÅDE      RØDT BEREDSKAB

\*\*\* BESKYTTELSESSKJOLD HAR FARLIGT LAVT NIVEAU

```

-----
* * *
* *
*
PHOTON-TORPEDOER: 9
TOTAL ENERGI: 2954
BESKYTTELSESSKJOLD: 0
KLINGONER TILBAGE: 15
-----

```

KOMMANDO ?BES

ENERGI TIL RØDIGHED: 2954 ENHEDER  
HVOR MANGE ENHEDER SKAL ANVENDES TIL SKJOLD ?1000

BESKYTTELSESKONTROLLEN RAPPORTERER: 'SKJOLDET BRUGER NU 1000 ENHEDER SOM BEORDRET'.

KOMMANDO ?PHA

EN DATAMAT-FEJL HINDRER NØJAGTIGHED.  
PHASERKANONER INDSTILLET PÅ MÅLET.  
ENERGI TIL RÅDIGHED: 1954 ENHEDER.

HVOR MANGE ENHEDER SKAL AFFYRES ?1000

KLINGON I SEKTOR 7 , 1 ER RAMT MED 407 ENHEDER.

\*\*\* KLINGON TILINTETGJORT \*\*\*

KOMMANDO ?COM

STYRE DATAMAT UDE AF DRIFT.

KOMMANDO ?O

INDTAST EN AF FØLGENDE KOMMANDOER:

- NAV (STYRING AF MOTORER)
- NSS (NÆR-SENSOR SCANNER)
- FSS (FJERN-SENSOR SCANNER)
- PHA (AFFYR PHASER-KANONER)
- TOR (AFFYR PHOTON-TORPEDOER)
- BES (FORØG/FORMINDSK BESKYTTELSESSKJOLD)
- SKA (SKADESRAPPORT)
- COM (BRUG AF STYRECOMPUTER)
- XXX (AFSLUT MISSIONEN)

KOMMANDO ?FSS

FJERN-SENSOR SCANNER RAPPORT FOR KVADRANT 7 , 4

```

-----
: 005 : 004 : 003 :
-----
: 002 : 008 : 005 :
-----
: 007 : 002 : 007 :
-----

```

KOMMANDO ?NAV

KURS (0-9) ?1

AFSTANDSFAKTOR (0-8) ?4

DU BEVÆGER DIG NU IND I KVADRANT 'ARCTURUS IV'.

```

-----
*          *          STJERNEDATO: 3603
*          *          TILSTAND: GRØN
*          *          KVADRANT: 7 , 8
*          *          SEKTOR: 7 , 7
*          *          PHOTON-TORPEDOER: 9
*          *          TOTAL ENERGI: 1912
<*>       *          BESKYTTELSESSKJOLD: 1000
*          *          KLINGONER TILBAGE: 14
-----

```

KOMMANDO ?FSS

FJERN-SENSOR SCANNER RAPPORT FOR KVADRANT 7 , 8

```

-----
: 007 : 008 : *** :
-----
: 006 : 005 : *** :
-----
: 003 : 104 : *** :
-----

```

KOMMANDO ?NAV

KURS (0-9) ?7

AFSTANDSFAKTOR (0-8) ?1

DU BEVÆGER DIG NU IND I KVADRANT 'SPICA IV'.

KAMP-OMRÅDE RØDT BEREDSKAB

```

-----
*          *          STJERNEDATO: 3604
*          *          TILSTAND: *RØD*
*          *          KVADRANT: 8 , 8
*          *          SEKTOR: 7 , 7
*          *          PHOTON-TORPEDOER: 9
*          *          TOTAL ENERGI: 1894
<*>       *          BESKYTTELSESSKJOLD: 1000
+K+       *          KLINGONER TILBAGE: 14
-----

```

KOMMANDO ?PHA

EN DATAMAT-FEJL HINDRER NØJAGTIGHED.  
PHASERKANONER INDSTILLET PÅ MÅLET.  
ENERGI TIL RÅDIGHED: 894 ENHEDER.

HVOR MANGE ENHEDER SKAL AFFYRES ?100

KLINGON I SEKTOR 8 , 2 ER RAMT MED 49 ENHEDER.

(SENSORERNE VISER, AT KLINGON HAR 99.5225279 ENHEDER TILBAGE)

ENTERPRISE RAMT MED 56 ENHEDER AF KLINGON  
PLACERET I SEKTOR 8 , 2

<BESKYTTELSESSKJOLD MINDSKET TIL 944 ENHEDER>

KOMMANDO ?

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " SUPER STAR TREK"
130 PRINT "-----"
140 REM Super Startrek - 16/5 1978
150 PRINT
160 PRINT " "
170 PRINT " "
180 PRINT " "
190 PRINT " "
200 PRINT " "
210 PRINT
220 PRINT " RUMSKIBET USS ENTERPRISE ... NCC-1701"
230 PRINT
240 PRINT
250 PRINT
260 Z$ = " "
270 M$ = Z$ + Z$
280 DIM G(8,8),C(10,2),K(3,3),N(3),Z(8,8),D(8)
290 T = INT ( RND (1) * 20 + 20) * 100
300 TO = T
310 L6 = 1
320 T9 = 25 + INT ( RND (1) * 10)
330 D0 = 0:E = 3000:E0 = E
340 P = 10:P0 = P:S9 = 200:S = 0:B9 = 0:K9 = 0
350 X$ = "" : X0$ = " ER "
360 DEF FN D(D) = SQR ((K(I,1) - S1) ^ 2 + (K(I,2) - S2) ^ 2)
370 DEF FN R(R) = INT ( RND (1) * 7.98 + 1.01)
380 REM Intialisering af Enterprises' position
390 Q1 = FN R(1):Q2 = FN R(1):S1 = FN R(1):S2 = FN R(1)
400 FOR I = 1 TO 9
410 C(I,1) = 0:C(I,2) = 0
420 NEXT I
430 C(3,1) = - 1:C(2,1) = - 1:C(4,1) = - 1
440 C(4,2) = - 1:C(5,2) = - 1:C(6,2) = - 1
450 C(1,2) = 1:C(2,2) = 1:C(6,1) = 1
460 C(7,1) = 1:C(8,1) = 1:C(8,2) = 1:C(9,2) = 1
470 FOR I = 1 TO 8
480 D(I) = 0
490 NEXT I
500 A1$ = "NAVNSSFSSPHATORBESSKACOMXXX"
510 REM Dan det der skal eksistere i galaksen
520 REM K3=antal klingorner
530 REM B3=antal rumbaser
540 REM S3=antal stjerner
550 FOR I = 1 TO 8
560 FOR J = 1 TO 8
570 K3 = 0:Z(I,J) = 0:R1 = RND (1)
580 IF R1 > .98 THEN K3 = 3:K9 = K9 + 3: GOTO 610
590 IF R1 > .95 THEN K3 = 2:K9 = K9 + 2: GOTO 610
600 IF R1 > .80 THEN K3 = 1:K9 = K9 + 1
610 B3 = 0

```

```

620 IF RND (1) > .96 THEN B3 = 1:B9 = B9 + 1
630 G(I,J) = K3 * 100 + B3 * 10 + FN R(1)
640 NEXT J
650 NEXT I
660 IF K9 > T9 THEN T9 = K9 + 1
670 IF B9 < > 0 THEN 720
680 IF G(Q1,Q2) < 200 THEN G(Q1,Q2) = G(Q1,Q2) + 100:K9 = K9 + 1
690 B9 = 1
700 G(Q1,Q2) = G(Q1,Q2) + 10
710 Q1 = FN R(1):Q2 = FN R(1)
720 K7 = K9
730 IF B9 < > 1 THEN X$ = "S":X0$ = " ER "
740 PRINT "DU HAR FÆRET FØLGENDE ORDRE:"
750 PRINT
760 PRINT "ØDELÆG DE";K9;"FJENDTLIGE KLINGON RUMSKIBE, DER"
770 PRINT "HAR INVADERET GALAKSEN FØR DISSE NÅR AT ANGRIBE PLANET-"
780 PRINT "FØDERATIONENS HOVEDKVARTER."
790 PRINT
800 PRINT "ANGREBET VIL SKE PÅ STJERNEDATO";T0 + T9;" , HVILKET"
810 PRINT "GIVER DIG";T9;"DAGE TIL AT UDFØRE ORDREN INDEN DET"
820 PRINT "ER FOR SENT."
830 PRINT
840 PRINT "DER ER";B9;"RUMBASER I GALAKSEN, DER KAN FORSYNE"
850 PRINT "DIT SKIB MED BRÆNDSTOF OG VÆBEN, OG HVOR DU KAN FÅ DIT"
860 PRINT "RUMSKIB REPARERET."
870 PRINT
880 PRINT "TAST <RETURN> NÅR DU ER PARAT TIL AT BEGYNDE ";
890 INPUT W5$
900 I = RND (1)
910 REM Tilbage hertil når Enterprise går ind i ny kvadrant
920 Z4 = Q1:Z5 = Q2
930 K3 = 0:B3 = 0:S3 = 0:G5 = 0
940 D4 = .5 * RND (1)
950 Z(Q1,Q2) = G(Q1,Q2)
960 IF Q1 < 1 OR Q1 > 8 OR Q2 < 1 OR Q2 > 8 THEN 1180
970 GOSUB 7620
980 PRINT
990 IF T0 < > T THEN 1030
1000 PRINT "DIN MISSION BEGYNDE MED DIT RUMSKIB PLACERET I DEN"
1010 PRINT "GALAKTISKE KVADRANT ";G2$;"."
1020 GOTO 1040
1030 PRINT "DU BEVÆGER DIG NU IND I KVADRANT ";G2$;"."
1040 PRINT
1050 K3 = INT (G(Q1,Q2) * .01)
1060 B3 = INT (G(Q1,Q2) * .1) - 10 * K3
1070 S3 = G(Q1,Q2) - 100 * K3 - 10 * B3
1080 IF K3 = 0 THEN 1140
1090 PRINT "KAMP-OMRÅDE          RØDT BEREDSKAB"
1100 PRINT
1110 IF S > 200 THEN 1140
1120 PRINT "*** BESKYTTELSSESKJOLD HAR FARLIGT LAVT NIVEAU"
1130 PRINT
1140 FOR I = 1 TO 3
1150 K(I,1) = 0
1160 K(I,2) = 0
1170 NEXT I
1180 FOR I = 1 TO 3
1190 K(I,3) = 0
1200 NEXT I
1210 Q$ = Z$ + Z$ + Z$ + Z$ + Z$ + Z$ + Z$ + LEFT$ (Z$,17)
1220 REM Position i kvadrant, placer Klingoner (K3),
1230 REM Rumbaser (B3) og Stjerner (S3) andre steder
1240 A$ = "<*>"
1250 Z1 = S1:Z2 = S2
1260 GOSUB 7380
1270 IF K3 < 1 THEN 1350
1280 FOR I = 1 TO K3
1290 GOSUB 7310
1300 A$ = "+K+"
1310 Z1 = R1:Z2 = R2
1320 GOSUB 7380
1330 K(I,1) = R1:K(I,2) = R2:K(I,3) = S9 * (0.5 + RND (1))
1340 NEXT I
1350 IF B3 < 1 THEN 1400
1360 GOSUB 7310
1370 A$ = ">!<<"
1380 Z1 = R1:B4 = R1:Z2 = R2:B5 = R2
1390 GOSUB 7380
1400 FOR I = 1 TO S3
1410 GOSUB 7310
1420 A$ = " * "
1430 Z1 = R1:Z2 = R2
1440 GOSUB 7380
1450 NEXT I
1460 GOSUB 5560
1470 PRINT
1480 IF S + E > 10 THEN IF E > 10 OR D(7) = 0 THEN 1600
1490 PRINT
1500 PRINT "*****"
1510 PRINT "*** ALVORLIG FEJL ***"
1520 PRINT "*****"
1530 PRINT
1540 PRINT "DU HAR NETOP SØRGET FOR, AT DIT RUMSKIB"
1550 PRINT "ER STRANDET I RUMMET. DU HAR IKKE ENERGI"
1560 PRINT "NOK TIL AT MANØVRERE SKIBET OG STYRINGEN"
1570 PRINT "AF BESKYTTELSSESKJOLDET KAN I ØJEBLIKKET"
1580 PRINT "IKKE STILLES OM TIL MASKINRUMMET!!!!"
1590 GOTO 5290
1600 PRINT "KOMMANDO ";
1610 INPUT A$
1620 PRINT
1630 FOR I = 1 TO 9
1640 IF LEFT$ (A$,3) < > MID$ (A$,3 * I - 2,3) THEN 1660
1650 ON I GOTO 1800,1460,3180,3390,3840,4530,4700,6080,5360
1660 NEXT I
1670 PRINT "INDTAST EN AF FØLGENDE KOMMANDOER:"
1680 PRINT " NAV (STYRING AF MOTORER)"
1690 PRINT " NSS (NÆR-SENSOR SCANNER)"
1700 PRINT " FSS (FJERN-SENSOR SCANNER)"
1710 PRINT " PHA (AFFYR PHASER-KANONER)"
1720 PRINT " TOR (AFFYR PHOTON-TORPEDOER)"
1730 PRINT " BES (FORØG/FORMINDSK BESKYTTELSSESKJOLD)"
1740 PRINT " SKA (SKADESRAPPORT)"
1750 PRINT " COM (BRUG AF STYRECOMPUTER)"
1760 PRINT " XXX (AFSLUT MISSIONEN)"
1770 PRINT
1780 GOTO 1470
1790 REM Kontrol af kurs begynder her
1800 PRINT "          KURS (0-9) ";
1810 INPUT C1
1820 PRINT
1830 IF C1 = 9 THEN C1 = 1
1840 IF C1 > = 1 AND C1 < 9 THEN 1870
1850 PRINT "LØJTNANT SULU RAPPORTERER: 'FEJLAGTIGE KURS-DATA'"
1860 GOTO 1470
1870 X$ = "8"
1880 IF D(1) < 0 THEN X$ = "0.2"
1890 PRINT "AFSTANDSFAKTOR (0-";X$;") ";
1900 INPUT W1
1910 PRINT
1920 IF D(1) < 0 AND W1 > .2 THEN 1990
1930 IF W1 > 0 AND W1 < = 8 THEN 2020
1940 IF W1 = 0 THEN 1470
1950 PRINT "MASKINMESTER SCOTT RAPPORTERER: 'MOTORERNE KAN IKKE'"
1960 PRINT "          HOLDE TIL EN AFSTANDS-"
1970 PRINT "          FAKTOR PÅ";W1;"!!'"
1980 GOTO 1470
1990 PRINT "KURSMOTORERNE ER BESKADIGET."
2000 PRINT "MAKSIMAL HASTIGHED ER DERFOR KUN 0.2."
2010 GOTO 1470
2020 N = INT (W1 * 8 + .5)
2030 IF E - N > = 0 THEN 2130
2040 PRINT "RAPPORT FRA MASKINRUMMET: 'UTILSTRÆKKELIG ENERGI TIL'"
2050 PRINT "          RÅDIGHED. KAN IKKE MANØVRERE MED'"
2060 PRINT "          AFSTANDSFAKTOR";W1;"!!'"
2070 IF S < N - E OR D(7) < 0 THEN 1470
2080 PRINT "RAPPORT FRA KONTROLRUMMET: ";S;"ENERGIENHEDER"
2090 PRINT "          AFLEDES I ØJEBLIKKET TIL"
2100 PRINT "          BESKYTTELSSESKJOLDET."
2110 GOTO 1470
2120 REM Klingoner flyttes/skyder mod Enterprise
2130 FOR I = 1 TO K3
2140 IF K(I,3) = 0 THEN 2220
2150 A$ = " "
2160 Z1 = K(I,1):Z2 = K(I,2)
2170 GOSUB 7380
2180 GOSUB 7310
2190 K(I,1) = Z1:K(I,2) = Z2
2200 A$ = "+K+"
2210 GOSUB 7380
2220 NEXT I
2230 GOSUB 5040
2240 D1 = 0:D6 = W1
2250 IF W1 > = 1 THEN D6 = 1
2260 FOR I = 1 TO 8
2270 IF D(I) > = 0 THEN 2360
2280 D(I) = D(I) + D6
2290 IF D(I) > = .1 AND D(I) < 0 THEN D(I) = - .1: GOTO 2360
2300 IF D(I) < 0 THEN 2360
2310 IF D1 < > 1 THEN D1 = 1: PRINT "SKADESRAPPORT: "
2320 PRINT "          ";
2330 R1 = I
2340 GOSUB 7440
2350 PRINT G2$;" REPARERET."
2360 NEXT I
2370 IF RND (1) > .2 THEN 2520

```



```

2380 R1 = FN R(1)
2390 IF RND (1) > = .6 THEN 2460
2400 D(R1) = D(R1) - ( RND (1) * 5 + 1)
2410 PRINT "SKADESRAPPORT: "
2420 GOSUB 7440
2430 PRINT G2$;" SKADET."
2440 PRINT
2450 GOTO 2520
2460 D(R1) = D(R1) + RND (1) * 3 + 1
2470 PRINT "SKADESRAPPORT: "
2480 GOSUB 7440
2490 PRINT G2$;" - REPARATIONSSITUATIONEN FORBEDRET."
2500 PRINT
2510 REM Begynd at bevæge rumskibet
2520 A$ = " "
2530 Z1 = INT (S1):Z2 = INT (S2)
2540 GOSUB 7380
2550 X1 = C(C1,1) + (C(C1 + 1,1) - C(C1,1)) * (C1 - INT (C1))
2560 X = S1:Y = S2
2570 X2 = C(C1,2) + (C(C1 + 1,2) - C(C1,2)) * (C1 - INT (C1))
2580 Q4 = Q1:Q5 = Q2
2590 FOR I = 1 TO N
2600 S1 = S1 + X1:S2 = S2 + X2
2610 IF S1 < 1 OR S1 > = 9 OR S2 < 1 OR S2 > = 9 THEN 2840
2620 S8 = INT (S1) * 24 + INT (S2) * 3 - 26
2630 IF MID$(Q$,S8,2) = " " THEN 2700
2640 S1 = INT (S1 - X1):S2 = INT (S2 - X2)
2650 PRINT "KURSMOTORERNE KOBLET FRA I ";
2660 PRINT "SEKTOR";S1;" ";S2
2670 PRINT "GRUNDET ELENDIG NAVIGATION."
2680 PRINT
2690 GOTO 2720
2700 NEXT I
2710 S1 = INT (S1):S2 = INT (S2)
2720 A$ = "<*>"
2730 Z1 = INT (S1):Z2 = INT (S2)
2740 GOSUB 7380
2750 GOSUB 3100
2760 T8 = 1
2770 IF W1 < 1 THEN T8 = .1 * INT (10 * W1)
2780 T = T + T8
2790 IF T > T0 + T9 THEN 5290
2800 REM Se efter om sammenkobling foretaget
2810 REM Hent kommando
2820 GOTO 1460
2830 REM Kvadrantgrænser overskredet
2840 X = 8 * Q1 + X + N * X1:Y = 8 * Q2 + Y + N * X2
2850 Q1 = INT (X / 8):Q2 = INT (Y / 8)
2860 S1 = INT (X - Q1 * 8):S2 = INT (Y - Q2 * 8)
2870 IF S1 = 0 THEN Q1 = Q1 - 1:S1 = 8
2880 IF S2 = 0 THEN Q2 = Q2 - 1:S2 = 8
2890 X5 = 0
2900 IF Q1 < 1 THEN X5 = 1:Q1 = 1:S1 = 1
2910 IF Q1 > 8 THEN X5 = 1:Q1 = 8:S1 = 8
2920 IF Q2 < 1 THEN X5 = 1:Q2 = 1:S2 = 1
2930 IF Q2 > 8 THEN X5 = 1:Q2 = 8:S2 = 8
2940 IF X5 = 0 THEN 3050
2950 PRINT "LØJTNANT UHURA RAPPORTERER, AT DER ER INDLØBET EN"
2960 PRINT "MEDDELELSE FRA FØDERATIONENS OVERKOMMANDO. DEN LYDER:"
2970 PRINT " 'TILLADELSE TIL AT OVERSKRIDE GALAKSENS GRÆNSER"
2980 PRINT " 'ER HERMED *NÅGTET*. SLUK FOR MOTORERNE."
2990 PRINT
3000 PRINT "MASKINMESTER SCOTT RAPPORTERER: 'MOTORER SLÆT FRA'."
3010 PRINT
3020 PRINT "ENTERPRISE ER I SEKTOR";S1;" ";S2
3030 PRINT " I KVADRANT";Q1;" ";Q2;" ."
3040 IF T > T0 + T9 THEN 5290
3050 IF 8 * Q1 + Q2 = 8 * Q4 + Q5 THEN 2720
3060 T = T + 1
3070 GOSUB 3100
3080 GOTO 920
3090 REM Manøverenergi
3100 E = E - N - 10
3110 IF E > = 0 THEN RETURN
3120 PRINT "DER TAPPES ENERGI FRA BESKYTTELSESSKJOLDET, SÅ"
3130 PRINT "DEN ØNSKEDE MANØVER KAN FULDFØRES."
3140 S = S + E:E = 0
3150 IF S < = 0 THEN S = 0
3160 RETURN
3170 REM Fjern-sensor scanner
3180 IF D(3) < 0 THEN PRINT "FJERN-SENSOR SCANNER UDE AF DRIFT": GOTO
1470
3190 PRINT "FJERN-SENSOR SCANNER RAPPORT ";
3200 PRINT "FOR KVADRANT";Q1;" ";Q2
3210 PRINT
3220 O1$ = "-----"
3230 PRINT O1$
3240 FOR I = Q1 - 1 TO Q1 + 1
3250 N(1) = - 1:N(2) = - 2:N(3) = - 3
3260 FOR J = Q2 - 1 TO Q2 + 1
3270 IF I > 0 AND I < 9 AND J > 0 AND J < 9 THEN N(J - Q2 + 2) =
G(I,J):Z(I,J) = G(I,J)
3280 NEXT J
3290 FOR L = 1 TO 3
3300 PRINT ": ";
3310 IF N(L) < 0 THEN PRINT "*** ";: GOTO 3330
3320 PRINT RIGHT$ ( STR$ (N(L) + 1000),3);" ";
3330 NEXT L
3340 PRINT ":"
3350 PRINT O1$
3360 NEXT I
3370 GOTO 1470
3380 REM Phaser
3390 IF D(4) < 0 THEN PRINT "PHASER-KANONER UDE AF DRIFT.": GOTO 1470
3400 IF K3 > 0 THEN 3460
3410 PRINT "VIDENSKABSMANDEN DR. SPOCK RAPPORTERER: 'SENSOR-SCANNERE"
3420 PRINT " VISER, AT DER IKKE"
3430 PRINT " ER FJENDTLIGE SKIBE"
3440 PRINT " I DENNE KVADRANT."
3450 GOTO 1470
3460 IF D(8) < 0 THEN PRINT "EN DATAMAT-FEJL HINDRER NØJAGTIGHED."
3470 PRINT "PHASERKANONER INDSTILLET PÅ MÅLET."
3480 PRINT "ENERGI TIL RÅDIGHED.":E;"ENHEDER."
3490 PRINT
3500 PRINT "HVOR MANGE ENHEDER SKAL AFFYRES ";
3510 INPUT X
3520 PRINT
3530 IF X < = 0 THEN 1470
3540 IF E - X < 0 THEN 3480
3550 E = E - X
3560 IF D(7) < 0 THEN X = X * RND (1)
3570 H1 = INT (X / K3)
3580 FOR I = 1 TO 3
3590 IF K(I,3) < = 0 THEN 3800
3600 H = INT ((H1 / FN D(0)) * ( RND (1) + 2))
3610 IF H > .15 * K(I,3) THEN 3650
3620 PRINT "SENSORERNE VISER INGEN SKADEVIRKNING PÅ FJENDEN"
3630 PRINT "PÅ POSITION";K(I,1);" ";K(I,2)
3640 GOTO 3800
3650 K(I,3) = K(I,3) - H
3660 PRINT "KLINGON I SEKTOR";K(I,1);" ";K(I,2);
3670 PRINT "ER RAMT MED";H;"ENHEDER."
3680 PRINT
3690 IF K(I,3) < = 0 THEN PRINT "*** KLINGON TILINTETGJORT ***": GOTO
3730
3700 PRINT "(SENSORERNE VISER, AT KLINGON HAR";
3710 PRINT K(I,3);"ENHEDER TILBAGE)"
3720 GOTO 3800
3730 K3 = K3 - 1:K9 = K9 - 1
3740 Z1 = K(I,1):Z2 = K(I,2)
3750 A$ = " "
3760 GOSUB 7380
3770 K(I,3) = 0:G(Q1,Q2) = G(Q1,Q2) - 100
3780 Z(Q1,Q2) = G(Q1,Q2)
3790 IF K9 < = 0 THEN 5490
3800 NEXT I
3810 GOSUB 5040
3820 GOTO 1470
3830 REM Photon-torpedo
3840 IF P < = 0 THEN PRINT "ALLE PHOTON-TORPEDOER ER BRUGT!": GOTO 1470
3850 IF D(5) < 0 THEN PRINT "PHOTON RØRENE ER UDE AF DRIFT.": GOTO 1470
3860 PRINT "PHOTON-TORPEDO KURS (1-9) ";
3870 INPUT C1
3880 IF C1 = 9 THEN C1 = 1
3890 IF C1 > = 1 AND C1 < 9 THEN 3920
3900 PRINT "SERGEANT CHEKOV RAPPORTERER: 'FEJLAGTIGE KURS DATA'."
3910 GOTO 1470
3920 X1 = C(C1,1) + (C(C1 + 1,1) - C(C1,1)) * (C1 - INT (C1))
3930 E = E - 2:P = P - 1
3940 X2 = C(C1,2) + (C(C1 + 1,2) - C(C1,2)) * (C1 - INT (C1))
3950 X = S1:Y = S2
3960 PRINT
3970 PRINT "TORPEDOBANE:"
3980 X = X + X1:Y = Y + X2
3990 X3 = INT (X + .5):Y3 = INT (Y + .5)
4000 IF X3 < 1 OR X3 > 8 OR Y3 < 1 OR Y3 > 8 THEN 4490
4010 PRINT " ";X3;" ";Y3
4020 A$ = " "
4030 Z1 = X:Z2 = Y
4040 GOSUB 7540
4050 IF Z3 < > 0 THEN 3980
4060 A$ = "+K+"
4070 Z1 = X:Z2 = Y
4080 GOSUB 7540
4090 IF Z3 = 0 THEN 4200
4100 PRINT

```

```

4110 PRINT "*** KLINGON TILINTETGJORT ***"
4120 K3 = K3 - 1:K9 = K9 - 1
4130 IF K9 < = 0 THEN 5490
4140 FOR I = 1 TO 3
4150 IF X3 = K(I,1) AND Y3 = K(I,2) THEN 4180
4160 NEXT I
4170 I = 3
4180 K(I,3) = 0
4190 GOTO 4420
4200 A$ = " * "
4210 Z1 = X:Z2 = Y
4220 GOSUB 7540
4230 IF Z3 = 0 THEN 4280
4240 PRINT "STJERNEN PR POSITION";X3;" ";Y3
4250 PRINT "ABSORBEREDE TORPEDO-ENERGIEN HELT."
4260 GOSUB 5040
4270 GOTO 1470
4280 A$ = ">!<"
4290 Z1 = X:Z2 = Y
4300 GOSUB 7540
4310 IF Z3 = 0 THEN 3860
4320 PRINT "*** RUMBASEN TILINTETGJORT ***"
4330 B3 = B3 - 1:B9 = B9 - 1
4340 IF B9 > 0 OR K9 > T - T0 - T9 THEN 4390
4350 PRINT "DET VAR DET! DU ER HERMED AFSKEDIGET SOM KAPTAJN,"
4360 PRINT "OG DU DØMMES TIL 99 STJERNEDATOERS HÆRDT ARBEJDE"
4370 PRINT "PR CYGNUS 12!!"
4380 GOTO 5360
4390 PRINT "PLANET FØDERATIONEN STUDERER NU DIT GENERALIEBLAD"
4400 PRINT "OG OVERVEJER OM DU SKAL STILLES FOR EN KRIGSRET."
4410 D0 = 0
4420 Z1 = X:Z2 = Y
4430 A$ = " "
4440 GOSUB 7380
4450 G(Q1,Q2) = K3 * 100 + B3 * 10 + S3
4460 Z(Q1,Q2) = G(Q1,Q2)
4470 GOSUB 5040
4480 GOTO 1470
4490 PRINT "TORPEDOEN RAMTE VED SIDEN AF."
4500 GOSUB 5040
4510 GOTO 1470
4520 REM Beskyttelseskjoldet
4530 IF D(7) < 0 THEN PRINT "BESKYTTELSESSKJOLDET UDE AF DRIFT.": GOTO
1470
4540 PRINT "ENERGI TIL RÅDIGHED:";E + S;"ENHEDER"
4550 PRINT "HVOR MANGE ENHEDER SKAL ANVENDES TIL SKJOLD ";
4560 INPUT X
4570 PRINT
4580 IF X < 0 OR S = X THEN PRINT "<SKJOLDET ÆNDRRET>": GOTO 1470
4590 IF X < = E + S THEN 4640
4600 PRINT "BESKYTTELSESKONTROLLEN RAPPORTERER: 'DETTE ER INGEN'"
4610 PRINT "SKATTEKISTE!."
4620 PRINT " <SKJOLDET ÆNDRRET>"
4630 GOTO 1470
4640 E = E + S - X:S = X
4650 PRINT "BESKYTTELSESKONTROLLEN RAPPORTERER: 'SKJOLDET BRUGER NU'"
4660 PRINT " "; INT(S);"ENHEDER"
4670 PRINT " SOM BEORDRET!."
4680 GOTO 1470
4690 REM Skadesrapport
4700 IF D(6) > = 0 THEN 4920
4710 PRINT "SKADESRAPPORT IKKE TILGÆNGELIG."
4720 IF D0 = 0 THEN 1470
4730 D3 = 0
4740 FOR I = 1 TO 8
4750 IF D(I) < 0 THEN D3 = D3 + .1
4760 NEXT I
4770 IF D3 = 0 THEN 1470
4780 PRINT
4790 D3 = D3 + D4
4800 IF D3 > = 1 THEN D3 = .9
4810 PRINT "DER STÅR TEKNIKERE PARAT TIL AT REPARERE DIT SKIB."
4820 PRINT "ANSLÆT REPARATIONSTID:";.01 * INT(100 * D3);
4830 PRINT "STJERNEDAGE."
4840 PRINT
4850 PRINT "GIVER DU ORDRE TIL AT FORETAGE REPARATIONEN ('J'/'N') ";
4860 INPUT A$
4870 IF LEFT$(A$,1) < > "J" THEN 1470
4880 FOR I = 1 TO 8
4890 IF D(I) < 0 THEN D(I) = 0
4900 NEXT I
4910 T = T + D3 + .1
4920 PRINT
4930 PRINT "ENHED REPARATIONSTILSTAND"
4940 PRINT "-----"
4950 FOR R1 = 1 TO 8
4960 GOSUB 7440
4970 PRINT G2$; LEFT$(M$,36 - LEN(G2$)); INT(D(R1) * 100) * .01

4980 NEXT R1
4990 PRINT
5000 PRINT "*****"
5010 IF D0 < > 0 THEN 4730
5020 GOTO 1470
5030 REM Klingoner skyder
5040 IF K3 < = 0 THEN RETURN
5050 IF D0 < > 0 THEN PRINT "RUMBASENS SKJOLD BESKYTTER ENTERPRISE":
RETURN
5060 FOR I = 1 TO 3
5070 IF K(I,3) < = 0 THEN 5260
5080 H = INT((K(I,3) / FN D(1)) * (2 + RND(1)))
5090 S = S - H
5100 K(I,3) = K(I,3) / (3 + RND(1))
5110 PRINT
5120 PRINT "ENTERPRISE RAMT MED";H;"ENHEDER AF KLINGON"
5130 PRINT "PLACERET I SEKTOR";K(I,1);" ";K(I,2)
5140 PRINT
5150 IF S < = 0 THEN 5310
5160 PRINT "<BESKYTTELSESSKJOLD MINDSKET TIL";S;"ENHEDER>"
5170 PRINT
5180 IF H < 20 THEN 5260
5190 IF RND(1) > .6 OR H / S < = .02 THEN 5260
5200 R1 = FN R(1)
5210 D(R1) = D(R1) - H / S - .5 * RND(1)
5220 GOSUB 7440
5230 PRINT "SKADEKONTROLLEN RAPPORTERER: 'TRÆFFER HAR FORÅRSAGET'"
5240 PRINT " BESKADIGELSE PÅ"
5250 PRINT " ";G2$;"'"
5260 NEXT I
5270 RETURN
5280 REM Spillet er slut
5290 PRINT "STJERNEDATOEN ER: ";T
5300 GOTO 5360
5310 PRINT
5320 PRINT "ENTERPRISE ER TILINTETGJORT."
5330 PRINT "PLANET-FØDERATIONEN HAR TABT KAMPEN,"
5340 PRINT "OG VIL BLIVE BESEJRET."
5350 GOTO 5290
5360 PRINT "DA DIN MISSION OPHØRTE VAR DER"
5370 PRINT "ENDNU";K9;"KLINGON KRIGSRUMSKIBE TILBAGE."
5380 PRINT
5390 PRINT
5400 IF B9 = 0 THEN 5480
5410 PRINT "PLANET-FØDERATIONEN HAR BRUG FOR EN NY"
5420 PRINT "KAPTAJN TIL EN LIGNENDE MISSION....."
5430 PRINT "-- HVIS DER ER EN FRIVILLIG, BEDES"
5440 PRINT "DENNE TRÆDE FREM OG INDTASTE 'JAVEL' ";
5450 INPUT A$
5460 PRINT
5470 IF A$ = "JAVEL" THEN 290
5480 GOTO 7930
5490 PRINT "TILLYKKE, KAPTAJN! DET SIDSTE KLINGON"
5500 PRINT "KRIGSRUMSKIB ER TILINTETGJORT!"
5510 PRINT
5520 PRINT "DIN EFFEKTIVITET HAR VÆRET";
5530 PRINT 1000 * (K7 / (T - T0)) ^ 2
5540 GOTO 5380
5550 REM Nærsensor scanning og start
5560 FOR I = S1 - 1 TO S1 + 1
5570 FOR J = S2 - 1 TO S2 + 1
5580 IF INT(I + .5) < 1 OR INT(I + .5) > 8 OR INT(J + .5) < 1 OR
INT(J + .5) > 8 THEN 5630
5590 A$ = ">!<"
5600 Z1 = I:Z2 = J
5610 GOSUB 7540
5620 IF Z3 = 1 THEN 5670
5630 NEXT J
5640 NEXT I
5650 D0 = 0
5660 GOTO 5740
5670 D0 = 1
5680 C$ = "SAMMENKOBLET"
5690 E = E0:P = P0
5700 PRINT "BESKYTTELSESSKJOLDET MINDSKET GRUNDET SAMMENKOBLING."
5710 PRINT
5720 S = 0
5730 GOTO 5770
5740 IF K3 > 0 THEN C$ = "*RØD*": GOTO 5770
5750 C$ = "GRØN"
5760 IF E < E0 * .1 THEN C$ = "GUL"
5770 IF D(2) > = 0 THEN 5820
5780 PRINT
5790 PRINT "*** NÆR-SENSOR SCANNERE ER UDE AF DRIFT ***"
5800 PRINT
5810 RETURN
5820 O1$ = "-----"
5830 PRINT O1$

```

```


5840 FOR I = 1 TO 8
5850 FOR J = (I - 1) * 24 + 1 TO (I - 1) * 24 + 22 STEP 3
5860 PRINT " "; MID$(Q$,J,3);
5870 NEXT J
5880 ON I GOTO 5890,5910,5930,5950,5970,5990,6010,6030
5890 PRINT "          STJERNEDATO: "; INT (T * 10) * .1
5900 GOTO 6040
5910 PRINT "          TILSTAND: ";C$
5920 GOTO 6040
5930 PRINT "          KVADRANT: ";Q1;" ";Q2
5940 GOTO 6040
5950 PRINT "          SEKTOR: ";S1;" ";S2
5960 GOTO 6040
5970 PRINT "          PHOTON-TORPEDOER: "; INT (P)
5980 GOTO 6040
5990 PRINT "          TOTAL ENERGI: "; INT (E + S)
6000 GOTO 6040
6010 PRINT "          BESKYTTELSESSKJOLD: "; INT (S)
6020 GOTO 6040
6030 PRINT "          KLINGONER TILBAGE: "; INT (K9)
6040 NEXT I
6050 PRINT O1$
6060 RETURN
6070 REM Styredatamat
6080 IF D(8) < 0 THEN PRINT "STYREDATAMAT UDE AF DRIFT.": GOTO 1470
6090 PRINT "STYREDATAMAT VENTER PÅ DINE INSTRUKTIONER ";
6100 INPUT A
6110 IF A < 0 THEN 1470
6120 PRINT
6130 H8 = 1
6140 ON A + 1 GOTO 6310,6600,6770,7260,6880,6250
6150 PRINT "FØLGENDE FUNKTIONER KAN KALDES:"
6160 PRINT " 0 = SAMMENFATTENDE GALAKTISK RAPPORT"
6170 PRINT " 1 = STATUS RAPPORT"
6180 PRINT " 2 = PHOTON-TORPEDO DATA"
6190 PRINT " 3 = NAVIGATIONSDATA TIL RUMBASE"
6200 PRINT " 4 = KURS/AFSTANDS-BEREGRNER"
6210 PRINT " 5 = KORT OER GALAKSEN MED OMRÅDENAVNE"
6220 PRINT
6230 GOTO 6090
6240 REM Galaktisk rapport
6250 H8 = 0:G5 = 1
6260 PRINT "          GALAKSEN"
6270 PRINT "          ====="
6280 PRINT
6290 GOTO 6350
6300 REM Sammenfattende galaktisk rapport
6310 PRINT
6320 PRINT "          ";
6330 PRINT "STYREDATAMAT RAPPORT OVER KVADRANT";Q1;" ";Q2
6340 PRINT
6350 PRINT "          1      2      3      4      5      6      7      8"
6360 PRINT "          -----"
6370 FOR I = 1 TO 8
6380 PRINT I;
6390 IF H8 = 0 THEN 6460
6400 FOR J = 1 TO 8
6410 PRINT " ";
6420 IF Z(I,J) = 0 THEN PRINT "****"; GOTO 6440
6430 PRINT RIGHT$( STR$( Z(I,J) + 1000),3);
6440 NEXT J
6450 GOTO 6540
6460 Z4 = I:Z5 = 1
6470 GOSUB 7620
6480 JO = INT (15 - .5 * LEN (G2$))
6490 PRINT TAB( JO);G2$;
6500 Z5 = 5
6510 GOSUB 7620
6520 JO = INT (39 - .5 * LEN (G2$))
6530 PRINT TAB( JO);G2$;
6540 PRINT
6550 PRINT "          -----"
6560 NEXT I
6570 PRINT
6580 GOTO 1470
6590 REM Status rapport
6600 PRINT "***** STATUS RAPPORT *****"
6610 PRINT
6620 X$ = ""
6630 IF K9 > 1 THEN X$ = "ER"
6640 PRINT "KLINGON";X$;" TILBAGE";K9
6650 PRINT "MISSIONEN SKAL FULDFØRES I LØBET AF";
6660 PRINT .1 * INT ((TO + T9 - T) * 10);"STJERNEDAGE."
6670 X$ = "ER"
6680 IF B9 < 2 THEN X$ = "E": IF B9 < 1 THEN 6730
6690 PRINT "PLANET-FØDERATIONEN HAR";B9;"RUMBAS";X$;
6700 PRINT " I GALAKSEN."
6710 X$ = ""
6720 GOTO 4700
6730 PRINT "PR GRUND AF DIN EGEN DUMHED ER DU ALENE I"
6740 PRINT "GALAKSEN -- DU HAR INGEN RUMBASER TILBAGE!"
6750 GOTO 4700
6760 REM Torpedo, rumbase-nav, retning/afstand beregning
6770 IF K3 < = 0 THEN 3410
6780 X$ = ""
6790 IF K3 > 1 THEN X$ = "E"
6800 PRINT "FRA ENTERPRISE TIL KLINGON KRIGSRUMSKIB";X$
6810 PRINT
6820 H8 = 0
6830 FOR I = 1 TO 3
6840 IF K(I,3) < = 0 THEN 7240
6850 W1 = K(I,1):X = K(I,2)
6860 C1 = S1:A = S2
6870 GOTO 6990
6880 PRINT "KURS/AFSTAND BEREGRNER:"
6890 PRINT
6900 PRINT "DU ER I KVADRANT";Q1;" ";Q2;
6910 PRINT "SEKTOR";S1;" ";S2;"."
6920 PRINT
6930 PRINT "INDTAST VENLIGST FØLGENDE:"
6940 PRINT " START-KOORDINATER (X,Y) ";
6950 INPUT C1,A
6960 PRINT " SLUT-KOORDINATER (X,Y) ";
6970 INPUT W1,X
6980 PRINT
6990 X = X - A:A = C1 - W1
7000 IF X < 0 THEN 7120
7010 IF A < 0 THEN 7140
7020 IF X > 0 THEN 7040
7030 IF A = 0 THEN C1 = 5: GOTO 7050
7040 C1 = 1
7050 IF ABS (A) < = ABS (X) THEN 7090
7060 PRINT " KURS =";
7070 PRINT C1 + ((( ABS (A) - ABS (X)) + ABS (A)) / ABS (A))
7080 GOTO 7210
7090 PRINT " KURS =";
7100 PRINT C1 + ( ABS (A) / ABS (X))
7110 GOTO 7210
7120 IF A > 0 THEN C1 = 3: GOTO 7150
7130 IF X < > 0 THEN C1 = 5: GOTO 7050
7140 C1 = 7
7150 IF ABS (A) > = ABS (X) THEN 7190
7160 PRINT " KURS =";
7170 PRINT C1 + ((( ABS (X) - ABS (A)) + ABS (X)) / ABS (X))
7180 GOTO 7210
7190 PRINT " KURS =";
7200 PRINT C1 + ( ABS (X) / ABS (A))
7210 PRINT "AFSTAND ="; SQR (X ^ 2 + A ^ 2) / L6
7220 L6 = 1
7230 IF H8 = 1 THEN 1470
7240 NEXT I
7250 GOTO 1470
7260 IF B3 < > 0 THEN 7650
7270 PRINT "HR. SPOCK RAPPORTERER: 'SENSORERNE VISER INGEN RUMBASER"
7280 PRINT " I DENNE KVADRANT'."
7290 GOTO 1470
7300 REM Find fri plads i kvadranten hvor ting kan placeres
7310 R1 = FN R(1):R2 = FN R(1)
7320 A$ = " "
7330 Z1 = R1:Z2 = R2
7340 GOSUB 7540
7350 IF Z3 = 0 THEN 7310
7360 RETURN
7370 REM Indfør kvadrant-data i liste
7380 S8 = INT (Z2 - .5) * 3 + INT (Z1 - .5) * 24 + 1
7390 IF LEN (A$) < > 3 THEN PRINT "PROGRAMFEJL - PROGRAM STOPPET!":
GOTO 7930
7400 IF S8 = 1 THEN Q$ = A$ + RIGHT$( Q$,189): RETURN
7410 IF S8 = 190 THEN Q$ = LEFT$( Q$,189) + A$: RETURN
7420 Q$ = LEFT$( Q$,S8 - 1) + A$ + RIGHT$( Q$,190 - S8): RETURN
7430 REM Udskriv enhedens navn
7440 ON R1 GOTO 7450,7460,7470,7480,7490,7500,7510,7520
7450 G2$ = "KURSMOTORER": RETURN
7460 G2$ = "NÆR-SENSOR SCANNER": RETURN
7470 G2$ = "FJERN-SENSOR SCANNER": RETURN
7480 G2$ = "STYRING AF PHASER-KANONER": RETURN
7490 G2$ = "STYRING AF PHOTON-TORPEDOER": RETURN
7500 G2$ = "SKADESRAPPORT": RETURN
7510 G2$ = "STYRING AF BESKYTTELSESSKJOLD": RETURN
7520 G2$ = "STYRECOMPUTER": RETURN
7530 REM Sammenligning med kvadrantliste
7540 Z1 = INT (Z1 + .5):Z2 = INT (Z2 + .5)
7550 S8 = (Z2 - 1) * 3 + (Z1 - 1) * 24 + 1
7560 Z3 = 0
7570 IF MID$( Q$,S8,3) < > A$ THEN RETURN
7580 Z3 = 1

```

```

7590 RETURN
7600 REM Kvadrantnavn i G2$ fra Z4,Z5 (=Q1,Q2)
7610 REM Kaldes med G5=1 hvis kun områdenavn ønskes
7620 IF Z5 < = 4 THEN ON Z4 GOTO
7700,7710,7720,7730,7740,7750,7760,7770
7630 GOTO 7780
7640 REM Navigations data
7650 PRINT "FRA ENTERPRISE TIL RUMBASE:"
7660 W1 = B4:X = B5:L6 = 8
7670 PRINT
7680 W1 = B4:X = B5:L6 = 8
7690 GOTO 6860
7700 G2$ = "ANTARES": GOTO 7870
7710 G2$ = "RIGEL": GOTO 7870
7720 G2$ = "PROCYON": GOTO 7870
7730 G2$ = "WEGA": GOTO 7870
7740 G2$ = "CANOPUS": GOTO 7870
7750 G2$ = "ALTAIR": GOTO 7870
7760 G2$ = "SAGITTARIUS": GOTO 7870
7770 G2$ = "POLLUX": GOTO 7870
7780 ON Z4 GOTO 7790,7800,7810,7820,7830,7840,7850,7860
7790 G2$ = "SIRIUS": GOTO 7870
7800 G2$ = "DENEBO": GOTO 7870
7810 G2$ = "CAPELLA": GOTO 7870
7820 G2$ = "BETELGEUZE": GOTO 7870
7830 G2$ = "ALDEBARAN": GOTO 7870
7840 G2$ = "REGULUS": GOTO 7870
7850 G2$ = "ARCTURUS": GOTO 7870
7860 G2$ = "SPICA"
7870 IF G5 < > 1 THEN ON Z5 GOTO
7890,7900,7910,7920,7890,7900,7910,7920
7880 RETURN
7890 G2$ = G2$ + " I": RETURN
7900 G2$ = G2$ + " II": RETURN
7910 G2$ = G2$ + " III": RETURN
7920 G2$ = G2$ + " IV": RETURN
7930 END

```



# Synonym

Et synonym til et ord er et andet ord, som på dansk har samme (eller næsten samme) betydning som det første ord. Dette program præsenterer en øvelse i at finde på synonymer til nogle almindelige ord.

Datamaten vælger et ord og spørger dig derefter om et synonym til dette. Du får at vide, om det ord, du indtastede, er rigtigt eller forkert. Hvis du ikke kan komme på noget synonym, kan du taste »HJÆLP«, så vil datamaten give dig et eksempel.

Du kan indføre dine egne ord i DATA-sætningerne (linienumre 800 til 890). Tallet efter DATA i linie 790 angiver det totale antal DATA-sætninger. I hver DATA-sætning angiver det første tal antallet af ord på samme linie.

Kan du finde på en måde at lave dette program om til et mere generelt anvendeligt CAI-program, der kan behandle alle mulige emner?

Walt Koetke ved Lexington High School, Massachusetts skrev dette program.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                        SYNONYM
-----
```

```
ET ORDS SYNONYM ER ET ANDET ORD, DER
BETYDER - ELLER NÆSTEN BETYDER - DET
SAMME SOM DET OPRINDELIGE ORD.
```

```
JEG VÆLGER ET ORD, OG DU SKAL INDSTASTE
ET SYNONYM FOR ORDET.
```

```
HVIS DU IKKE KAN FINDE PÅ ET SYNONYM,
SÅ INDSTAST 'HJÆLP', OG JEG VIL
GIVE DIG ET BRUGBART SYNONYM.
```

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: FORREST --- ?FØRST
```

FINT

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: ENS --- ?HJÆLP
```

ET SYNONYM FOR ENS ER ENSARTET.

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: ENS --- ?LIGNENDE
```

GODT!

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: GLÆDE --- ?LYKKE
```

FINT

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: HUL --- ?HOSPITAL
```

PRØV IGEN.

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: HUL --- ?HJÆLP
```

ET SYNONYM FOR HUL ER FÆLDE.

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: HUL --- ?FÆLDE
```

RIGTIGT

```
HVILKET ANDET ORD ER ET SYNONYM TIL
ORDET: MODEL --- ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                        SYNONYM"
130 PRINT "-----"
140 PRINT
150 DIM R$(5),W$(10),L(30),R(30)
160 R$(1) = "RIGTIGT"
170 R$(2) = "KORREKT"
180 R$(3) = "FINT"
190 R$(4) = "GODT!"
200 R$(5) = "OK!"
210 C = 0
220 PRINT "ET ORDS SYNONYM ER ET ANDET ORD, DER"
230 PRINT "BETYDER - ELLER NÆSTEN BETYDER - DET"
240 PRINT "SAMME SOM DET OPRINDELIGE ORD."
250 PRINT
260 PRINT "JEG VÆLGER ET ORD, OG DU SKAL INDSTASTE"
270 PRINT "ET SYNONYM FOR ORDET."
280 PRINT
290 PRINT "HVIS DU IKKE KAN FINDE PÅ ET SYNONYM,"
300 PRINT "SÅ INDSTAST 'HJÆLP', OG JEG VIL"
310 PRINT "GIVE DIG ET BRUGBART SYNONYM."
320 PRINT
330 RESTORE
340 C = C + 1
350 READ N
360 IF C > N THEN 750
370 N1 = INT ( RND (1) * N + 1)
380 IF R(N1) = 1 THEN 370
390 R(N1) = 1
400 FOR I = 1 TO N1
410 READ N2
420 FOR J = 1 TO N2
430 READ W$(J)
440 NEXT J
450 NEXT I
460 FOR J = 1 TO N2
470 L(J) = J
480 NEXT J
490 L(0) = N2
500 G = 1
510 PRINT
520 L(G) = L(L(0))
530 L(0) = N2 - 1
540 PRINT "HVILKET ANDET ORD ER ET SYNONYM TIL"
550 PRINT "ORDET: ";W$(G);" --- ";
560 INPUT A$
570 PRINT
580 IF A$ = "HJÆLP" THEN 680
590 FOR K = 1 TO N2
600 IF G = K THEN 620
610 IF A$ = W$(K) THEN 660
620 NEXT K
630 PRINT "PRØV IGEN."
640 PRINT
650 GOTO 540
660 PRINT R$( RND (1) * 5 + 1)
670 GOTO 330
680 G1 = INT ( RND (1) * L(0) + 1)
690 PRINT "ET SYNONYM FOR ";W$(G);" ER ";
700 PRINT W$(L(G1));"."
710 PRINT
720 L(G1) = L(L(0))
730 L(0) = L(0) - 1
740 GOTO 540
750 PRINT
760 PRINT "SYNONYMTESTEN ER OVERSTÆT."
770 PRINT
780 GOTO 900
790 DATA 10
800 DATA 5, "FORREST", "FORAN", "FREMME", "FØRST",
"SPIDSEN"
810 DATA 5, "ENS", "IDENTISK", "SAMME", "ENSARTET",
"LIGNENDE"
820 DATA 5, "MODEL", "FORBILLEDE", "MØNSTER",
"BILLEDE", "FORLÆG"
830 DATA 5, "LILLE", "KORT", "BITTE", "LIDEN", "FIN"
840 DATA 6, "STOP", "HOLDT", "STANDS", "PAUSE",
"AFBRYDELSE", "HVIL"
850 DATA 6, "HUS", "RESIDENS", "DOMICIL", "BOLIG",
"BYGNING", "EJENDOM"
860 DATA 5, "HUL", "GRUBE", "SÆNKNING", "GRAV",
"FÆLDE"
870 DATA 5, "STEMME", "RØST", "TALEGAVE", "LYD",
"STONE"
880 DATA 6, "GLÆDE", "BEGEJSTRING", "FORNØJE",
"HENRYKKE", "LYKKE", "GAMMEN"
890 DATA 5, "SMERTE", "KVAL", "PINE", "SVIE", "VE"
900 END
```

# Sænke slagskibe


Reglerne forklares ikke i programmet, så læs nedenstående spillebeskrivelse grundigt.

Sænke slagskibe spilles på et kvadreret net, der indeholder 10 gange 10 felter i et retvinklet koordinatsystem (x,y). Hver spiller har fire skibe: slagskib (5 felter), krydser (3 felter) og 2 destroyere (2 felter hver). Disse skibe kan placeres på det kvadrerede net enten vandret, lodret eller diagonalt, men de må ikke ligge oven i hinanden. Skibene skifter ikke plads under spillet.

Hver af spillerne placerer sine skibe på nettet, uden at modstanderen ser hvor. Spillerne skiftes nu til at skyde på hinandens net i et forsøg på at sænke modstanderens skibe først. Så længe alle felter i slagskibet ikke er ramt, har spilleren 3 skud, for krydserens vedkommende 2 skud og for destroyerne 1 skud. I begyndelsen af spillet har hver spiller derfor  $3+2+1+1=7$  skud. Spilleren indtaster alle sine skud på en gang, og datamaten fortæller, hvilke af dens skibe, der er ramt. Skuddene indtastes som et koordinatsæt (x,y). Den spiller, som sænker alle modstanderens skibe har vundet.

Vigtig note: Dine skibe og datamatens skibe er selvfølgelig placeret på to forskellige net, der hver måler 10 gange 10 felter.

Programmet er skrevet af Lawrence Siegel, Shaker Heights, Ohio.



MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

## SÆNKE SLAGSKIBE

INDTAST KOORDINATERNE FOR...

SLAGSKIBE (5 FELTER)

KOORDINAT ?10,10  
 KOORDINAT ?9,9  
 KOORDINAT ?8,8  
 KOORDINAT ?7,7  
 KOORDINAT ?6,6

KRYDSE (3 FELTER)

KOORDINAT ?3,5  
 KOORDINAT ?2,6  
 KOORDINAT ?1,7

A-DESTROYER (2 FELTER)

KOORDINAT ?1,10  
 KOORDINAT ?2,10

B-DESTROYER (2 FELTER)

KOORDINAT ?6,7  
 KOORDINAT ?6,8

ØNSKER DU AT BEGYNDE ?JA

VIL DU SE MINE SKUD ?JA

OMGANG 1

DU HAR 7 SKUD.

?6,6  
 ?6,7  
 ?10,10  
 ?10,9  
 ?8,7  
 ?1,1  
 ?4,4

DU HAR RAMT MIT SLAGSKIB.

DU HAR RAMT MIN A-DESTROYER.

JEG HAR 7 SKUD.

3,4  
 4,5  
 3,6  
 4,3  
 5,4  
 5,6  
 4,7

OMGANG 2

DU HAR 7 SKUD.

?5,5  
 ?6,1  
 ?6,2  
 ?6,10  
 ?1,10  
 ?6,5  
 ?6,4

DU HAR RAMT MIT SLAGSKIB.

DU HAR RAMT MIT SLAGSKIB.

DU HAR RAMT MIT SLAGSKIB.

JEG HAR 7 SKUD.

8,7  
 9,8  
 8,9  
 9,6  
 10,7  
 10,9  
 9,10

OMGANG 3

DU HAR 7 SKUD.

?6,3  
 ?9,10  
 ?9,9  
 ?8,5  
 ?8,3  
 ?9,5  
 ?9,8

DU HAR RAMT MIT SLAGSKIB.

DU HAR RAMT MIN A-DESTROYER.

JEG HAR 3 SKUD.

6,1  
 7,2  
 6,3

OMGANG 4

DU HAR 7 SKUD.

?10,1  
 ?10,2  
 ?10,3  
 ?10,4  
 ?1,7  
 ?1,6  
 ?1,5

DU HAR RAMT MIN B-DESTROYER.

DU HAR RAMT MIN B-DESTROYER.

JEG HAR 2 SKUD.

8,1  
 9,2

OMGANG 5

DU HAR 7 SKUD.

?2,2  
 ?3,4  
 ?2,5  
 ?2,6  
 ?2,7  
 ?2,8  
 ?1,2

DU HAR RAMT MIN KRYDSE.

DU HAR RAMT MIN KRYDSE.

JEG HAR 2 SKUD.

10,6  
 10,8

OMGANG 6

DU HAR 7 SKUD.

?2,9  
 ?2,10  
 ?3,10  
 ?4,10  
 ?5,10  
 ?9,7  
 ?8,7

DU HAR ALLEREDE SKUDT DER I OMGANG 1

?10,1

DU HAR ALLEREDE SKUDT DER I OMGANG 4

?10,6

DU HAR RAMT MIN KRYDSE.

JEG HAR 0 SKUD.

DU HAR VUNDET.

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          SÅNKE SLAGSKIBE"
130 PRINT "-----"
150 PRINT
160 DIM A(10,10),B(10,10),C(7),D(7)
170 DIM E(12),F(12),G(12),H(12),K(10,10)
180 DEF FN A(K) = (5 - K) * 3 - 2 * INT (K / 4) +
SGN (K - 1) - 1
190 DEF FN B(K) = K + INT (K / 4) - SGN (K - 1)
200 Z8 = 0
210 FOR W = 1 TO 12
220 E(W) = - 1
230 H(W) = - 1
240 NEXT W
250 FOR X = 1 TO 10
260 FOR Y = 1 TO 10
270 B(X,Y) = 0
280 NEXT Y
290 NEXT X
300 FOR X = 1 TO 12
310 F(X) = 0
320 G(X) = 0
330 NEXT X
340 FOR X = 1 TO 10
350 FOR Y = 1 TO 10
360 A(X,Y) = 0
370 NEXT Y
380 NEXT X
390 FOR K = 4 TO 1 STEP - 1
400 U6 = 0
410 GOSUB 2290
420 IF V + V2 + V * V2 = 0 THEN 410
430 IF Y + V * FN B(K) > 10 THEN 410
440 IF Y + V * FN B(K) < 1 THEN 410
450 IF X + V2 * FN B(K) > 10 THEN 410
460 IF X + V2 * FN B(K) < 1 THEN 410
470 U6 = U6 + 1
480 IF U6 > 25 THEN 340
490 FOR Z = 0 TO FN B(K)
500 F(Z + FN A(K)) = X + V2 * Z
510 G(Z + FN A(K)) = Y + V * Z
520 NEXT Z
530 U8 = FN A(K)
540 IF U8 > U8 + FN B(K) THEN 610
550 FOR Z2 = U8 TO U8 + FN B(K)
560 IF U8 < 2 THEN 600
570 FOR Z3 = 1 TO U8 - 1
580 IF SQR ((F(Z3) - F(Z2)) ^ 2 + (G(Z3) - G(Z2)) ^
2) < 3.59 THEN 410
590 NEXT Z3
600 NEXT Z2
610 FOR Z = 0 TO FN B(K)
620 A(F(Z + U8),G(Z + U8)) = .5 + SGN (K - 1) * (K -
1.5)
630 NEXT Z
640 NEXT K
650 PRINT "INDTAST KOORDINATERNE FOR..."
660 PRINT
670 PRINT "SLAGSKIBE (5 FELTER)"
680 FOR X = 1 TO 5
690 PRINT "KOORDINAT ";
700 INPUT Y,Z
710 B(Y,Z) = 3
720 NEXT X
730 PRINT
740 PRINT "KRYDSER (3 FELTER)"
750 FOR X = 1 TO 3
760 PRINT "KOORDINAT ";
770 INPUT Y,Z
780 B(Y,Z) = 2
790 NEXT X
800 PRINT
810 PRINT "A-DESTROYER (2 FELTER)"
820 FOR X = 1 TO 2
830 PRINT "KOORDINAT ";
840 INPUT Y,Z
850 B(Y,Z) = 1
860 NEXT X
870 PRINT
880 PRINT "B-DESTROYER (2 FELTER)"
890 FOR X = 1 TO 2
900 PRINT "KOORDINAT ";
910 INPUT Y,Z
920 B(Y,Z) = .5
930 NEXT X
940 PRINT
950 PRINT "ØNSKER DU AT BEGYNDE ";

```

```

960 INPUT J$
970 PRINT
980 IF J$ < > "HVOR" THEN 1180
990 PRINT "SLAGSKIB"
1000 FOR Z = 1 TO 5
1010 PRINT F(Z);",";G(Z)
1020 NEXT Z
1030 PRINT
1040 PRINT "KRYDSER"
1050 PRINT F(6);",";G(6)
1060 PRINT F(7);",";G(7)
1070 PRINT F(8);",";G(8)
1080 PRINT
1090 PRINT "A-DESTROYER"
1100 PRINT F(9);",";G(9)
1110 PRINT F(10);",";G(10)
1120 PRINT
1130 PRINT "B-DESTROYER"
1140 PRINT F(11);",";G(11)
1150 PRINT F(12);",";G(12)
1160 PRINT
1170 GOTO 950
1180 C = 0
1190 PRINT "VIL DU SE MINE SKUD ";
1200 INPUT K$
1210 IF LEFT$(J$,1) < > "J" THEN 1950
1220 REM Start
1230 IF LEFT$(J$,1) < > "J" THEN 1280
1240 C = C + 1
1250 PRINT
1260 PRINT "OMGANG";C
1270 PRINT "-----"
1280 A = 0
1290 FOR W = .5 TO 3 STEP .5
1300 FOR X = 1 TO 10
1310 FOR Y = 1 TO 10
1320 IF B(X,Y) = W THEN 1360
1330 NEXT Y
1340 NEXT X
1350 GOTO 1370
1360 A = A + INT (W + .5)
1370 NEXT W
1380 FOR W = 1 TO 7
1390 C(W) = 0
1400 D(W) = 0
1410 F(W) = 0
1420 G(W) = 0
1430 NEXT W
1440 P3 = 0
1450 FOR X = 1 TO 10
1460 FOR Y = 1 TO 10
1470 IF A(X,Y) > 10 THEN 1490
1480 P3 = P3 + 1
1490 NEXT Y
1500 NEXT X
1510 PRINT "DU HAR";A;"SKUD."
1520 IF P3 > = (A) THEN 1570
1530 PRINT
1540 PRINT "DU HAR FLERE SKUD END DER ER"
1550 PRINT
1560 GOTO 2260
1570 IF A < > 0 THEN 1610
1580 PRINT
1590 PRINT "JEG HAR VUNDET."
1600 GOTO 3640
1610 FOR W = 1 TO A
1620 INPUT X,Y
1630 IF X < > INT (X) THEN 1690
1640 IF X > 10 THEN 1690
1650 IF X < 1 THEN 1690
1660 IF Y < > INT (Y) THEN 1690
1670 IF Y > 10 THEN 1690
1680 IF Y > = 1 THEN 1710
1690 PRINT "IKKE GYLDIGT....PRØV IGEN."
1700 GOTO 1620
1710 IF A(X,Y) > 10 THEN 1760
1720 C(W) = X
1730 D(W) = Y
1740 NEXT W
1750 GOTO 1790
1760 PRINT "DU HAR ALLEREDE SKUDT DER I OMGANG";
1770 PRINT A(X,Y) - 10
1780 GOTO 1620
1790 FOR W = 1 TO A
1800 IF A(C(W),D(W)) = 3 THEN 1870
1810 IF A(C(W),D(W)) = 2 THEN 1890
1820 IF A(C(W),D(W)) = 1 THEN 1910
1830 IF A(C(W),D(W)) = .5 THEN 1930

```

```

1840 A(C(W),D(W)) = 10 + C
1850 NEXT W
1860 GOTO 1950
1870 PRINT "DU HAR RAMT MIT SLAGSKIB."
1880 GOTO 1840
1890 PRINT "DU HAR RAMT MIN KRYDSER."
1900 GOTO 1840
1910 PRINT "DU HAR RAMT MIN A-DESTROYER."
1920 GOTO 1840
1930 PRINT "DU HAR RAMT MIN B-DESTROYER."
1940 GOTO 1840
1950 A = 0
1960 IF LEFT$(J$,1) = "J" THEN 2010
1970 C = C + 1
1980 PRINT
1990 PRINT "OMGANG";C
2000 PRINT "-----"
2010 A = 0
2020 FOR W = .5 TO 3 STEP .5
2030 FOR X = 1 TO 10
2040 FOR Y = 1 TO 10
2050 IF A(X,Y) = W THEN 2090
2060 NEXT Y
2070 NEXT X
2080 GOTO 2100
2090 A = A + INT (W + .5)
2100 NEXT W
2110 P3 = 0
2120 FOR X = 1 TO 10
2130 FOR Y = 1 TO 10
2140 IF A(X,Y) > 10 THEN 2160
2150 P3 = P3 + 1
2160 NEXT Y
2170 NEXT X
2180 PRINT "JEG HAR";A;"SKUD."
2190 IF P3 > (A) THEN 2250
2200 PRINT
2210 PRINT "JEG HAR FLERE SKUD END DER ER"
2220 PRINT "FRIE FELTER."
2230 PRINT
2240 GOTO 1580
2250 IF A < > 0 THEN 2340
2260 PRINT
2270 PRINT "DU HAR VUNDET."
2280 GOTO 3640
2290 X = INT ( RND (1) * 10 + 1)
2300 Y = INT ( RND (1) * 10 + 1)
2310 V = INT (3 * RND (1) - 1)
2320 V2 = INT (3 * RND (1) - 1)
2330 RETURN
2340 FOR W = 1 TO 12
2350 IF H(W) > 0 THEN 3180
2360 NEXT W
2370 REM Tilfældighed
2380 W = 0
2390 R3 = 0
2400 GOSUB 2290
2410 RESTORE
2420 R2 = 0
2430 R3 = R3 + 1
2440 IF R3 > 100 THEN 2390
2450 IF X > 10 THEN 2490
2460 IF X > 0 THEN 2500
2470 X = 1 + INT ( RND (1) * 2.5)
2480 GOTO 2500
2490 X = 10 - INT ( RND (1) * 2.5)
2500 IF Y > 10 THEN 2540
2510 IF Y > 0 THEN 2650
2520 Y = 1 + INT ( RND (1) * 2.5)
2530 GOTO 2650
2540 Y = 10 - INT ( RND (1) * 2.5)
2550 GOTO 2650
2560 F(W) = X
2570 G(W) = Y
2580 IF W = (A) THEN 2760
2590 IF R2 = 6 THEN 2410
2600 READ X1,Y1
2610 R2 = R2 + 1
2620 DATA 1,1,-1,1,1,-3,1,1,0,2,-1,1
2630 X = X + X1
2640 Y = Y + Y1
2650 IF X > 10 THEN 2590
2660 IF X < 1 THEN 2590
2670 IF Y > 10 THEN 2590
2680 IF Y < 1 THEN 2590
2690 IF B(X,Y) > 10 THEN 2590
2700 FOR Q9 = 1 TO W
2710 IF F(Q9) < > X THEN 2730
2720 IF G(Q9) = Y THEN 2590
2730 NEXT Q9
2740 W = W + 1
2750 GOTO 2560
2760 IF LEFT$(K$,1) < > "J" THEN 2800
2770 FOR Z5 = 1 TO A
2780 PRINT F(Z5);";";G(Z5)
2790 NEXT Z5
2800 FOR W = 1 TO A
2810 IF B(F(W),G(W)) = 3 THEN 2880
2820 IF B(F(W),G(W)) = 2 THEN 2900
2830 IF B(F(W),G(W)) = 1 THEN 2940
2840 IF B(F(W),G(W)) = .5 THEN 2920
2850 B(F(W),G(W)) = 10 + C
2860 NEXT W
2870 GOTO 1230
2880 PRINT "JEG RAMTE DIT SLAGSKIB."
2890 GOTO 2950
2900 PRINT "JEG RAMTE DIN KRYDSER."
2910 GOTO 2950
2920 PRINT "JEG RAMTE DIN B-DESTROYER."
2930 GOTO 2950
2940 PRINT "JEG RAMTE DIN A-DESTROYER."
2950 FOR Q = 1 TO 12
2960 IF E(Q) < > - 1 THEN 3110
2970 E(Q) = 10 + C
2980 H(Q) = B(F(W),G(W))
2990 M3 = 0
3000 FOR M2 = 1 TO 12
3010 IF H(M2) < > H(Q) THEN 3030
3020 M3 = M3 + 1
3030 NEXT M2
3040 IF M3 < > INT (H(Q) + .5) + 1 + INT ( INT
(H(Q) + .5) / 3) THEN 2850
3050 FOR M2 = 1 TO 12
3060 IF H(M2) < > H(Q) THEN 3090
3070 E(M2) = - 1
3080 H(M2) = - 1
3090 NEXT M2
3100 GOTO 2850
3110 NEXT Q
3120 PRINT "PROGRAMMET AFBRUDT:"
3130 FOR Q = 1 TO 12
3140 PRINT "E(";Q;") =";E(Q)
3150 PRINT "H(";Q;") =";H(Q)
3160 NEXT Q
3170 GOTO 3640
3180 REM Listning
3190 FOR R = 1 TO 10
3200 FOR S = 1 TO 10
3210 K(R,S) = 0
3220 NEXT S
3230 NEXT R
3240 FOR U = 1 TO 12
3250 IF E(U) < 10 THEN 3400
3260 FOR R = 1 TO 10
3270 FOR S = 1 TO 10
3280 IF B(R,S) < 10 THEN 3310
3290 K(R,S) = - 10000000
3300 GOTO 3380
3310 FOR M = SGN (1 - R) TO SGN (10 - R)
3320 FOR N = SGN (1 - S) TO SGN (10 - S)
3330 IF N + M + N * M = 0 THEN 3360
3340 IF B(R + M,S + N) < > E(U) THEN 3360
3350 K(R,S) = K(R,S) + E(U) - S * INT (H(U) + .5)
3360 NEXT N
3370 NEXT M
3380 NEXT S
3390 NEXT R
3400 NEXT U
3410 FOR R = 1 TO A
3420 F(R) = R
3430 G(R) = R
3440 NEXT R
3450 FOR R = 1 TO 10
3460 FOR S = 1 TO 10
3470 Q9 = 1
3480 FOR M = 1 TO A
3490 IF K(F(M),G(M)) > = K(F(Q9),G(Q9)) THEN 3510
3500 Q9 = M
3510 NEXT M
3520 IF R > (A) THEN 3540
3530 IF R = S THEN 3610
3540 IF K(R,S) < K(F(Q9),G(Q9)) THEN 3610
3550 FOR M = 1 TO A
3560 IF F(M) < > R THEN 3590
3570 IF G(M) = S THEN 3610
3580 NEXT M
3590 F(Q9) = R
3600 G(Q9) = S
3610 NEXT S
3620 NEXT R
3630 GOTO 2760
3640 END

```



# Tal

Til forskel fra andre spil hvor man skal gætte et tilfældigt tal, som datamaten har valgt (Gæt, Omkring, Stjerner, etc.), får du til Tal kun ét gæt pr. spil. Du vinder eller taber points, alt efter hvor nær du var på det tilfældige tal, datamaten tænkte på. Nogle gange får du en jackpot, som fordobler dit pointtal. Du har vundet, når du har fået 500 points.

Tom Adameix skrev dette program, da han var elev ved Curtis Junior High School i Sudbury, Massachusetts.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
TAL  
-----
```

```
DU HAR 100 POINT. VED AT GÆTTE TAL, DER  
LIGGER MELLEM 1 OG 5 KAN DU VINDE EL-  
LER TABE POINT, AFHÆNGIG AF, HVOR TÆT  
DU KOMMER PÅ DET TILFÆLDIGE TAL SOM  
DATAMATEN HAR VALGT.
```

```
NOGENGANGE UDBETALES EN 'JACKPOT', HVOR  
DIT POINT-TAL FORDOBLES!!
```

```
DU VINDER NÅR DU HAR 500 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?3  
DU HAR 101 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?2  
DU HAR 96 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?3  
DU HAR 91 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?3  
DU HAR 86 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?3  
DU HAR 91 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?1  
DU HAR 86 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?4  
DU FIK JACKPOT!!!  
DU HAR 172 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?2  
DU HAR 167 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?4  
DU FIK JACKPOT!!!  
DU HAR 334 POINT.
```

```
GÆT ET TAL MELLEM 1 OG 5 ?5  
DU FIK JACKPOT!!!  
!!!!!! DU VINDER !!!!!  
DU HAR 668 POINT.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "TAL"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "DU HAR 100 POINT. VED AT GÆTTE TAL, DER"  
160 PRINT "LIGGER MELLEM 1 OG 5 KAN DU VINDE EL-"  
170 PRINT "LER TABE POINT, AFHÆNGIG AF, HVOR TÆT"  
180 PRINT "DU KOMMER PÅ DET TILFÆLDIGE TAL SOM"  
190 PRINT "DATAMATEN HAR VALGT."  
200 PRINT  
210 PRINT "NOGENGANGE UDBETALES EN 'JACKPOT', HVOR"  
220 PRINT "DIT POINT-TAL FORDOBLES!!"  
230 PRINT  
240 PRINT "DU VINDER NÅR DU HAR 500 POINT."  
250 P = 100  
260 DEF FN R(X) = INT (5 * RND (1) + 1)  
270 PRINT  
280 PRINT "GÆT ET TAL MELLEM 1 OG 5 ";  
290 INPUT G  
300 R = FN R(1)  
310 S = FN R(1)  
320 T = FN R(1)  
330 U = FN R(1)  
340 V = FN R(1)  
350 IF G = R THEN 410  
360 IF G = S THEN 430  
370 IF G = T THEN 450  
380 IF G = U THEN 480  
390 IF G = V THEN 500  
400 IF G > 5 THEN 270  
410 P = P - 5  
420 GOTO 510  
430 P = P + 5  
440 GOTO 510  
450 P = P + P  
460 PRINT "DU FIK JACKPOT!!!"  
470 GOTO 510  
480 P = P + 1  
490 GOTO 510  
500 P = P - (P * .5)  
510 IF P > 500 THEN 540  
520 PRINT "DU HAR";P;"POINT."  
530 GOTO 270  
540 PRINT "!!!!!! DU VINDER !!!!!!"  
550 PRINT "DU HAR";P;"POINT."  
560 END
```

# Terning

Dette er egentlig ikke noget spil. Programmet kaster to terninger et stort antal gange og udskriver derefter hyppighederne for de forskellige kast. Det er interessant at se, hvor mange kast der skal til, før den teoretiske fordeling opnås:

2 1/36 2.777 ...%

3 2/36 5.555 ...%

4 3/36 8.333 ...%

osv.

Daniel Freidus skrev dette program, da han gik i syvende klasse på Harrison Junior-Senior High School, Harrison, New York.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
                    TERNING  
-----
```

```
DETTE PROGRAM EFTERLIGNER KAST MED TO  
TERNINGER.
```

```
DU INDTASTER SELV DET ANTAL GANGE  
DATAMATEN SKAL 'KASTE' TERNINGERNE.
```


```
PAS PÅ, FOR MEGET STORE TAL TAGER OGSÅ  
MEGET LANG TID.....DET GÆLDER ISÅR FOR  
TAL OVER 5000.
```

```
HVOR MANGE KAST ?1000
```

ANTAL ØJNE	ANTAL GANGE
2	31
3	47
4	97
5	105
6	139
7	165
8	137
9	104
10	100
11	47
12	28

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                    TERNING"  
130 PRINT "-----"  
140 PRINT  
150 DIM F(12)  
160 PRINT "DETTE PROGRAM EFTERLIGNER KAST MED TO"  
170 PRINT "TERNINGER."  
180 PRINT  
190 PRINT "DU INDTASTER SELV DET ANTAL GANGE"  
200 PRINT "DATAMATEN SKAL 'KASTE' TERNINGERNE."  
210 PRINT  
220 PRINT "PAS PÅ, FOR MEGET STORE TAL TAGER OGSÅ"  
230 PRINT "MEGET LANG TID.....DET GÆLDER ISÅR FOR"  
240 PRINT "TAL OVER 5000."  
250 FOR Q = 1 TO 12  
260 F(Q) = 0  
270 NEXT Q  
280 PRINT  
290 PRINT "HVOR MANGE KAST ";  
300 INPUT X  
310 FOR S = 1 TO X  
320 A = INT (6 * RND (1) + 1)  
330 B = INT (6 * RND (1) + 1)  
340 R = A + B  
350 F(R) = F(R) + 1  
360 NEXT S  
370 PRINT  
380 PRINT "ANTAL ØJNE","ANTAL GANGE"  
390 FOR V = 2 TO 12  
400 PRINT V,F(V)  
410 NEXT V  
420 PRINT  
430 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";  
440 INPUT Z$  
450 IF LEFT$(Z$,1) = "J" THEN 250  
460 END
```



# Tog

Tog er et program, hvor datamaten fremstiller regnestykker ved hjælp af nogle udgangsværdier, der er valgt tilfældigt, men som har tilknytning til forholdene mellem tid, afstand og hastighed (afstand er jo lig hastighed gange tid). Du indtaster derefter dit svar på problemet, hvorefter datamaten bedømmer dit svar.

Tog er et eksempel på et program skrevet af en elev. Den største morskab (og nytte) får man af at SKRIVE programmer som dette, i modsætning til bare at løse de problemer programmet danner. Byt programmer med andre – du løser deres problemer, og de løser dine!

Programmet blev oprindeligt skrevet i FOCAL af en elev, for at andre i hans klasse skulle bruge det. Det blev indsendt til os af Walt Koetke, Lexington High School, Massachusetts.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
                        TOG  
-----
```

```
TID-HASTIGHEDS ØVELSE.
```

```
EN BIL, DER KØRER MED 53 KM/T  
KAN KØRE EN VIS STRÆKNING PÅ 18  
TIMER MINDRE END ET TOG, DER KØRER  
MED 20 KM/T.
```

```
HVOR LANG TID TAGER TUREN I BIL ?30
```

```
DESVÆRRE, DU VAR 64 %  
VED SIDEN AF.
```

```
DET RIGTIGE SVAR ER 10.9090909 TIMER.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
EN BIL, DER KØRER MED 41 KM/T  
KAN KØRE EN VIS STRÆKNING PÅ 18  
TIMER MINDRE END ET TOG, DER KØRER  
MED 37 KM/T.
```

```
HVOR LANG TID TAGER TUREN I BIL ?75
```

```
DESVÆRRE, DU VAR 122 %  
VED SIDEN AF.
```

```
DET RIGTIGE SVAR ER 166.5 TIMER.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
EN BIL, DER KØRER MED 53 KM/T  
KAN KØRE EN VIS STRÆKNING PÅ 13  
TIMER MINDRE END ET TOG, DER KØRER  
MED 38 KM/T.
```

```
HVOR LANG TID TAGER TUREN I BIL ?50
```

```
DESVÆRRE, DU VAR 34 %  
VED SIDEN AF.
```

```
DET RIGTIGE SVAR ER 32.9333334 TIMER.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
EN BIL, DER KØRER MED 40 KM/T  
KAN KØRE EN VIS STRÆKNING PÅ 12  
TIMER MINDRE END ET TOG, DER KØRER  
MED 28 KM/T.
```

```
HVOR LANG TID TAGER TUREN I BIL ?15
```

```
DESVÆRRE, DU VAR 87 %  
VED SIDEN AF.
```

```
DET RIGTIGE SVAR ER 28 TIMER.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                        TOG"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "TID-HASTIGHEDS ØVELSE."  
160 PRINT  
170 C = INT (25 * RND (1)) + 40  
180 D = INT (15 * RND (1)) + 5  
190 T = INT (19 * RND (1)) + 20  
200 PRINT "EN BIL, DER KØRER MED";C;"KM/T"  
210 PRINT "KAN KØRE EN VIS STRÆKNING PÅ";D  
220 PRINT "TIMER MINDRE END ET TOG, DER KØRER"  
230 PRINT "MED";T;"KM/T."  
240 PRINT  
250 PRINT "HVOR LANG TID TAGER TUREN I BIL ";  
260 INPUT A  
270 PRINT  
280 V = D * T / (C - T)  
290 E = INT ( ABS ((V - A) * 100 / A) + .5)  
300 IF E > 5 THEN 330  
310 PRINT "FINT! SVARET LIGGER INDEN FOR";E;"%."  
320 GOTO 360  
330 PRINT "DESVÆRRE, DU VAR";E;"%"  
340 PRINT "VED SIDEN AF."  
350 PRINT  
360 PRINT "DET RIGTIGE SVAR ER";V;"TIMER."  
370 PRINT  
380 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";  
390 INPUT AS  
400 PRINT  
410 IF LEFTS (AS,1) = "J" THEN 170  
420 END
```

# Tyrefægtning

I denne simulerede tyrefægtning er du matadoren, dvs. Den som har hovedrollen, og som skal dræbe tyren, eller som selv bliver dræbt (eller stikker af fra arenaen).

Hver gang tyren passerer dig, kan du forsøge et af følgende:

- 0 = Farlig bevægelse med indersiden af kappen (Veronica)
- 1 = Mindre farlig bevægelse med ydersiden af kappen
- 2 = Almindeligt sving med kappen

Du kan også forsøge at dræbe tyren:

- 4 = Over hornene
- 5 = I brystkassen

Publikum bestemmer, hvilken belønning du fortjener, måske posthumt, hvis det bliver nødvendigt. Jo mere mod du udviser, jo større belønning får du. Det kan også være ganske morsomt at overleve. Jo bedre indsats toreadorerne og picadorerne gør, jo større chancer har du for at vinde.

David Sweet fra Dartmouth skrev den oprindelige version af programmet. Senere er det blevet modificeret af elever ved Lexington High School og til slut af Steve North fra Creative Computing.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                TYREFÆGTNING
-----
```

```
ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J
```

```
GODAFTEN ALLE TILHÆNGERE AF BLODSUDGY-
DELSER. NU HAR DU EN ENESTRENDE CHANCE
FOR AT DRÆBE EN TYR.
```

```
HVERGANG TYREN PASSERER DIG, KAN DU
ENTEN FORSØGE:
```

```
0=VERONICA (FARLIG BEVÆGELSE MED
INDERSIDEN AF KAPPEN).
```

```
1=MINDRE FARLIG BEVÆGELSE MED YDER-
SIDEN AF KAPPEN.
```

```
2=ALMINDELIGT SVING MED KAPPEN.
```

```
<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?
```

```
DU KAN OGSÅ FORSØGE AT DRÆBE TYREN:
```

```
4=OVER HORNENE      5=I BRYSTKASSEN
```

```
MEN HVIS JEG VAR DIG, SÅ VILLE JEG
VENTE TIL EFTER SYVENDE PASSAGE.
```

```
TILSKUERNE BESTEMMER HVILKEN BELØN-
NING DU SKAL HAVE. OGSÅ POSTHUMT, HVIS
DET SKULLE BLIVE NØDVENDIGT!
JO MERE MOD, JO STØRRE BELØNNING.
```

```
JO BEDRE PICADORERNE OG TOREADORERNE
UDFØRER DERES HVERV, JO STØRRE ER DIN
CHANCE FOR AT LYKKES.
```

```
<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?
```

```
DU HAR TRUKKET EN GOD TYR.
```

```
PICADORERNE HAR GJORT ET GOD JOB.
```

```
TOREADORERNE HAR GJORT ET ELENDIG JOB.
1 TOREADORERNE BLEV DRÆBT.
```

```
PASSAGE NUMMER 1
```

```
TYREN ANGRIBER DIG! DU ER TOREADOREN...
```

```
VIL DU DRÆBE TYREN ?N
```

```
HVILKEN KAPPEBEVÆGELSE VIL DU GØRE ?2
```

```
PASSAGE NUMMER 2
```

```
TYREN ANGRIBER DIG! DU ER TOREADOREN...
```

```
VIL DU DRÆBE TYREN ?1
```

```
FORKERT SVAR....TAST 'N' ELLER 'J'.
```

```
?N
```

```
HVILKEN KAPPEBEVÆGELSE VIL DU GØRE ?0
```

```
PASSAGE NUMMER 3
```

```
NU KOMMER TYREN.
FORSØGER DU AT DRÆBE DEN ?J
```

```
KAPPEBEVÆGELSE?0
TYREN HAR SÆRET DIG.....
MEN DU ER STADIG I LIVE.
VIL DU STIKKE AF FRA ARENAEN ?J
```

```
DIN £$^&* KUJON!
```

```
TILSKUERNE RÆBER 'ØV' I 10 MINUTTER..
HVIS DU VISER DIG I EN TYREFÆGTERARENA
IGEN, VIL DE SLÅ DIG IHJEL...ALTSÅ FØR
TYREN GØR DET!
```

```
ADIOS AMIGO.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                TYREFÆGTNING"
130 PRINT "-----"
140 DEF FN A(K) = INT ( RND (1) * 2 + 1)
150 DEF FN C(Q) = FN D(Q) * RND (1)
160 DEF FN D(Q) = (4.5 + L / 6 - (D(1) + D(2)) *
2.5 + 4 * D(4) + 2 * D(5) - D(3) ^ 2 / 120 - A)
170 PRINT
180 L = 1
190 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
200 INPUT Z$
210 PRINT
220 IF LEFT$(Z$,1) = "N" THEN 590
230 PRINT "GODAFTEN ALLE TILHÆNGERE AF BLODSUDGY--"
240 PRINT "DELSER. NU HAR DU EN ENESTRENDE CHANCE"
250 PRINT "FOR AT DRÆBE EN TYR."
260 PRINT
270 PRINT "HVERGANG TYREN PASSERER DIG, KAN DU"
280 PRINT "ENTEN FORSØGE:"
290 PRINT
300 PRINT "0=VERONICA (FARLIG BEVÆGELSE MED"
310 PRINT "  INDERSIDEN AF KAPPEN)."
```

```

960 PRINT
970 PRINT "KAPPEBEVÆGELSE";
980 GOTO 1050
990 PRINT "TYREN ANGRIBER DIG! DU ER TOREADOREN..."
1000 PRINT
1010 PRINT "VIL DU DRÆBE TYREN ";
1020 GOSUB 2160
1030 IF Z1 = 1 THEN 1400
1040 PRINT "HVILKEN KAPPEBEVÆGELSE VIL DU GØRE ";
1050 INPUT E
1060 IF E < > INT ( ABS ( E ) ) THEN 1080
1070 IF E < 3 THEN 1110
1080 PRINT "TAG DET ROLIGT, IDIOT!"
1090 PRINT "INDTAST ET RIGTIGT TAL!"
1100 GOTO 1050
1110 IF E = 0 THEN 1170
1120 IF E = 1 THEN 1150
1130 M = .5
1140 GOTO 1180
1150 M = 2
1160 GOTO 1180
1170 M = 3
1180 L = L + M
1190 F = ( 6 - A + M / 10 ) * RND ( 1 ) / ((D(1) + D(2)
+ D(3) / 10) * 5)
1200 IF F < .51 THEN 870
1210 PRINT "TYREN HAR SÆRET DIG....."
1220 ON FN A(O) GOTO 1230,1260
1230 PRINT "OG DU ER DØD!"
1240 D(4) = 1.5
1250 GOTO 1570
1260 PRINT "MEN DU ER STADIG I LIVE."
1270 PRINT "VIL DU STIKKE AF FRA ARENAEN ";
1280 GOSUB 2160
1290 IF Z1 = 2 THEN 1330
1300 PRINT "DIN £$*&* KUJON!"
1310 D(4) = 0
1320 GOTO 1570
1330 PRINT "DU ER MODIG."
1340 PRINT "MEGET MODIG, MEN OGSÅ MEGET DUM."
1350 ON FN A(O) GOTO 1360,1380
1360 D(4) = 2
1370 GOTO 870
1380 PRINT "DU BLEV SÆRET IGEN."
1390 GOTO 1220
1400 Z = 1
1410 PRINT "SANDHEDENS ØJEBLIK ER KOMMET."
1420 PRINT "HVORDAN VIL DU DRÆBE TYREN ";
1430 INPUT H
1440 IF H = 4 THEN 1490
1450 IF H = 5 THEN 1490
1460 PRINT "DER GIK PANIK I DIG....."
1470 PRINT "TYREN HAR SÆRET DIG..."
1480 GOTO 1220
1490 K = ( 6 - A ) * 10 * RND ( 1 ) / ((D(1) + D(2)) * 5
* D(3))
1500 IF J = 4 THEN 1550
1510 IF K > .2 THEN 1210
1520 PRINT "DU HAR DRÆBT TYREN!"
1530 D(5) = 2
1540 GOTO 1580
1550 IF K > .8 THEN 1210
1560 GOTO 1520
1570 PRINT
1580 PRINT
1590 IF D(4) < > 0 THEN 1650
1600 PRINT "TILSKUERNE RÅBER 'ØV' I 10 MINUTTER.."
1610 PRINT "HVIS DU VISER DIG I EN TYREFÅGTERARENA"
1620 PRINT "IGEN, VIL DE SLÅ DIG IHJEL...ALTSÅ FØR"
1630 PRINT "TYREN GØR DET!"
1640 GOTO 1820
1650 IF D(4) < > 2 THEN 1680
1660 PRINT "TILSKUERNE RÅBER I BEGEJSTRING!"
1670 GOTO 1700
1680 IF D(5) < > 2 THEN 1700
1690 PRINT "TILSKUERNE RÅBER"
1700 PRINT "TILSKUERNE TILKENDER DIG:"
1710 IF FN C(Q) < 2.4 THEN 1810
1720 IF FN C(Q) < 4.9 THEN 1790
1730 IF FN C(Q) < 7.4 THEN 1760
1740 PRINT "OLE!! DU ER 'MUY HOMBRE'!! OLE! OLE!"
1750 GOTO 1820
1760 PRINT "BEGGE TYRENS ØRER."
1770 PRINT "OLE!"
1780 GOTO 1820
1790 PRINT "TYRENS ENE ØRE."
1800 GOTO 1820
1810 PRINT "INGENTING OVERHOVEDET."

```

```

1820 PRINT
1830 PRINT "ADIOS AMIGO.."
1840 GOTO 2270
1850 B = 3 / A * RND ( 1 )
1860 IF B < .37 THEN 1980
1870 IF B < .5 THEN 1960
1880 IF B < .63 THEN 1940
1890 IF B < .87 THEN 1920
1900 C = .1
1910 GOTO 1990
1920 C = .2
1930 GOTO 1990
1940 C = .3
1950 GOTO 1990
1960 C = .4
1970 GOTO 1990
1980 C = .5
1990 T = INT ( 10 * C + .2 )
2000 PRINT A$;" HAR GJORT ET ";L$(T);" JOB."
2010 IF T < 4 THEN 2140
2020 IF T = 5 THEN 2110
2030 ON FN A(K) GOTO 2070,2090
2040 IF A$ = "TOREADORERNE" THEN 2060
2050 PRINT "EN AF ";A$;"S HESTE BLEV DRÆBT."
2060 ON FN A(K) GOTO 2070,2090
2070 PRINT "EN AF ";A$;" BLEV DRÆBT."
2080 GOTO 2140
2090 PRINT "INGEN AF ";A$;" BLEV DRÆBT."
2100 GOTO 2140
2110 IF A$ = "TOREADORERNE" THEN 2130
2120 PRINT FN A(K);"AF ";A$;"S HESTE BLEV DRÆBT."
2130 PRINT FN A(K);A$;" BLEV DRÆBT."
2140 PRINT
2150 RETURN
2160 INPUT A$
2170 PRINT
2180 IF LEFT$( A$,1 ) = "J" THEN 2230
2190 IF LEFT$( A$,1 ) = "N" THEN 2250
2200 PRINT "FORKERT SVAR....TAST 'N' ELLER 'J'.."
2210 PRINT
2220 GOTO 2160
2230 Z1 = 1
2240 GOTO 2260
2250 Z1 = 2
2260 RETURN
2270 END

```

# Tyveknægt

En enarmet tyveknægt er en spillemaskine, der spiser mønter, lige så hurtigt som du kan fodre den. Når du har indkastet en mønt, trækker du i en arm og sætter derved tre tromler i bevægelse. Hvis tromlerne standser på en sådan måde, at bestemte symboler står på række, får du en vis udbetaling. Den første enarmede tyveknægt hed »Liberty Bell«, og den blev opfundet i 1895 af Charles Fey i San Francisco. Fey nægtede at sælge eller overlade produktionsrettighederne til andre, hvorfor H.S. Mills fra Chicago byggede en lignende, men meget bedre maskine, der hed »Operators Bell«. Det er denne maskine, der har overlevet næsten uændret til i dag.

På »Operators Bell« og andre almindelige tyveknægte findes der 20 symboler på hver tromle, men hver tromle indeholder ikke lige mange symboler (kirsebær, citroner, appelsiner, etc.). Af de 8.000 mulige kombinationer udgør den forventede udbetaling til spilleren 7.049, eller 89,11 kroner for hver 100 kroner som stoppes i maskinen. Det er nogle af de laveste odds på noget kasino-spil overhovedet!

I vores program er gevinstudbetalingen mere liberal, den giver faktisk spilleren en fordel på 11%, dvs. en forventet udbetaling på 111 kroner for hver 100 kroners indsats.

Det oprindelige program er skrevet af Fred Mirabelle og Bob Harper.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

TYVEKNÆGT

DU BEFINDER DIG PÅ ET KASINO, OG DU  
STÅR FORAN EN ENARMET TYVEKNÆGT. DU  
KAN SPILLE FOR ET BELØB MELLEML 1 OG  
100 KRONER.

NÅR DU ØNSKER AT TRÆKKE I ARMEN, SKAL  
DU BLOT TRYKKE <RETURN> NÅR DU HAR  
GJORT DIN INDSATS.

HVAD ER DIN INDSATS ?10

APPELSIN KLOKKE KLOKKE

DOBBELT!!!!  
DU VANDT!!!  
DINE AKTIVER ER PÅ 30 KR.

VIL DU FORTSÆTTE ?J

HVAD ER DIN INDSATS ?10

BLOMME KLOKKE KLOKKE

DOBBELT!!!!  
DU VANDT!!!  
DINE AKTIVER ER PÅ 60 KR.

VIL DU FORTSÆTTE ?J

HVAD ER DIN INDSATS ?30

BAR KLOKKE KLOKKE

DOBBELT!!!!  
DU VANDT!!!  
DINE AKTIVER ER PÅ 150 KR.

VIL DU FORTSÆTTE ?J

HVAD ER DIN INDSATS ?10

CITRON APPELSIN BAR

DU TABTE DIN INDSATS.  
DINE AKTIVER ER PÅ 140 KR.

VIL DU FORTSÆTTE ?J

HVAD ER DIN INDSATS ?100

CITRON KLOKKE CITRON

DU TABTE DIN INDSATS.  
DINE AKTIVER ER PÅ 40 KR.

VIL DU FORTSÆTTE ?J

HVAD ER DIN INDSATS ?30

KIRSEBÆR APPELSIN APPELSIN

DOBBELT!!!!  
DU VANDT!!!  
DINE AKTIVER ER PÅ 130 KR.

VIL DU FORTSÆTTE ?N

DU KAN HENTE DIN GEVINST  
HOS KASSEREREN.

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                TYVEKNÆGT"
130 PRINT "-----"
140 PRINT
150 REM Programmeret af Fred Mirabelle og
160 REM Bob Harper den 29. januar 1973
170 PRINT "DU BEFINDER DIG PÅ ET KASINO, OG DU"
180 PRINT "STÅR FORAN EN ENARMET TYVEKNÆGT. DU"
190 PRINT "KAN SPILLE FOR ET BELØB MELLEML 1 OG"
200 PRINT "100 KRONER."
210 PRINT
220 PRINT "NÅR DU ØNSKER AT TRÆKKE I ARMEN, SKAL"
230 PRINT "DU BLOT TRYKKE <RETURN> NÅR DU HAR"
240 PRINT "GJORT DIN INDSATS."
250 P = 0
260 PRINT
270 PRINT "HVAD ER DIN INDSATS ";
280 INPUT M
290 IF M > 100 THEN 1040
300 IF M < 1 THEN 1060
310 M = INT (M)
320 X = INT (6 * RND (1) + 1)
330 Y = INT (6 * RND (1) + 1)
340 Z = INT (6 * RND (1) + 1)
350 PRINT
360 IF X = 1 THEN 1090
370 IF X = 2 THEN 1110
380 IF X = 3 THEN 1130
390 IF X = 4 THEN 1150
400 IF X = 5 THEN 1170
410 IF X = 6 THEN 1190
420 IF Y = 1 THEN 1210
430 IF Y = 2 THEN 1230
440 IF Y = 3 THEN 1250
450 IF Y = 4 THEN 1270
460 IF Y = 5 THEN 1290
470 IF Y = 6 THEN 1310
480 IF Z = 1 THEN 1330
490 IF Z = 2 THEN 1350
500 IF Z = 3 THEN 1370
510 IF Z = 4 THEN 1390
520 IF Z = 5 THEN 1410
530 IF Z = 6 THEN 1430
540 IF X = Y THEN 700
550 IF X = Z THEN 730
560 IF Y = Z THEN 750
570 PRINT
580 PRINT "DU TABTE DIN INDSATS."
590 P = P - M
600 PRINT "DINE AKTIVER ER PÅ";P;"KR."
610 PRINT
620 PRINT "VIL DU FORTSÆTTE ";
630 INPUT A$
640 IF LEFT$(A$,1) < > "N" THEN 260
650 PRINT
660 IF P < 0 THEN 770
670 IF P = 0 THEN 800
680 IF P > 0 THEN 820
690 GOTO 1520
700 IF Y = Z THEN 850
710 IF Y = 1 THEN 980
720 GOTO 1470
730 IF Z = 1 THEN 980
740 GOTO 560
750 IF Z = 1 THEN 980
760 GOTO 1470
770 PRINT "VÆR SR GOD AT BETALE!"
780 PRINT "LAD DINE PENGE LIGGE VED TERMINALEN."
790 GOTO 1520
800 PRINT "HOVSA. DET GIK LIGE OP!"
810 GOTO 1520
820 PRINT "DU KAN HENTE DIN GEVINST"
830 PRINT "HOS KASSEREREN."
840 GOTO 1520
850 IF Z = 1 THEN 920
860 PRINT
870 PRINT "**** TOP GEVINST ****"
880 PRINT
890 PRINT "DU FIK GEVINST!"
900 P = (((10 * M) + M) + P)
910 GOTO 600
920 PRINT
930 PRINT "**** JACKPOT ****"
940 PRINT
950 PRINT "DU FIK GEVINST!"
960 P = (((100 * M) + M) + P)
970 GOTO 600

```

```

980 PRINT
990 PRINT "*** DOBBELT BAR ***"
1000 PRINT
1010 PRINT "DU FIK GEVINST!"
1020 P = ((5 * M) + M) + P)
1030 GOTO 600
1040 PRINT "HUSETS MAKSIMUMGRÆNSE ER 100 KRONER."
1050 GOTO 260
1060 PRINT "HUSETS MINIMUMGRÆNSE ER 1 KRONE."
1070 GOTO 260
1080 GOTO 320
1090 PRINT "BAR";
1100 GOTO 420
1110 PRINT "KLOKKE";
1120 GOTO 420
1130 PRINT "APPELSIN";
1140 GOTO 420
1150 PRINT "CITRON";
1160 GOTO 420
1170 PRINT "BLOMME";
1180 GOTO 420
1190 PRINT "KIRSEBÆR";
1200 GOTO 420
1210 PRINT " BAR";
1220 GOTO 480
1230 PRINT " KLOKKE";
1240 GOTO 480
1250 PRINT " APPELSIN";
1260 GOTO 480
1270 PRINT " CITRON";
1280 GOTO 480
1290 PRINT " BLOMME";
1300 GOTO 480
1310 PRINT " KIRSEBÆR";
1320 GOTO 480
1330 PRINT " BAR"
1340 GOTO 540
1350 PRINT " KLOKKE"
1360 GOTO 540
1370 PRINT " APPELSIN"
1380 GOTO 540
1390 PRINT " CITRON"
1400 GOTO 540
1410 PRINT " BLOMME"
1420 GOTO 540
1430 PRINT " KIRSEBÆR"
1440 GOTO 540
1450 RETURN
1460 RETURN
1470 PRINT
1480 PRINT "DOBBELT!!!"
1490 PRINT "DU VANDT!!!"
1500 P = ((2 * M) + M) + P)
1510 GOTO 600
1520 END

```

# Tændstikker

I spillet Tændstikker begynder du med 23 tændstikker på et bord. Ved hvert træk tager du 1, 2 eller 3 tændstikker. Du skiftes med datamaten, og den, som tager den sidste tændstik, har tabt.

Den letteste måde at komme frem til en god strategi er at begynde ved spillets slutning. Da det er din agt at lade den sidste tændstik blive tilbage, så din modstander kan tage den, skal der ved dit sidste træk være 4, 3 eller 2 tændstikker tilbage, sådan at du kan tage 3, 2 eller 1 og lade en enkelt blive tilbage. Derfor må du ved det forudgående træk lade 5 tændstikker tilbage, så modstanderen (uanset hvor mange han tager) lader 4, 3 eller 2 tilbage. Hvis du på denne måde regner tilbage lige til spillets begyndelse, vil du opdage, at spillet kan vindes på det allerførste træk. Heldigvis lader datamaten dig begynde, så hvis du spiller rigtigt, vinder du altid.

Når du har lært at beherske Tændstikker, kan du gå over til Bunke og senere til Nim.

Denne version af Tændstikker er oprindeligt skrevet af Bob Albrecht fra People's Computer Company.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
TÆNDSTIKKER  
-----
```

```
DETTE SPIL HEDDER 'TÆNDSTIKKER'.
```

```
NÅR DET ER DIN TUR KAN DU TAGE 1, 2  
ELLER 3 TÆNDSTIKKER. DET GÆLDER OM  
IKKE AT TAGE DEN SIDSTE TÆNDSTIK.
```

```
LAD OS SLÅ PLAT OG KRONE OM HVEM DER  
STARTER. HVIS DET BLIVE KRONE, SÅ  
STARTER JEG.
```

```
.....PLAT!, DU STARTER!
```

```
DIN TUR.....  
DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-  
KER. HVOR MANGER TAGER DU ?3
```

```
DER ER NU IALT 20 TILBAGE.
```

```
MIN TUR!!!  
JEG TAGER 1 TÆNDSTIKKER.
```

```
DER ER NU IALT 19 TILBAGE.
```

```
DIN TUR.....  
DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-  
KER. HVOR MANGER TAGER DU ?2
```

```
DER ER NU IALT 17 TILBAGE.
```

```
MIN TUR!!!  
JEG TAGER 2 TÆNDSTIKKER.
```

```
DER ER NU IALT 15 TILBAGE.
```

```
DIN TUR.....  
DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-  
KER. HVOR MANGER TAGER DU ?3
```

```
DER ER NU IALT 12 TILBAGE.
```

```
MIN TUR!!!  
JEG TAGER 1 TÆNDSTIKKER.
```

```
DER ER NU IALT 11 TILBAGE.
```

```
DIN TUR.....  
DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-  
KER. HVOR MANGER TAGER DU ?2
```

```
DER ER NU IALT 9 TILBAGE.
```

```
MIN TUR!!!  
JEG TAGER 2 TÆNDSTIKKER.
```

```
DER ER NU IALT 7 TILBAGE.
```

```
DIN TUR.....  
DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-  
KER. HVOR MANGER TAGER DU ?3
```

```
DER ER NU IALT 4 TILBAGE.
```

```
MIN TUR!!!  
JEG TAGER 3 TÆNDSTIKKER.  
STAKKELS DIG. DU TOG DEN SIDSTE  
TÆNDSTIK. HAR, HAR, JEG FIK DIG!
```

```
FARVEL....NUMMER SJOK!
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          TÆNDSTIKKER"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "  DETTE SPIL HEDDER 'TÆNDSTIKKER'. "  
160 PRINT  
170 PRINT "NÅR DET ER DIN TUR KAN DU TAGE 1, 2"  
180 PRINT "ELLER 3 TÆNDSTIKKER. DET GÆLDER OM"  
190 PRINT "IKKE AT TAGE DEN SIDSTE TÆNDSTIK."  
200 PRINT  
210 PRINT "LAD OS SLÅ PLAT OG KRONE OM HVEM DER"  
220 PRINT "STARTER. HVIS DET BLIVE KRONE, SÅ"  
230 PRINT "STARTER JEG."  
240 N = 23  
250 Q = INT (2 * RND (1))  
260 IF Q = 1 THEN GOTO 310  
270 PRINT  
280 PRINT ".....PLAT!, DU STARTER!"  
290 PRINT  
300 GOTO 410  
310 PRINT  
320 PRINT ".....KRONE, JEG STARTER."  
330 PRINT  
340 PRINT "GØR DIG PARAT TIL ET NEDERLAG!"  
350 PRINT  
360 PRINT "JEG TAGER 2 TÆNDSTIKKER."  
370 N = N - 2  
380 PRINT  
390 PRINT "DER ER NU IALT";N;"TILBAGE."  
400 PRINT  
410 PRINT "DIN TUR....."  
420 PRINT "DU KAN TAGE 1, 2, ELLER 3 TÆNDSTIK-"  
430 PRINT "KER. HVOR MANGER TAGER DU ?";  
440 INPUT K  
450 PRINT  
460 IF K > 3 THEN GOTO 680  
470 IF K < = 0 THEN GOTO 680  
480 N = N - K  
490 PRINT "DER ER NU IALT";N;"TILBAGE."  
500 PRINT  
510 IF N = 4 THEN GOTO 580  
520 IF N = 3 THEN GOTO 600  
530 IF N = 2 THEN GOTO 620  
540 IF N < = 1 THEN GOTO 780  
550 Z = 4 - K  
560 GOTO 630  
570 PRINT  
580 Z = 3  
590 GOTO 630  
600 Z = 2  
610 GOTO 630  
620 Z = 1  
630 PRINT "MIN TUR!!!"  
640 PRINT "JEG TAGER";Z;"TÆNDSTIKKER."  
650 N = N - Z  
660 IF N < = 1 THEN GOTO 720  
670 GOTO 380  
680 PRINT "MEGET MORSOMT, HA, HA, HA."  
690 PRINT "VIL DU SPILLE ELLER PJATTE?"  
700 PRINT "HVOR MANGE TÆNDSTIKKER TAGER DU ?";  
710 GOTO 440  
720 PRINT "STAKKELS DIG. DU TOG DEN SIDSTE"  
730 PRINT "TÆNDSTIK. HAR, HAR, JEG FIK DIG!"  
740 PRINT  
750 PRINT "FARVEL....NUMMER SJOK!"  
760 PRINT  
770 GOTO 820  
780 PRINT "DU VANDT....BLOMKÅLSØRE!!!"  
790 PRINT "DU TROR VIS DU ER SNART....HVA'???"  
800 PRINT "LAD OS SPILLE IGEN, FOR SÅ SKAL DU"  
810 PRINT "FR SÅ HATTEN DEN PASSER!!!"  
820 END
```



# Tårnene i Hanoi

Dette er en simulation af et logisk spil, som kommer fra Mellem-østen. Det kaldes nogle gange for »Faraos pinde«, men det almindeligste navn er Tårnene i Hanoi.

Ifølge sagnet findes der et hemmeligt selskab af munke som bor under byen Hanoi. De har tre store tårne eller pinde, på hvilke store guldskiver i forskellige størrelser kan placeres. Ved at flytte en skive ad gangen og aldrig placere en større skive oven på en mindre forsøger munkene at flytte hele tårnet af skiver fra den venstre pind til den højre pind. Ifølge sagnet vil verden gå under, når det er lykkedes dem at flytte tårnet, der består af 64 skiver. Hvor mange træk skal de foretage for at gøre dette? Hvis de arbejder døgnet rundt og kan flytte en skive i minuttet, hvor mange år vil munkene så være om det?

I datamatudgaven af spillet har du tre lodrette pinde. På den venstre pind mellem 2 og 7 skiver af forskellig størrelse, den største nederst, og den mindste øverst. Det er nu din opgave at flytte alle skiverne til den pind, der står længst til højre. Du må dog kun flytte en skive ad gangen, og du må aldrig placere en skive oven på en, der er mindre.

Skiverne har i spillet navn efter deres størrelse, dvs. den mindste har nummer 3, de næste numrene 5, 7, 9, 11, 13 og 15. Hvis du spiller med færre end 7 skiver, anvendes de højeste værdier altid først. Hvis du derfor bruger to skiver, får de numrene 13 og 15. Programmets instruktioner vil forklare resten. Held og lykke.

Charles Lund skrev dette program, mens han gik i The American School i Haag, Holland.

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

TÅRNE I HANOI

SPILET OM TÅRNE I HANOI.

DET ER DIN OPGAVER AT FLYTTE ALLE PLADERNE FRA DET VENSTRE TIL DET HØJRE TÅRN. DU MÅ KUN FLYTTE EN PLADE AD GANGEN, OG DU MÅ ALDRIG LÆGGE EN STØRRE PLADE PÅ EN MINDRE.

HVOR MANGE PLADER VIL DU FLYTTE (7 ER MAKSIMUM) ?3

I DETTE PROGRAM FORETAGES OVERFØRSELLEN AF PLADERNE VED HJÆLP AF ET TAL. TALLET 3 ER DEN MINDSTE PLADE, TALLET 5 DEN NÆSTE STØRRELSE, TALLET 7 DEN NÆSTE, OSV., HELT OP TIL TALLET 15. HVIS DU UDFØRER SPILET MED 2 PLADER, SÅ HEDDER DEN MINDSTE AF PINDENE NUMMERERES FRA VENSTRE MOD HØJRE, 1, 2 OG 3.

VI BEGYNDER MED ALLE PLADERNE PÅ PIND 1, OG DET GÆLDER OM AT FLYTTE DEM OVER PÅ PIND 3.

HELD OG LYKKE.

```

*           *           *
*           *           *
*           *           *
*           *           *
*****
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?11  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?3

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?13  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?2

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?11  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?2

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?15  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?3

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?11  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?1

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?13  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?3

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****

```

HVILKEN PLADE ØNSKER DU AT FLYTTE ?11  
HVILKEN PIND SKAL PLADEN PLACERES PÅ ?3

```

*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*           *           *
*****
*****
*****

```

TILLYKKE!!!!

DU HAR UDFØRT OPGAVER PÅ 7 TRÆK.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

TAK FOR SPILET!

```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "              TÅRNET I HANOI"
130 PRINT "-----"
140 PRINT
150 REM Initialisering
160 DIM T(7,3)
170 E = 0
180 FOR D = 1 TO 7
190 FOR N = 1 TO 3
200 T(D,N) = 0
210 NEXT N
220 NEXT D
230 PRINT "SPILLET OM TÅRNET I HANOI."
240 PRINT
250 PRINT "DET ER DIN OPGAVER AT FLYTTE ALLE PLADERNE FRA DET VENSTRE"
260 PRINT "TIL DE HØJRE TÅRN. DU MÅ KUN FLYTTE EN PLADE AD GANGEN,"
270 PRINT "OG DU MÅ ALDRIG LÆGGE EN STØRRE PLADE PÅ EN MINDRE."
280 PRINT
290 PRINT "HVOR MANGE PLADER VIL DU FLYTTE (7 ER MAKSIMUM) ";
300 INPUT S
310 PRINT
320 M = 0
330 FOR Q = 1 TO S
340 IF Q = S THEN 470
350 NEXT Q
360 E = E + 1
370 IF E > 2 THEN 400
380 PRINT "DESVÆRRE, DEN OPGAVER KAN JEG IKKE LØSE FOR DIG"
390 GOTO 280
400 PRINT "OK, GENI!..... HVIS DU IKKE KAN SPILLE SPILLET ORDENTLIGT"
410 PRINT "SÅ KAN JEG LIGE SE GODT TAGER MIT SPIL UNDER ARMEN OG GÅ"
420 PRINT "HJEM. FARVEL MED DIG!!!"
430 GOTO 1810
440 PRINT "PLADERNE 13, OG DEN STØRSTE 15. NÅR DU SPILLER MED 3"
450 PRINT "PLADER, HEDDER DE 11, 13 OG 15, OSV."
460 PRINT
470 PRINT "I DETTE PROGRAM FORETAGES OVERFØRSELN AF PLADERNE VED HJÆLP"
480 PRINT "AF ET TAL."
490 PRINT "TALLET 3 ER DEN MINDSTE PLADE, TALLET 5 DEN NÆSTE STØRRE-"
500 PRINT "SE, TALLET 7 DEN NÆSTE, OSV., HELT OP TIL TALLET 15. HVIS"
510 PRINT "DU UDFØRER SPILLET MED 2 PLADER, SÅ HEDDER DEN MINDSTE AF"
520 PRINT "PINDENE NUMMERERES FRA VENSTRE MOD HØJRE, 1, 2 OG 3."
530 PRINT
540 PRINT "VI BEGYNDER MED ALLE PLADERNE PÅ PIND 1, OG DET GÆLDER OM"
550 PRINT "AT FLYTTE DEM OVER PÅ PIND 3."
560 PRINT
570 PRINT "HELD OG LYKKE."
580 PRINT
590 Y = 7
600 D = 15
610 FOR X = S TO 1 STEP - 1
620 T(Y,1) = D
630 D = D - 2
640 Y = Y - 1
650 NEXT X
660 GOSUB 1620
670 PRINT
680 PRINT "HVIKEN PLADE ØNSKER DU AT FLYTTE ";
690 E = 0
700 INPUT D
710 IF (D - 3) * (D - 5) * (D - 7) * (D - 9) * (D - 11) * (D - 13) * (D -
15) = 0 THEN 800
720 PRINT
730 PRINT "ULOVLIGT TRÆK.... DU KAN KUN TASTE 3,5,7,9,11,13 ELLER 15."
740 PRINT
750 E = E + 1
760 IF E > 1 THEN 780
770 GOTO 680
780 PRINT "HOLD OP MED AT SPILDE MIN TID. FIND EN ANDEN AT GENERERE!"
790 GOTO 1810
800 REM Se efter om angiven skive ligger under en anden
810 FOR R = 1 TO 7
820 FOR C = 1 TO 3
830 IF T(R,C) = D THEN 860
840 NEXT C
850 NEXT R
860 FOR Q = R TO 1 STEP - 1
870 IF T(Q,C) = 0 THEN 890
880 IF T(Q,C) < D THEN 910
890 NEXT Q
900 GOTO 940
910 PRINT
920 PRINT "DEN PLADE LIGGER NEDEN UNDER EN ANDEN. VÆLG EN ANDEN."
930 GOTO 670
940 E = 0
950 PRINT "HVIKEN PIND SKAL PLADEN PLACERES PÅ ";
960 INPUT N
970 PRINT
980 IF (N - 1) * (N - 2) * (N - 3) = 0 THEN 1070
990 E = E + 1
1000 IF E > 1 THEN 1040
1010 PRINT "JEG GÅR UD FRA, AT DER ER TALE OM ET FEJLSLAG...DENNE GÅR
1020 PRINT "MEN PAS PÅ! JEG TILLADER KUN 1 FEJL!"
1030 GOTO 950
1040 PRINT "JEG HAR ADVARET DIG, MEN DU HØRER ÅBENBART IKKE EFTER!"
1050 PRINT "FARVEL MED DIG, KARL SMART!"
1060 GOTO 1810
1070 FOR R = 1 TO 7
1080 IF T(R,N) < > 0 THEN 1120
1090 NEXT R
1100 GOTO 1180
1110 REM Se efter om pladen lægges på en større plade
1120 IF D < T(R,N) THEN 1180
1130 PRINT "DU KAN IKKE LÆGGE EN STOR PLADE PÅ EN MINDRE PLADE, DEN"
1140 PRINT "KAN GÅ I STYKKER AF DET!!"
1150 PRINT "SÅ DERFOR, ";
1160 GOTO 670
1170 REM Flyt plade
1180 FOR V = 1 TO 7
1190 FOR W = 1 TO 3
1200 IF T(V,W) = D THEN 1240
1210 NEXT W
1220 NEXT V
1230 REM Find fri plads på pind N
1240 FOR U = 1 TO 7
1250 IF T(U,N) < > 0 THEN 1300
1260 NEXT U
1270 U = 7
1280 GOTO 1310
1290 REM Flyt pladen og sæt tidligere plads til 0
1300 U = U - 1
1310 T(U,N) = T(V,W)
1320 T(V,W) = 0
1330 REM Udskriv
1340 GOSUB 1620
1350 REM Alt i orden?
1360 M = M + 1
1370 FOR R = 1 TO 7
1380 FOR C = 1 TO 2
1390 IF T(R,C) < > 0 THEN 1430
1400 NEXT C
1410 NEXT R
1420 GOTO 1490
1430 IF M < = 128 THEN 670
1440 PRINT
1450 PRINT "JEG ER KED AF AT MÅTTE MEDDELE DIG, AT JEG HAR FRET STRI
1460 PRINT "ORDRER OM AT STOPPE DIG, NÅR DU HAR GJORT 128 TRÆK. SPI
1470 PRINT "ER DERFOR FORBI NU!"
1480 GOTO 1810
1490 IF M < > 2 ^ S - 1 THEN 1520
1500 PRINT
1510 PRINT "TILLYKKE!!!"
1520 PRINT "DU HAR UDFØRT OPGAVERN PÅ";M;"TRÆK."
1530 PRINT
1540 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1550 INPUT A$
1560 IF LEFT$(A$,1) = "N" THEN 1780
1570 IF LEFT$(A$,1) = "J" THEN 140
1580 PRINT
1590 PRINT "INDTAST VENLIGST 'JA' ELLER 'NEJ' ";
1600 INPUT A$
1610 GOTO 1560
1620 REM Udskriftsrutine
1630 FOR K = 1 TO 7
1640 Z = 10
1650 FOR J = 1 TO 3
1660 IF T(K,J) = 0 THEN 1720
1670 PRINT TAB( Z - INT( T(K,J) / 2));
1680 FOR V = 1 TO T(K,J)
1690 PRINT "*";
1700 NEXT V
1710 GOTO 1730
1720 PRINT TAB( Z);"*";
1730 Z = Z + 21
1740 NEXT J
1750 PRINT
1760 NEXT K
1770 RETURN
1780 PRINT
1790 PRINT "TAK FOR SPILLET!"
1800 PRINT
1810 END

```

# Ubådsjagt

I dette spil er du kaptajn på HMS Computer. En fjendtlig ubåd har bevæget sig ind i vore farvande, og det er din opgave at tilintetgøre den. Du kan selv vælge størrelsen på den »terning vand«, i hvilken du ønsker at søge efter ubåden. Derefter bestemmer datamaten, hvor mange dybvandsbomber du har til rådighed.

Hver dybvandsbombe affyres ved at indtaste tre tal, hvor de første to er overfladekoordinater (X,Y), det tredje dybden. Efter hver eksplosion angiver din 'sonar', hvor eksplosionen er i forhold til ubåden.

Dana Nofle skrev dette program, mens han var elev på Acton High School, Acton, Massachusetts.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
UBÅDSJAGT  
-----
```

```
SØGEOMRÅDETS STØRRELSE ?10
```

```
DU ER KAPTAJN PÅ 'HMS COMPUTER'. EN  
FJENDTLIG UBÅD BEFINDER SIG INDEN FOR  
DIT HAVOMRÅDE. DET ER DIN OPGAVE AT  
DESTRUERE DEN.
```

```
DU HAR 4 DYBVANDSBOMBER.
```

```
DU SKAL INDTASTE TRE KOORDINATER, DER  
ANGIVER HVOR DU ØNSKER, AT DYBVANDS-  
BOMBEN SKAL EKSPLODERE.
```

```
DE FØRSTE TO KOORDINATER ANGIVER OVER-  
FLADE-KOORDINATERNE, MENS DEN TREDIE  
ANGIVER DYBDEN.
```

```
HELD OG LYKKE !
```

```
FORSØG NUMMER 1 ?5,5,2
```

```
SONAR RAPPORTERER:  
SKUDET VAR NORDVESTLIGT  
OG RIGTIG DYBDE.
```

```
FORSØG NUMMER 2 ?3,3,2
```

```
SONAR RAPPORTERER:  
SKUDET VAR NORDVESTLIGT  
OG RIGTIG DYBDE.
```

```
FORSØG NUMMER 3 ?8,8,2
```

```
SONAR RAPPORTERER:  
SKUDET VAR NORDØSTLIGT  
OG RIGTIG DYBDE.
```

```
FORSØG NUMMER 4 ?7,4,2
```

```
SONAR RAPPORTERER:  
SKUDET VAR NORDLIGT  
OG RIGTIG DYBDE.
```

```
DU ER BLEVET TORPEDERET...  
FORLAD OMGÆRENDE SKIBET....
```

```
U-BÅDEN VAR PÅ POSITION 7 , 2 , 2
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
DET ER I ORDEN. HÅBER DU HAR MORET DIG.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          UBÅDSJAGT"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "SØGEOMRÅDETS STØRRELSE ";  
160 INPUT G  
170 PRINT  
180 N = INT ( LOG (G) / LOG (2) ) + 1  
190 PRINT "DU ER KAPTAJN PÅ 'HMS COMPUTER'. EN"  
200 PRINT "FJENDTLIG UBÅD BEFINDER SIG INDEN FOR"  
210 PRINT "DIT HAVOMRÅDE. DET ER DIN OPGAVE AT"  
220 PRINT "DESTRUERE DEN."  
230 PRINT  
240 PRINT "DU HAR";N;"DYBVANDSBOMBER."  
250 PRINT  
260 PRINT "DU SKAL INDTASTE TRE KOORDINATER, DER"  
270 PRINT "ANGIVER HVOR DU ØNSKER, AT DYBVANDS-"  
280 PRINT "BOMBEN SKAL EKSPLODERE."  
290 PRINT  
300 PRINT "DE FØRSTE TO KOORDINATER ANGIVER OVER-"  
310 PRINT "FLADE-KOORDINATERNE, MENS DEN TREDIE"  
320 PRINT "ANGIVER DYBDEN."  
330 PRINT  
340 PRINT "HELD OG LYKKE !"  
350 PRINT  
360 A = INT ( G * RND (1) )  
370 B = INT ( G * RND (1) )  
380 C = INT ( G * RND (1) )  
390 FOR D = 1 TO N  
400 PRINT "FORSØG NUMMER";D;  
410 INPUT X,Y,Z  
420 PRINT  
430 IF ABS (X - A) + ABS (Y - B) + ABS (Z - C) =  
0 THEN 530  
440 GOSUB 630  
450 PRINT  
460 NEXT D  
470 PRINT "DU ER BLEVET TORPEDERET..."  
480 PRINT "FORLAD OMGÆRENDE SKIBET...."  
490 PRINT  
500 PRINT "U-BÅDEN VAR PÅ POSITION";  
510 PRINT A;" ";B;" ";C  
520 GOTO 560  
530 PRINT "B A N G ! !"  
540 PRINT  
550 PRINT "DU FANDT UBÅDEN PÅ";D;"FORSØG!"  
560 PRINT  
570 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";  
580 INPUT AS  
590 PRINT  
600 IF LEFT$ (AS,1) = "J" THEN 330  
610 PRINT "DET ER I ORDEN. HÅBER DU HAR MORET DIG."  
620 GOTO 740  
630 PRINT "SONAR RAPPORTERER:"  
640 PRINT "SKUDET VAR ";  
650 IF Y > B THEN PRINT "NORD";  
660 IF Y < B THEN PRINT "SYD";  
670 IF X > (A) THEN PRINT "ØST";  
680 IF X < (A) THEN PRINT "VEST";  
690 IF X < > A OR Y < > B THEN PRINT "LIG":  
PRINT "OG ";  
700 IF Z > C THEN PRINT "FOR DYBT."  
710 IF Z < C THEN PRINT "FOR HØJT."  
720 IF Z = C THEN PRINT "RIGTIG DYBDE."  
730 RETURN  
740 END
```

# Ugedag

Dette program giver dig oplysning om, på hvilken ugedag du blev født (eller ugedagen for en anden interessant dato). Programmet kan dog ikke beregne ugedage for datoer, der ligger før 1582.

Du indtaster blot dags dato (som dag, måned, år) og derefter din fødselsdato på samme måde. Derefter udskriver datamaten den ugedag, du blev født på, hvor gammel du er nu, og hvor meget tid du til i dag har brugt på at spise, sove, arbejde og slappe af.

Programmet er en modificeret udgave af et program fra GE, der var skrevet af Tom Kloos fra Oregon Museum of Science and Industry.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
UGEDAG  
-----
```

```
'UGEDAG' ER ET PROGRAM, DER GIVER DIG  
VIGTIGE OPLYSNINGER OM ENHVER DAG  
SOM MÅTTE INTERESSERE DIG.
```

```
INDTAST DAGS DATO I FORMEN: DAG,MÅNED,  
ÅR (F.EKS. 24,12,1985) ?7,1,1978
```

```
INDTAST DIN FØDSELSDATO (ELLER ANDEN  
DATO SOM INTERESSERER DIG ?2,12,1999
```

```
2 / 12 1999 FALDER PÅ EN TORS DAG
```

```
ÅRUN
```

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
UGEDAG  
-----
```

```
'UGEDAG' ER ET PROGRAM, DER GIVER DIG  
VIGTIGE OPLYSNINGER OM ENHVER DAG  
SOM MÅTTE INTERESSERE DIG.
```

```
INDTAST DAGS DATO I FORMEN: DAG,MÅNED,  
ÅR (F.EKS. 24,12,1985) ?1,12,1977
```

```
INDTAST DIN FØDSELSDATO (ELLER ANDEN  
DATO SOM INTERESSERER DIG ?12,4,1952
```

```
12 / 4 1952 VAR EN LØRDAG
```

```
ÅR MDR DAGE  
-- -- --
```

```
DIN ALDER, HVIS DEN  
INDTASTEDE FØDSELSDATO ER 25 7 19  
DU HAR SOVET 8 11 24  
DU HAR SPIST 4 4 10  
DU HAR ARBEJDET/LEGET 5 10 27  
DU HAR HVILET 6 4 18
```


```
DU KAN TRÆKKE DIG TILBAGE ÅR 2017 .
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "UGEDAG"  
130 PRINT "-----"  
140 PRINT  
150 PRINT  
160 PRINT "'UGEDAG' ER ET PROGRAM, DER GIVER DIG"  
170 PRINT "VIGTIGE OPLYSNINGER OM ENHVER DAG"  
180 PRINT "SOM MÅTTE INTERESSERE DIG."  
190 PRINT  
200 PRINT "INDTAST DAGS DATO I FORMEN: DAG,MÅNED,"  
210 PRINT "ÅR (F.EKS. 24,12,1985) ";  
220 INPUT D1,M1,Y1  
230 PRINT  
240 REM Dette program beregner ugedagen for  
250 REM alle datoer efter 1582.  
260 DEF FN A(A) = INT (A / 4)  
270 DEF FN B(A) = INT (A / 7)  
280 DIM T(12)  
290 REM Indlæs grundværdier for måneder  
300 FOR I = 1 TO 12  
310 READ T(I)  
320 NEXT I  
330 PRINT "INDTAST DIN FØDSELSDATO (ELLER ANDEN"  
340 PRINT "DATO SOM INTERESSERER DIG ";  
350 INPUT D,M,Y  
360 PRINT  
370 I1 = INT ((Y - 1500) / 100)  
380 REM Er dagen efter 1582?  
390 IF Y - 1582 < 0 THEN 1450  
400 A = I1 * 5 + (I1 + 3) / 4  
410 I2 = INT (A - FN B(A) * 7)  
420 Y2 = INT (Y / 100)  
430 Y3 = INT (Y - Y2 * 100)  
440 A = Y3 / 4 + Y3 + D + T(M) + I2  
450 B = INT (A - FN B(A) * 7) + 1  
460 IF M > 2 THEN 570  
470 IF Y3 = 0 THEN 540  
480 T1 = INT (Y - FN A(Y) * 4)  
490 IF T1 < > 0 THEN 570  
500 IF B < > 0 THEN 520  
510 B = 6  
520 B = B - 1  
530 GOTO 570  
540 A = I1 - 1  
550 T1 = INT (A - FN A(A) * 4)  
560 IF T1 = 0 THEN 500  
570 IF B < > 0 THEN 590  
580 B = 7  
590 IF (Y1 * 12 + M1) * 31 + D1 < (Y * 12 + M) * 31  
+ D THEN 650  
600 IF (Y1 * 12 + M1) * 31 + D1 = (Y * 12 + M) * 31  
+ D THEN 630  
610 PRINT D;" / ";M;"Y;"VAR EN "  
620 GOTO 670  
630 PRINT D;" / ";M;"Y;"ER EN "  
640 GOTO 670  
650 PRINT D;" / ";M;"Y;"FALDER PÅ EN "  
660 REM Udskriv den ugedag som datoer falder på  
670 IF B < > 1 THEN 690  
680 PRINT "SØNDAG"  
690 IF B < > 2 THEN 710  
700 PRINT "MANDAG"  
710 IF B < > 3 THEN 730  
720 PRINT "TIRSDAG"  
730 IF B < > 4 THEN 750  
740 PRINT "ONSDAG"  
750 IF B < > 5 THEN 770  
760 PRINT "TORS DAG"  
770 IF B < > 6 THEN 790  
780 GOTO 1400  
790 IF B < > 7 THEN 810  
800 PRINT "LØRDAG"  
810 IF (Y1 * 12 + M1) * 31 + D1 = (Y * 12 + M) * 31  
+ D THEN 1350  
820 I5 = Y1 - Y  
830 PRINT  
840 I6 = M1 - M  
850 I7 = D1 - D  
860 IF I7 > = 0 THEN 890  
870 I6 = I6 - 1  
880 I7 = I7 + 30  
890 IF I6 > = 0 THEN 920  
900 I5 = I5 - 1  
910 I6 = I6 + 12  
920 IF I5 < 0 THEN 1470
```

```

930 IF I7 < > 0 THEN 970
940 IF I6 < > 0 THEN 970
950 PRINT "***** TILLYKKE MED FØDSELSDAGEN *****"
960 PRINT
970 PRINT "                ÅR MDR DAGE"
980 PRINT "                -- ----"
990 PRINT "DIN ALDER, HVIS DEN"
1000 PRINT "INDTASTEDE FØDSELSDATO ER ";
1010 GOSUB 1670
1020 A8 = (I5 * 365) + (I6 * 30) + I7 + INT (I6 / 2)
1030 K5 = I5
1040 K6 = I6
1050 K7 = I7
1060 REM Beregn pensionsalder
1070 E = Y + 65
1080 REM Beregn tid brugt på følgende ting
1090 F = .35
1100 PRINT "DU HAR SOVET          ";
1110 GOSUB 1530
1120 F = .17
1130 PRINT "DU HAR SPIST          ";
1140 GOSUB 1530
1150 F = .23
1160 IF K5 > 3 THEN 1190
1170 PRINT "DU HAR LEGET          ";
1180 GOTO 1230
1190 IF K5 > 9 THEN 1220
1200 PRINT "DU HAR LEGET/STUDERET ";
1210 GOTO 1230
1220 PRINT "DU HAR ARBEJDET/LEGET ";
1230 GOSUB 1530
1240 GOTO 1750
1250 PRINT "DU HAR HVILET          ";
1260 IF K5 > 9 AND K5 < 100 THEN PRINT " ";: GOTO
1280
1270 IF K5 < 10 THEN PRINT " ";
1280 PRINT K5;
1290 IF K6 < 10 THEN PRINT " ";
1300 PRINT K6;
1310 IF K7 < 10 THEN PRINT " ";
1320 PRINT K7
1330 PRINT
1340 PRINT "DU KAN TRÆKKE DIG TILBAGE ÅR";E;". "
1350 PRINT
1360 PRINT
1370 PRINT
1380 PRINT
1390 GOTO 1800
1400 IF D = 13 THEN 1430
1410 PRINT "FREDAG "
1420 GOTO 810
1430 PRINT "FREDAG DEN 13. !!! PAS PÅ!"
1440 GOTO 810
1450 PRINT "JEG KAN IKKE BEREGNE DATOER, DER"
1460 PRINT "LIGGER FØR ÅR MDLXXXII."
1470 GOTO 1370
1480 REM Tabel for månedsværdier der
1490 REM bruges i beregningen
1500 DATA 0,3,3,6,1,4,6,2,5,0,3,5
1510 REM Denne dato bruges nu i beregningerne
1520 REM Beregn tid i år, måneder og dage
1530 K1 = INT (F * A8)
1540 I5 = INT (K1 / 365)
1550 K1 = K1 - (I5 * 365)
1560 I6 = INT (K1 / 30)
1570 I7 = K1 - (I6 * 30)
1580 K5 = K5 - I5
1590 K6 = K6 - I6
1600 K7 = K7 - I7
1610 IF K7 > = 0 THEN 1640
1620 K7 = K7 + 30
1630 K6 = K6 - 1
1640 IF K6 > 0 THEN 1670
1650 K6 = K6 + 12
1660 K5 = K5 - 1
1670 IF I5 > 9 AND I5 < 100 THEN PRINT " ";: GOTO
1690
1680 IF I5 < 10 THEN PRINT " ";
1690 PRINT I5;
1700 IF I6 < 10 THEN PRINT " ";
1710 PRINT I6;
1720 IF I7 < 10 THEN PRINT " ";
1730 PRINT I7
1740 RETURN
1750 IF K6 = 12 THEN 1770
1760 GOTO 1250
1770 K5 = K5 + 1
1780 K6 = 0
1790 GOTO 1250
1800 END

```



opåh .

# Appendix

## Basic-test

Dette program bør indtastes og køres før bogens andre programmer.

Ved hjælp af BASIC-TEST vil du hurtigt være i stand til at undersøge hvilke af bogens BASIC-instruktioner, der fejlfortolkes af den BASIC din datamat bruger. Indtast programmet omhyggeligt og kørs det. Hvis din computer kan køre programmet og giver de samme resultater (bortset fra RND-funktionen – selvfølgelig) som nedenstående kørsel, så vil du ikke få problemer med bogens programmer. Husk dog, at de værdier som funktionerne beregner heller ikke behøver at passe på de sidste decimaler. Hvis BASIC-TEST standser midt i kørslen eller giver fejlmeddelelse, så find ud af, hvilken programlinie der ikke udføres som forventet. Arbejd med problemet og prøv forskellige muligheder, indtil din version af BASIC-TEST udskriver de samme resultater som prøvekørslen. På den måde kan du være sikker på, at problemet er løst tilfredsstillende, og at den samme løsning vil kunne anvendes i bogens andre programmer.

Programmet er skrevet af Christian Borup.

```
-----
                        BASIC-TEST
-----

*** DETTE ER ET BASIC TEST PROGRAM ***

  3.333
-3.333

*** INDBYGGEDE FUNKTIONER ***

-190240727
-981737473
.193779633
1.27931201
28.0222826
1.2038728
1.82565057
5.555      0      5.555
-6         0      5
-1         0      1

OBS!
NEDENSTÅENDE VÆRDIER KAN AFVIGE PÅ DIN
COMPUTER, DE SKAL BLOT VÆRE STØRRE END
O OG MINDRE END 1

.93753437   .0544707574
.246017374   .220233661
.975331284   .253108841
.869225068   .533446289
.892159924   .225467183

***** ** *

*** HOP OG LØKKER ***

GOTO UDFØRT
GOSUB UDFØRT
IF-THEN UDFØRT
IF-THEN GOSUB UDFØRT
ON-GOTO UDFØRT
ON-GOSUB UDFØRT

-2      -2
-2      -1
-2      0
-2      1
-2      2
0       0
0       1
0       2
2       2

*** DATA-SÆTNINGER ***

5 4 3 2 1
EDCBA54321

1

*** OPERATORER ***

-29.623704
-17.0332965

0      0
1      1
0      0
1      1

*** BRUGER-DEFINERED E FUNKTIONER ***

-44      -36.663

*** STRENGE ***

65      A
66      B
67      C
68      D
69      E

5      1
4      2
3      3
2      4
1      5

13

COM*PUTER S*PIL

3-2-1-0

*** DENNE LØKKE RINGER KLOKKEN ***

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
100 PRINT
110 PRINT "-----"
120 PRINT "          BASIC-TEST"
130 PRINT "-----"
140 REM DETTE ER EN REMARK-SÆTNING
150 PRINT
160 A = 0:B = 2
170 C = 3.333:C1 = - 3.333
180 D = 5.555:D1 = - 5.555
190 K$ = "COMPUTER SPIL"
200 DIM A(5),B(2,2),A$(10)
210 PRINT "*** DETTE ER ET BASIC TEST PROGRAM ***"
220 PRINT
230 PRINT C
240 PRINT C1
250 PRINT
260 PRINT "*** INDBYGGEDE FUNKTIONER ***"
270 PRINT
280 PRINT SIN (C)
290 PRINT COS (C)
300 PRINT TAN (C)
310 PRINT ATN (C)
320 PRINT EXP (C)
330 PRINT LOG (C)
340 PRINT SQR (C)
350 PRINT ABS (D1), ABS (A), ABS (D)
360 PRINT INT (D1), INT (A), INT (D)
370 PRINT SGN (D1), SGN (A), SGN (D)
380 PRINT
390 PRINT "OBS!"
400 PRINT "NEDENSTÅENDE VÆRDIER KAN AFVIGE PÅ DIN"
410 PRINT "COMPUTER, DE SKAL BLOT VÆRE STØRRE END"
420 PRINT "O OG MINDRE END 1"
430 PRINT
440 FOR I = 1 TO 5
450 PRINT RND (1), RND (1)
460 NEXT I
470 PRINT
480 PRINT "*****"; TAB (12); "*"
490 PRINT TAB (13); "*" ; TAB (15); "*"
500 PRINT
510 PRINT
520 PRINT "HOP OG LØKKER ***"
530 PRINT
540 GOTO 1370
550 GOSUB 1390
560 IF C = 3.333 THEN 1410
570 IF C = 3.333 THEN GOSUB 1430
580 IF A < > 0 THEN 1450: PRINT "DENNE BASICS
IF-THEN PROCEDURE FORKERT": GOTO 580
590 ON B GOTO 1470,1450,1470
600 ON B GOSUB 1490,1510,1490
610 PRINT
620 FOR I = - 2 TO 2 STEP 2
630 FOR J = I TO 2
640 PRINT I,
650 PRINT J
660 NEXT J
670 NEXT I
680 PRINT
690 PRINT "DATA-SÆTNINGER ***"
700 PRINT
710 DATA 5,4,3,2,1
720 DATA "E","D","C","B","A"
730 DATA "5","4","3","2","1"
740 FOR I = 1 TO 5
750 READ A(I)
760 PRINT A(I);
770 NEXT I
780 PRINT
790 FOR I = 1 TO 10
800 READ A$(I)
810 PRINT A$(I);
820 NEXT I
830 PRINT
840 PRINT
850 RESTORE
860 FOR J = 1 TO 5
870 READ M
880 NEXT J
890 PRINT M
900 PRINT
910 PRINT "OPERATORER ***"
920 PRINT
```

```

930 PRINT (C + C) * D1 / B - C1 ^ 2
940 PRINT C + C * D1 / B - C1 ^ 2
950 PRINT ABS (1 < > 1), ABS (2 < 1)
960 PRINT ABS (1 = 1), ABS (2 > 1)
970 PRINT ABS ((1 > = 1) AND (2 > = 3))
980 PRINT ABS ((1 > = 1) OR (2 > = 3))
990 PRINT
1000 PRINT "*** BRUGER-DEFINERED FUNKTIONER ***"
1010 PRINT
1020 DEF FN R(X) = X * 10 + X / (X / X)
1030 X = - 4
1040 PRINT FN R(X), FN R(C1)
1050 PRINT
1060 PRINT "*** STRENGE ***"
1070 PRINT
1080 FOR I = 5 TO 1 STEP - 1
1090 PRINT ASC (A$(I)), CHR$ (70 - I)
1100 NEXT I
1110 PRINT
1120 FOR I = 10 TO 6 STEP - 1
1130 PRINT STR$ (I - 5), VAL (A$(I))
1140 NEXT I
1150 PRINT
1160 PRINT LEN (K$)
1170 PRINT
1180 PRINT LEFT$ (K$,3);"*";
1190 PRINT MID$ (K$,4,7);"*";
1200 PRINT RIGHT$ (K$,3)
1210 PRINT
1220 A$ = A$(8) + "-" + A$(9) + "-" + A$(10) + "-0"
1230 PRINT A$
1240 PRINT
1250 PRINT "*** DENNE LØKKE RINGER KLOKKEN ***"
1260 FOR I = 1 TO 5
1270 PRINT CHR$ (7);
1280 NEXT I
1290 PRINT
1300 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1310 INPUT A$
1320 PRINT
1330 RESTORE
1340 IF LEFT$ (A$,1) = "J" THEN 210
1350 GOTO 1360
1360 END
1370 PRINT "GOTO UDFØRT"
1380 GOTO 550
1390 PRINT "GOSUB UDFØRT"
1400 RETURN
1410 PRINT "IF-THEN UDFØRT"
1420 GOTO 570
1430 PRINT "IF-THEN GOSUB UDFØRT"
1440 RETURN
1450 PRINT "ON-GOTO UDFØRT"
1460 GOTO 600
1470 PRINT "FEJL I ON-GOTO"
1480 GOTO 600
1490 PRINT "FEJL I ON-GOSUB"
1500 RETURN
1510 PRINT "ON-GOSUB UDFØRT"
1520 RETURN
1530 END

```

# Fortegnelse over spillene opdelt i kategorier

## Morskab

FRASEGENERATOR · 69, bind 1  
GODDAV · 70, bind 1  
NAVN · 72, bind 2  
POESI · 21, bind 3  
RUSSISK ROULETTE · 30, bind 3  
SAKS-STEN-PAPIR · 31, bind 3  
UGEDAG · 72, bind 3

## Undervisning

AKTIER · 12, bind 1  
BORGERKRIG · 44, bind 1  
BYTTEPENGE · 53, bind 1  
DYRERIGET · 63, bind 1  
HAMMURABI · 9, bind 2  
HÆNGT MAND · 18, bind 2  
HØVDING · 22, bind 2  
KAST · 32, bind 2  
KEMIKER · 35, bind 2  
KONGE · 36, bind 2  
LITTERATUR QUIZ · 52, bind 2  
MATEMATISK TERNING · 62, bind 2  
PELSHANDLER · 14, bind 3  
SYNONYM · 57, bind 3  
TOG · 62, bind 3

## Billeder og kurver

BANNER · 28, bind 1  
BILLE · 33, bind 1  
HOP · 16, bind 2  
KALENDER · 27, bind 2  
KANIN · 31, bind 2  
LABYRINT · 47, bind 2  
LIV · 53, bind 2  
LIV FOR TO · 56, bind 2  
LOVE · 58, bind 2  
PLOT I 3-D · 20, bind 3  
RUDERE · 29, bind 3  
SINUSKURVE · 32, bind 3

## Gæt tal eller bogstaver

BOGSTAV · 36, bind 1  
GÆT · 8, bind 2  
HØJT-LAVT · 21, bind 2  
OMRING · 10, bind 3  
STJERNER · 46, bind 3  
TAL · 61, bind 3

## Fjern objekter

BUNKE · 52, bind 1  
LIGE VINDER · 49, bind 2  
LIGE VINDER, VERSION 2 · 50, bind 2  
NIM · 8, bind 3  
TÆNDSTIKKER · 68, bind 3

## Koordinatsystemer

BOMBARDEMENT · 39, bind 1  
HURKLE · 17, bind 2  
PIZZA · 18, bind 3  
SLAGSKIB · 34, bind 3  
SMUGORM · 39, bind 3  
SÆNKE SLAGSKIBE · 58, bind 3  
UBÅDSJAGT · 70, bind 3

## Logik

AWARI · 22, bind 1  
BAGELS · 24, bind 1  
BAGLÆNS · 26, bind 1  
BONDESPIL · 41, bind 1  
CIFFER · 54, bind 1  
DAM FOR EN · 59, bind 1  
DRONNINGESKAK · 61, bind 1  
FLIP-FLOP · 68, bind 1  
HAPS · 11, bind 2  
KRYDS OG BOLLE · 40, bind 2  
KRYDS OG BOLLE, VERSION 2 · 41, bind 2  
KRYDS OG BOLLE I 3-D · 42, bind 2  
KUBUS · 45, bind 2  
MASTER MIND · 59, bind 2  
NICOMACHUS · 7, bind 3  
ORD · 13, bind 3  
SOLITAIRE · 41, bind 3  
TÅRNET I HANOI · 69, bind 3

## Rumfart

MÅL · 63, bind 2  
MÅNELANDING · 65, bind 2  
MÅNELANDING, L.E.M. · 67, bind 2  
MÅNELANDING, RAKET · 70, bind 2  
ORBIT · 11, bind 3  
SPLAT · 44, bind 3  
SUPER STAR TREK INSTRUKTIONER · 47, bind 3  
SUPER STAR TREK X · 50, bind 3

## Sport

AMERIKANSK FODBOLD · 15, bind 1  
AMERIKANSK FODBOLD, VERSION 2 · 19, bind 1  
BASKETBALL · 30, bind 1  
BOKSNING · 37, bind 1  
BOWLING · 49, bind 1  
GOLF · 72, bind 1  
ISHOCKEY · 23, bind 2  
PILESPIL · 16, bind 3  
SLALOM · 37, bind 3  
TYREFÆGTNING · 64, bind 3

## Kasino og hasard

CRAPS · 56, bind 1  
ENOGTYVE · 65, bind 1  
HESTEVÆDDELØB · 14, bind 2  
POKER · 22, bind 3  
ROULETTE · 26, bind 3  
TERNING · 62, bind 3  
TYVEKNÆGT · 66, bind 3

## Kort- og brætspil

DAM · 57, bind 1  
GOMOKO · 7, bind 2  
HASARD · 13, bind 2  
KRIG · 39, bind 2

## Krigsspil

KAMP · 29, bind 2  
KAST BOMBERNE · 33, bind 2  
SKYTTE · 33, bind 3





# Borgens Computerbøger

David Ahl

## **Basic computerspil**

Bind 1-3, 80 sider pr. bd., ill.

Klassisk samling af spil til computere. Spillene er skrevet i BASIC og vil kunne bruges på de fleste computere.

Barker, Jacobi og Zimmermann

## **Mikrodatamaten som tegneredskab**

347 sider, ill.

Om at fremstille tegninger ved hjælp af mikrocomputere. Bogen er også håndbog i brugen af MONSTER-programmet, som gør datamaten til en tegnemaskine.

X.T. Bui

## **Bedre beslutninger med BASIC**

234 sider, ill.

Programsamling til brug for beslutningstagere. Behandler brugen af PERT, prognosemodeller, nulpunktsanalyse, investeringsmodeller m.m.

Oluf Danielsen

## **EDB i hverdagen**

42 sider, ill.

Debatbog til brug i folkeskolens samtidsorientering.

Andreas Dripke

## **Maskinkodeprogrammering med 6502**

183 sider

Bogen, som henvender sig til begyndere, gennemgår instruktionssættet for 6502-processoren og forklarer brugen af en assembler.

Walter A. Ettl

## **Wordstar på en nem måde**

192 sider

Indføring i brugen af det meget udbredte tekstbehandlingsprogram. Indeholder eksempler på færdige opstillinger af breve m.m. Velegnet som både opslagsbog og begyndervejledning.

Jørgen Feder

## **Pascal – også for begyndere**

ca. 385 sider

Den mest omfattende bog på dansk om programmeringssproget Pascal. Fungerer både som indføring og håndbog. Dækker de meget udbredte Poly-Pascal og Turbo-Pascal. Fyldig behandling af brugen af pointere.

Annie Fox og David Fox

## **Lænestols-BASIC**

304 sider, ill.

Helt elementær, humoristisk indføring i programmeringssproget BASIC. Kræver ikke adgang til en computer.

Eugene Galanter

## **Computeren – hverdagsredskab i hjem og skole**

240 sider, ill.

Lettilgængelig bog om børn og computere. Henvender sig til forældre og pædagoger.

Goldschlager og Lister

## **Datalogi og datamaskiner**

320 sider

Dybtgående gennemgang af moderne databehandlingsmetoder med hovedvægten på algoritmebegrebet og principperne for hvordan opgaver omformes til algoritmer.

Tord-Jöran Hallberg

## **Sådan virker mikrodatamaten**

100 sider, ill.

Gennemgår en tænkt mikrodatamat MM og forklarer i detaljer hvordan den programmeres i maskinkode.

Fietz, Kitza og Mantz

## **Håndbog i MSDOS/PCDOS**

ca. 180 sider

Bogen falder i to dele: Først fortælles om hvad operativsystemet er, hvilke opgaver det løser og hvordan man arbejder med det. Bogens anden del er et referenceafsnit, hvor hver enkelt kommando gennemgås. Bogen indeholder et appendiks om nogle af de nye kommandoer i version 3.0 og senere.

Ken Knecht

## **Microsoft BASIC**

261 sider

Populær og omhyggelig gennemgang af denne meget udbredte version af BASIC med mange programeksempler, herunder læsning og skrivning i filer.

Jørn Lorentzen og Henrik Nellager

## **Maskinkode med Amstrad**

340 sider

Om maskinkodeprogrammering på Amstrad CPC464, 664 og 6128. Forfatterne starter med Z80 processorens instruktionssæt og behandler derefter Amstrad-computerens særlige egenskaber m.h.t. grafik, farver og lyd. Stort afsnit om firmwarerutiner, der kan bruges som byggesten i egne programmer.

Jørn Lorentzen og Henrik Nellager

## **Maskinkode med ZX81**

240 sider

Populær begynderbog med mange eksempler, herunder komplet Space Invader spil.

Murtha og Waite

## **CP/M for begyndere**

208 sider

Om CP/M vers. 2.2. Beskriver operativsystemets virkemåde og opgaver. Læseren lærer assemblerprogrammering med DDT og ASM og bogen giver nyttige tips om brugen af STAT, SYSGEN, PIP og andre hjælpeprogrammer.

Glenford J. Myers

**Kunsten at teste edb-programmer**

246 sider

Fortæller om fejlfinding og afprøvning af edb-programmer. Opstiller et sæt regler og principper, som man kan gå frem efter, når programmer skal fejlrettes.

Erwin Neutzsky-Wulff

**Mikrodatamaten**

Programmering og anvendelse – en bog om ZX81 BASIC

280 sider

Underholdende og lærerig bog om programmering af hjemme-computeren ZX81.

Erwin Neutzsky-Wulff

**Programmering med Commodore BASIC**

220 sider, ill.

Om programmering af hjemmecomputeren VIC20. Mange program-eksempler. Kan også bruges med PET og de lidt større Commodore maskiner.

Erwin Neutzsky-Wulff

**BASIC med Commodore 64**

218 sider, ill.

Bogen er skrevet specielt til Commodore 64 hjemmecomputeren. Indeholder mange morsomme spil og lege som kan testes ind i computeren. Ideel som BASIC-lærebog til selvstudium.

Erwin Neutzsky-Wulff

**Amstrad BASIC**

200 sider

Bogen er skrevet til brug med Amstrad CPC464, 664 og 6128. Ideel som BASIC-lærebog til selvstudium. Udforsker computerens muligheder og egenskaber. Indeholder mange underholdende spil og lege, der er lige klar til at blive tastet ind i computeren.

Lon Poole og Mary Borchers

**En samling nyttige BASIC-programmer**

339 sider, ill.

Dækker et bredt spektrum af almindeligt forekommende beregnings-problemer inden for rentesregning, geometri, statistik m.m.

Lon Poole

**IBM PC – den personlige computer**

480 sider, ill.

Populær bog med hovedvægten på BASIC-programmering med IBM PC. Desuden en fyldig vejledning i brugen af operativsystemet PCDOS. Bogen kan også bruges sammen med IBM PC/XT og AT og med IBM-kompatible computere.

Inge-Lise Salomon

**Elementær datalære** for HF og tilsvarende undervisningstrin

304 sider, ill.

Fortæller om edb-maskiner, anvendelse af edb, mennesker og edb, samt programmering i COMAL. En grundig bog, der er velegnet til aften-skole og selvstudium, samt undervisning i gymnasiet.

Alan Simpson

**Lotus 1-2-3**

Det elektroniske kolonneark

332 sider, ill.

Vejledning i brugen af Lotus 1-2-3 regnearket. De forskellige faciliteter gennemgås, herunder brugen af grafik og database, samt brugen af makroer, der gør 1-2-3 til et meget kraftfuldt værktøj.

Ian Sinclair

**ZX Spectrum** og hvordan man får glæde af den

180 sider

Populær gennemgang af ZX Spectrum hjemmecomputeren.

Jesper Skavin

**Maskinkodeprogrammering med Z80**

224 sider

Instruktionssættet for Z80 processoren gennemgås og instruktionernes virkemåde forklares.

Ian Stewart og Robin Jones

**Lær BASIC-programmering med ZX Spectrum**

250 sider

Letforståelig bog om BASIC-programmering på ZX Spectrum hjemmecomputeren. Mange færdige programeksempler.

Ian Stewart og Robin Jones

**Maskinkode med Commodore 64**

200 sider, ill.

Bogen fortæller om hvordan man fra BASIC kan lave maskinkode-programmer. Mange underholdende og lærerige programeksempler.

Ludwigs, Poppensieker og Surowiecki

**UNIX for begyndere**

ca. 225 sider

Gennemgår UNIX-operativsystemet, der vinder stigende udbredelse i flerbrugersystemer.

Niels Søndergaard

**Lav dine egne computereventyr med BASIC**

200 sider, ill.

Første bog på dansk om en ny og underholdende anvendelse af computere. Computereventyr er tekstspil, hvor computeren på skærmen forklarer hvad der sker, og spørger dig hvad du vil gøre. Du finder dig måske i en ufremkommelig jungle, hvor du skal finde en skat og forsvare dig mod alskens farer.

Carl Townsend

**Programmering med dBASE II**

250 sider

Bogen fortæller detaljeret hvordan man selv kan bruge dBASE II til opbygning af databaser til mange forskellige formål. Talrige eksempler.









**Her er en samling af de bedste spil til computere.**

**Hvert spil er omtalt med en indledning, en listning af programmet lige klar til at taste ind, og et prøvespil der viser, hvordan spillet forløber.**

**Alle spillene er skrevet i Microsoft BASIC, som i forskellige udgaver findes på flertallet af microcomputere.**

**Bogen hjælper dig med at tilpasse spillene, hvis du har en computer med en anden BASIC.**

**Bogen henvender sig til alle brugere af mikrocomputere.**

**Bind 3**

**BORGEN**

ISBN 87-418-7686-5

